

GENEVE

The
MYARC 9640
Family Computer

MYARC ADVANCED BASIC

User's Manual

MYARC, Inc.
Basking Ridge, NJ

COPYRIGHT

Copyright ©1986 by MYARC, INC. All rights reserved. No part of this publication may be reproduced without the written permission of MYARC, INC., P.O. Box 140, Basking Ridge, NJ 07920

DISCLAIMER OF WARRANTY

MYARC, INC. makes no representation or warranties with respect to the contents hereof and specifically disclaims any implied warranties of merchantability or fitness for any particular purpose. MYARC, INC. software is sold or licensed "as is." The risk as to its quality and performance is with the buyer and not MYARC, INC. Further MYARC reserves the right to revise this publication and to make changes in the content hereof without obligations of MYARC to notify any person of such revisions or changes. MYARC also reserves the right to make design revisions or changes without obligations of MYARC to notify any person of such revisions or changes.

MYARC Advanced BASIC

This Manual contains an alphabetical listing of all MYARC Advanced BASIC commands, statements and functions with detailed explanations on each. The Appendix Section provides significant reference details that you will find necessary for effective programming.

MYARC Advanced BASIC is totally upward compatible with MYARC Extended BASIC II and with TI Extended BASIC so that you are already familiar with nearly all the referenced commands, statements, and functions.

In addition to many new commands, statements and functions that were not in MYARC Extended BASIC II, MYARC Advanced BASIC provides additional speed, power, flexibility, and/or sophistication.

Several MYARC Extended BASIC II commands and statements are no longer used in MYARC Advanced BASIC and many commands and statements that were used in MYARC Extended BASIC II have been revised and/or their descriptions modified to reflect the added flexibility that the User now will have and can take advantage of in MYARC Advanced BASIC.

To simplify I/O communication with external devices, a set of default I/O commands has been added to MYARC Advanced BASIC. These commands are described separately in the section "I/O Default Commands".

Different from MYARC Extended BASIC II and TI Extended BASIC, in MYARC Advanced BASIC the function "Break" is invoked by simultaneously depressing both the Control and Break keys. Accordingly wherever in this manual reference is made to "CLEAR", press the two keys, CONTROL + BREAK.

WE RECOMMEND THAT YOU CAREFULLY REVIEW THIS
ENTIRE MANUAL BEFORE PROCEEDING WITH ANY
SERIOUS PROGRAMMING.

MYARC Advanced BASIC

This manual contains information on the MYARC Advanced BASIC system. It is intended for users who are familiar with the MYARC system and who wish to learn more about the Advanced BASIC system. The information in this manual is intended to help you understand the system and to help you use it effectively.

The MYARC Advanced BASIC system is a powerful and flexible system that allows you to write and execute programs in a high-level language. It is designed to be easy to use and to provide a wide range of features and options. The system is designed to be used on a variety of computers and operating systems.

The MYARC Advanced BASIC system is designed to be used on a variety of computers and operating systems. It is designed to be easy to use and to provide a wide range of features and options. The system is designed to be used on a variety of computers and operating systems.

The MYARC Advanced BASIC system is designed to be used on a variety of computers and operating systems. It is designed to be easy to use and to provide a wide range of features and options. The system is designed to be used on a variety of computers and operating systems.

The MYARC Advanced BASIC system is designed to be used on a variety of computers and operating systems. It is designed to be easy to use and to provide a wide range of features and options. The system is designed to be used on a variety of computers and operating systems.

The MYARC Advanced BASIC system is designed to be used on a variety of computers and operating systems. It is designed to be easy to use and to provide a wide range of features and options. The system is designed to be used on a variety of computers and operating systems.

MYARC Advanced BASIC
Version 1.0
Copyright 1980

TABLE OF CONTENTS

COMMANDS, STATEMENTS and FUNCTIONS

ABS	9
*ACCEPT	10
ASC	13
ATN	14
**BCOLOR	14
**BEEP	15
BREAK	16
**BTIME	18
BYE	18
*CALL	19
**CDBL	20
CHAR	21
*CHARPAT	25
*CHARSET	26
CHR\$	27
**CINT	28
*CIRCLE	29
CLEAR	30
*CLOSE	31
**CLS	32
*COLOR	35
CONTINUE	37
COS	38
**CREAL	39
**CSNG	39
DATA	40
**DATE/DATE\$	42
DCOLOR	43
DEF	44
**DEFvartype	46
*DELETE	48
DELSPRITE	49
*DIM	50
*DISPLAY	52
DISPLAY...USING	55
DISTANCE	56
*DRAW	58
*DRAWTO	60
*END	62
EOF	63

ERR	64
EXP	66
**FILES	67
FILL	68
FOR TO	70
FREESPACE	73
*GCHAR	74
GOSUB	76
GOTO	78
*GRAPHICS	79
HCHAR	82
**HEX\$	83
IF THEN ELSE	84
IMAGE	86
INIT	89
INPUT	90
INT	94
JOYST	95
*KEY	96
**KILL	100
**LEFT\$	101
LEN	102
LET	103
LINK	105
LINPUT	106
LIST	108
**LLIST	110
LOAD	112
LOCATE	114
LOG	115
**LPR	116
**LPT	116
**LTRACE	117
MAGNIFY	118
*MARGIN	121
MAX	123
**MEMSET	124
MERGE	125
MIN	127
**MOD	128
MOTION	129
**MOUSE	130
*NEW	132
NEXT	133

*NUMBER	134
OLD	137
*ON BREAK	138
ON ERROR	140
ON GOSUB	142
ON GOTO	144
ON WARNING	146
OPEN	148
OPTION BASE	151
**OUT	152
PATTERN	152
PEEK	154
PEEKV	156
PI	157
POINT	158
POKEV	159
POS	161
POSITION	163
**PPR	164
PRINT	164
PRINT USING	169
RANDOMIZE	170
READ	171
REC	172
RECTANGLE	174
REM	176
RESEQUENCE	177
RESTORE	178
RETURN	180
**RIGHT\$	182
RND	183
RPT\$	184
RUN	185
SAVE	187
SAY	189
SCREEN	190
SEG\$	192
SGN	193
SIN	194
SOUND	195
**SPEED	197
SPGET	198
SPRITE	199
SQR	205
STOP	206

STR\$	207
SUB	208
SUBEND	211
SUBEXIT	212
**SWAP	213
TAB	214
TAN	215
TERMCHAR	216
**TIME/TIME\$	218
TRACE	219
UNBREAK	220
VAL	221
VALHEX	222
VCHAR	223
VERSION	225
**WEND	225
**WHILE	226
**I/O DEFAULT COMMANDS	227
CHDIR LPT PCM PPT	
COM MDM PMD PWD	
APPENDICES	229
*Appendix A: List of Commands, Statements, and Functions	230
*Appendix B: ASCII Code	232
Appendix C: Musical Tone Frequencies	234
Appendix D: Character Sets	235
Appendix E: Pattern-Identifier Conversion Table	235
Appendix F: Color Codes	236
Appendix G: Mathematical Functions	236
Appendix H: List of Speech Words	237
Appendix I: Adding Suffixes to Speech Words	240
Appendix J: Error Messages	246
**Appendix K: Summary of Graphics Modes	251
**Appendix L: Program Illustrating MOUSE Commands	252
**Appendix M: Additional Extended ASCII Codes for Keyboard Mode 6 ..	253

NOTES:

- * Revised from MYARC Extended BASIC II.
- ** New commands, statements or functions.

ABS

ABS

Format

ABS(numeric-expression)

Type

Numeric (REAL or DEFINT)

Description

The ABS function gives the absolute value of the numeric-expression.

If the value of the numeric-expression is positive or zero, ABS returns its value.

If the value of the numeric-expression is negative, ABS returns its negative (a positive number).

ABS always returns a non-negative number.

Examples

```
100 PRINT ABS(45.2)
PRINT ABS(45.2)
Prints 45.2
```

```
100 VV=ABS(-7.345)
VV=ABS(-7.345)
Sets VV equal to 7.345
```

ACCEPT

ACCEPT

Format

```
ACCEPT [[AT(row,column)] [BEEP] [ERASE ALL] [SIZE(numeric-expression)]
[INVERSE/BLINK] [CLIP] [VALIDATE(type[,...])]:]variable
```

Cross Reference

GRAPHICS, INPUT, LINPUT, MARGINS, TERMCHAR, BCOLOR, BTIME

Description

The ACCEPT instruction suspends program execution to enable you to enter data from the keyboard.

The options available with ACCEPT make it more versatile for keyboard input than the input statement. You can accept up to one line of input from any position within the screen window, sound a tone when the computer is ready to accept input, clear the screen window before accepting input, limit input to a specified number of characters, and define the types of valid input.

ACCEPT can be used as either a program statement or a command.

The data value entered from the keyboard is assigned to the variable you specify. If you specify a numeric variable, the data value entered from the keyboard must be a valid representation of a number. If you specify a string variable, the data value entered from the keyboard can be either a string or a number. Trailing spaces are removed.

A string value entered from the keyboard can optionally be enclosed in quotation marks. However, a string containing a comma, a quotation mark, or leading or trailing spaces must be enclosed in quotation marks. A quotation mark within a string is represented by two adjacent quotation marks.

You normally press ENTER to complete keyboard input; however, you can also use Alt 7 (AID), Alt 9 (BACK), Alt 5 (BEGIN), CLEAR, Alt 6 (PROC'D), DOWN ARROW, or UP ARROW. You can use the TERMCHAR function to determine which of those keys was pressed to exit from the previous ACCEPT, INPUT, or LINPUT instruction.

Note that pressing CLEAR during keyboard input normally causes a break in the program. However, if your program includes an ON BREAK NEXT statement, you can use CLEAR to exit from an input field.

Options

You can enter the following options, separated by a space in any order.

AT--Enables you to specify the location of the beginning of the input field. Row and column are relative to the upper-left corner of the screen window defined by the margins. The upper-left corner of the window defined by the margins is considered to be the intersection of row 1 and column 1 by an ACCEPT instruction that uses the AT option. If you do not use the AT option, the input field begins in the far left column of the bottom row of the window.

BEEP--Sounds a short tone to signal that the computer is ready to accept input.

ERASE ALL--Places a space character (ASCII code 32) in every character position in the screen window before accepting input.

SIZE--Enables you to specify a limit to the number of characters that can be entered as input. The limit is the absolute value of the numeric-expression. If the algebraic sign of the numeric-expression is positive, or if you do not use the SIZE option, the input field is cleared before input is accepted. If the numeric-expression is negative, the input field is not cleared, enabling you to place a value in the input field that may be accepted by pressing ENTER. If you do not use the SIZE option, or if the absolute value of the numeric-expression is greater than the number of characters remaining in the row (from the beginning of the input field to the right margin), the input field extends to the right margin.

VALIDATE--Enables you to specify the characters or the types of characters that are valid input. If you specify more than one type, a character from any of the specified types is valid. The types are as follows:

TYPE	VALID INPUT
ALPHA	All alphabetic characters.
UALPHA	All upper-case alphabetic characters.
LALPHA	All lower-case alphabetic characters.
DIGIT	All digits (0-9).
NUMERIC	All digits (0-9), the decimal point (.), the plus sign (+), the minus sign (-), and the upper-case letter E.

You can also use one or more string-expressions as types. The characters contained in the strings specified by the string-expressions are valid input.

The VALIDATE option only verifies data entered from the keyboard. If there is a default value in the input field (entered with DISPLAY), for example, the validate option has no effect on that value.

New Options

CLIP--Using the CLIP option, the string represented in the "DISPLAY AT" statement will be clipped at the end of a line rather than wrapping around to the next line, as it does in the default mode. The CLIP option is particularly useful when using "DISPLAY AT" within a window.

BLINK/INVERT--BLINK will cause the line displayed to BLINK on and off. This is only available in GRAPHICS(3,1) mode.

INVERT--Will cause the pixels in each character to invert their colors so the foreground- and background-colors will be inverted. This is only available in GRAPHICS(2,2), (2,3), (3,2), and (3,3) modes.

Examples

100 ACCEPT AT(3,5):Y

Accepts data at the third row, fifth column of the screen window into the variable Y.

100 ACCEPT VALIDATE("YN"):R\$

Accepts data containing Y and/or N into the variable R\$. (YYNN would be a valid entry.)

100 ACCEPT ERASE ALL:B

Accepts data into the variable B after putting the blank character into all positions in the screen window.

100 ACCEPT AT(R,C)SIZE(FIELDLEN)BEEP VALIDATE(DIGIT,"AYN"):X\$

Accepts a digit or the letters A, Y, or N into the variable X\$. The length of the input may be up to FIELDLEN characters. A field the length of FIELDLEN is filled with blank characters, and then the data value is accepted at row R, column C. A beep is sounded before acceptance of data.

Program

100 DIM NAME\$(20),ADDR\$(20)

110 DISPLAY AT (5,1)ERASE ALL:
L:"NAME:"

120 DISPLAY AT(7,1):"ADDRESS:"

130 DISPLAY AT(23,1):"TYPE A ? TO END ENTRY."

140 FOR S=1 TO 20

150 ACCEPT AT(5,7)VALIDATE(ALPHA,"?")BEEP SIZE(13):NAME \$(S)

160 IF NAME\$(S)="#" THEN 200

170 ACCEPT AT(7,10)SIZE(12):ADDR\$(S)

180 DISPLAY AT(7,10):" "

190 NEXT S

200 CALL CLEAR

210 DISPLAY AT(1,1):"NAME", "ADDRESS"

220 FOR T=1 TO S-1

230 DISPLAY AT(T+2,1):NAME\$(T),ADDR\$(T)

240 NEXT T

250 GOTO 250

(Press CLEAR to stop the program.)

ASC

ASC

Format

ASC(string-expression)

Cross Reference

CHR\$

Description

The ASC function returns the ASCII character code corresponding to the first character of the string-expression.

ASC is the inverse of the CHR\$ function.

The string-expression cannot be a null string.

Examples

```
100 PRINT ASC("A")
```

Prints 65 (the ASCII character code for the letter A).

```
100 B=ASC("1")
```

Sets B equal to 49 (the ASCII character code for the character 1).

```
100 DISPLAY ASC("HELLO")
```

Displays 72 (the ASCII character code for the letter H).

```
100 A$="DAVID"
```

```
110 PRINT ASC(A$)
```

Prints 68 in line 110.

ATN**ATN**

Format
 ATN(numeric-expression)

Cross Reference
 COS, SIN, TAN

Description
 The ATN function returns the angle (in radians) whose tangent is the value of the numeric-expression.

The value returned by ATN is always greater than $-\pi/2$ and less than $\pi/2$.

Examples

```
100 PRINT 4*ATN(-1)
Prints -3.141592654.
```

```
100 Q=PI/ATN(1.732)
Sets Q equal to 3.0000363894830.
```

BCOLOR**BCOLOR**

Format
 CALL BCOLOR(foreground,background)

Cross Reference
 BTIME, DISPLAY, ACCEPT

Description
 This command is used to set the foreground- and background-colors of the BLINK parameter used in conjunction with DISPLAY AT, ACCEPT AT and BTIME. The value of foreground- or background-color is 1 to 16 as given in Appendix F. This subroutine is applicable only to graphics 3,1 (Text 2) mode.

Example

```
100 CALL GRAPHICS(3,1)
110 CALL SCREEN(16,5)
120 CALL BCOLOR(16,7)
130 DISPLAY AT(5,1)ERASE ALL BLINK:"THIS IS BLINKING"
140 ACCEPT AT(5,1)BLINK SIZE(-28):A$
```

This program displays normal text in white with a dark blue background. The display area on line 5 will blink and alternately be white text on a dark red background and white text on a dark blue background.

BEEP**BEEP**

Cross Reference
DISPLAY AT, ACCEPT AT, SOUND

Description

The BEEP command sounds a short tone when encountered as a command or program statement. BEEP is also an option in DISPLAY AT and ACCEPT AT commands.

You can use BEEP either as a program statement or as a command.

Example

```
100 CALL GRAPHICS(4)
110 DEFINT I,R,C
120 CALL POINT(1,R,C)
130 FOR I=1 TO 25
140 C=(RND5)+1
150 R=(RND1)+1
160 CALL DRAWTO(R,C)
170 BEEP
180 NEXT I
190 GOTO 190
(Press CLEAR to stop the program.)
```

This program randomly selects the ROW COLUMN coordinates of 25 points and draws lines connecting them sequentially. Each time a line is drawn the BEEP sound is produced.

BREAK**BREAK**

Format

BREAK(line-number-list)

Cross Reference

CONTINUE, ON BREAK, UNBREAK

Description

The **BREAK** instruction sets a breakpoint at each program statement you specify. When the computer encounters a line at which you have set a breakpoint, your program stops running before that statement is executed.

BREAK is a valuable debugging aid. You can use **BREAK** to stop your program at a specific program line, so that you can check the values of variables at that point.

You can use **BREAK** line-number-list as either a program statement or a command.

The line-number-list consists of one or more line numbers, separated by commas. When a **BREAK** instruction is executed, breakpoints are set at the specified program lines. If you use **BREAK** as a program statement, line-number-list is optional. When a **BREAK** statement with no line-number-list is encountered, the computer stops running the program at that point.

If you use **BREAK** as a command, you must include a line-number-list.

Breakpoints

When your program stops at a breakpoint, the message Breakpoint in line number is displayed. While your program is stopped at a breakpoint, you can enter any valid command.

To resume program execution starting with the line at which the break occurred, enter the **CONTINUE** command. However, if you edit your program (add, delete, or change a program statement) you cannot use **CONTINUE**. (This prevents errors that could result from resuming execution in the middle of a revised program.) You also cannot use **CONTINUE** if you enter a **MERGE** or **SAVE** command or a **LIST** command with the file-specification option. Note that pressing **CLEAR** also causes a breakpoint to occur before the execution of the next program statement. When your program stops at a breakpoint, the computer performs the following operations:

It restores the default character definitions of all characters.

It restores the default foreground-color and background-color to all characters.

It restores the default screen color.

It deletes all sprites.

It resets the sprite magnification level to 1.

The graphics colors (see DCOLOR) and current position (see DRAWTO) are not affected. If the computer is in Pattern or Text Mode, the graphics mode and margin settings remain unchanged.

Removing Breakpoints

You can remove a breakpoint by using the UNBREAK instruction or by editing or deleting the line at which the breakpoint is set. When your program stops at a breakpoint, that breakpoint is automatically removed.

All breakpoints are removed when you use the NEW or SAVE command.

BREAK Errors

If the line-number-list includes an invalid line number (0 or a value greater than 32767), the message Bad line number is displayed. If the line-number-list includes a fractional or negative line number, the message Syntax error is displayed. In both cases, the BREAK instruction is ignored; that is, breakpoints are not set even at valid line numbers in the line-number-list. If you were entering BREAK as a program statement, it is not entered into your program.

If the line-number-list includes a line number that is valid (1-32767) but is not the number of a line in your program, or a fractional number greater than 1, the message

```
WARNING
LINE NOT FOUND
```

is displayed. If you were entering BREAK as a program statement, the line number is included in the warning message. A breakpoint is, however, set at any valid line in the line-number-list preceding the line number which caused the warning.

Examples

```
150 BREAK
```

BREAK as a statement causes a breakpoint before execution of the next line in the program.

```
100 BREAK 120,130
```

Causes breakpoints before execution of lines 120 and 130.

```
BREAK 10,400,130
```

As a command, causes breakpoints before execution of lines 10, 400, and 130.

BTIME

BTIME

Format

CALL BTIME(blinkrate-ON, blinkrate-OFF)

Cross Reference

BCOLOR, ACCEPT, DISPLAY

Description

This command is used to set the rate at which characters are set to BLINK in the DISPLAY AT and ACCEPT AT statements.

Blinkrate can be an integer from 0 to 15, representing actual blink rates between 0 and 2503.5 milliseconds in multiples of 166.9 milliseconds.

Example

```
100 CALL GRAPHICS(3,1)
110 CALL DCOLOR(15,5)
120 CALL BCOLOR(15,7)
130 FOR I=0 TO 15
140 CALL BTIME(I,I)
150 DISPLAY AT(5,1)ERASE ALL BLINK:"RATE OF BLINK= ";I
160 FOR DELAY=1 TO 1000::NEXT DELAY
170 NEXT I
180 END
```

The above program illustrates some of the possible blink rates.

BYE

BYE

Format

BYE

Description

The BYE command resets the computer. Always use BYE to exit from MYARC Advanced BASIC. The BYE command causes the computer to do the following:

Close all open files.

Erase the program and all variable values in memory.

Exit from MYARC Advanced BASIC.

Display the DOS command line.

CALL

CALL

Format

CALL subprogram-name[(parameter-list)]

Cross Reference

SUB

Description

The CALL instruction transfers program control to the specified subprogram.

You can use CALL as either a program statement or a command.

The CALL instruction transfers program control to the subprogram specified by the subprogram-name.

The optional parameter-list consists of one or more parameters separated by commas. Use of a parameter-list is determined by the subprogram you are calling. Some subprograms require a parameter-list, some do not use a parameter-list, and with some a parameter-list is optional.

You can use CALL as a program statement to call either a built-in MYARC Advanced BASIC subprogram or to call a subprogram that you write. After the subprogram is executed, program control returns to the statement immediately following the CALL statement.

You can use CALL as a command only to call a built-in MYARC Advanced BASIC subprogram, not to call a subprogram that you write.

Each of the following built-in subprograms is discussed separately in this manual:

CHAR	GCHAR	PEEKV
CHARPAT	GRAPHICS	POINT
CHARSET	HCHAR	POKEY
CIRCLE	INIT	POSITON
CLEAR	JOYST	RECTANGLE
COINC	KEY	SAY
COLOR	LINK	SCREEN
DCOLOR	LOAD	SOUND
DELSPRITE	LOCATE	SPGET
DISTANCE	MAGNIFY	SPRITE
DRAW	MARGIN	VCHAR
DRAWTO	MOTION	
ERR	PATTERN	
FILL	PEEK	

Program

The following program illustrates the use of CALL with a built-in subprogram (CLEAR) in line 100 and the use of a user-written subprogram (TIMES) in line 120.

```
100 CALL CLEAR
110 X=4
120 CALL TIMES(X)
130 PRINT X
140 STOP
200 SUB TIMES(Z)
210 Z=Z*PI
220 SUBEND
RUN
(SCREEN CLEARS)
12.56637061
```

CDBL

CDBL

Format
(numeric-expression)

Cross Reference
DEFtype, CINT, CSNG, CREAL

Description
Converts a number to double-precision. The numeric-expression must evaluate to either an integer, or a single- or a double-precision value.

CAUTION: mixed mode arithmetic is not allowed.

Arithmetic modes:
REAL: Real numbers and integers.

BINARY: Integers, single-precision, double-precision.

Mixing real numbers with either single- or double-precision will cause a mixed arithmetic mode error.

CHAR -Subprogram

CHAR

Format

```
CALL CHAR(character-code,pattern-string[,...])
```

Cross Reference

CHARPAT, CHARSET, COLOR, DCOLOR, GRAPHICS, HCHAR, SCREEN, SPRITE, VCHAR

Description

The CHAR subprogram enables you to define your own characters so that you can create graphics on the screen.

CHAR is the inverse of the CHARPAT subprogram.

Character-code is a numeric-expression with a value from 0 to 255, specifying the number of the character (codes 0-255). You can define any of the 256 characters and display them as characters and/or sprites.

The pattern-string specifies the definition of the character. The pattern-string, which may be up to 64 digits long, is a coded representation of the pixels that define up to four characters on the screen, as explained below. Any letters entered as part of a pattern-string must be upper case.

You can use the CHARSET subprogram to restore default character definitions of characters 32-95 inclusive. Also, when your program ends (either normally or because of an error), stops at a breakpoint, or changes graphics mode, all default character definitions (0-255) are restored.

The instructions that you can use to display characters on the screen vary according to the graphics mode. In all modes except Text Modes, you can use the SPRITE subprogram to display sprites on the screen.

If you use HCHAR or VCHAR to display a character on the screen and then later use CHAR to change the definition of that character, the result depends on the graphics mode.

In Pattern and Text Modes, the displayed character changes to the newly defined pattern.

In Bit Mapped Modes, the displayed character remains unchanged.

Graphics(1,X) Modes

In Graphics(1,1), (1,2), and (1,3) modes, each character is composed of 64 pixels in a grid eight pixels high and eight pixels wide, as explained below.

You can use the DISPLAY, DISPLAY USING, PRINT, and PRINT USING instructions and the HCHAR and VCHAR subprograms to display characters on the screen.

Other Graphics Modes

In Graphics(2,X) and (3,X), each character is composed of 48 pixels in a grid eight pixels high and six pixels wide. The eight by eight grid described below is used to define characters; however, the last two pixels in each pixel-row are ignored.

In these modes, you can use the DISPLAY, DISPLAY USING, PRINT, and PRINT USING instructions and the HCHAR and VCHAR subprograms to display characters on the screen. You cannot display sprites in Text Modes.

Character Definition--The Pattern String

Characters are defined by turning some pixels on and leaving others off. The space character (ASCII code 32) is a character with all the pixels turned off. Turning all the pixels on produces a solid block, eight pixels high and eight pixels wide.

The foreground-color is the color of the pixels that are on. The background-color is the color of the pixels that are off. (For more information see COLOR, DCOLOR, and SCREEN.)

When you enter MYARC Advanced BASIC, the characters are predefined with the appropriate pixels turned on. To redefine a character, you specify which pixels to turn on and which pixels to turn off.

For the purpose of defining characters, each pixel-row (eight pixels) is divided into two blocks (four pixels each). Each digit in the pattern-string is a code specifying the pattern of the four pixels in one block.

You define a character by describing the blocks from left to right and from top to bottom. The first two digits in the pattern-string describe the pattern for the first two blocks (pixel-row 1) of the grid, the next two digits define the next two blocks (pixel-row 2), and so on.

The computer uses a binary (base 2) code to represent the status of each pixel; you use hexadecimal (base 16) notation of the binary code to specify which pixels in a box are turned on and which pixels are turned off.

The following table shows all the possible on/off combinations of the four pixels in a block and the binary code and hexadecimal notation representing each combination.

BLOCK	BINARY CODE (0=OFF; 1=ON)	HEXADECIMAL NOTATION
—	0000	0
— X	0001	1
— X	0010	2
— XX	0011	3
— X	0100	4
— X X	0101	5
— XX	0110	6
— XXX	0111	7
X	1000	8
X X	1001	9
X X	1010	A
X XX	1011	B
XX	1100	C
XX X	1101	D
XXX	1110	E
XXXX	1111	F

A character definition consists of 16 hexadecimal digits; each digit represents one of the 16 blocks that comprise a character. As the pattern-string may be up to 64 digits long, you can define as many as four consecutive characters with one pattern-string.

If the length of the pattern-string is not a multiple of 16, the computer fills the pattern-string with zeros until its length is a multiple of 16.

Programs

For the dot pattern pictured below, you use "1898FF3D3C3CE404" as the pattern string for CALL CHAR. The following program uses this and one other string to make a figure "dance". This example will work only in Pattern Mode.

```

100 CALL CLEAR
110 A$="1898FF3D3C3CE404"
120 B$="1819FFBC3C3C2720"
130 CALL COLOR(27,7,12)
140 CALL VCHAR(12,16,244)
150 CALL CHAR(244,A$)
160 GOSUB 200

```

MYARC Advanced BASIC

```
170 CALL CHAR(244,B$)
180 GOSUB 200
190 GOTO 150
200 FOR DELAY=1 TO 150
210 NEXT DELAY
220 RETURN
RUN
```

(screen clears)
(character moves)
(Press CLEAR to stop the program.)

To make this example work in a Bit-Mapped Graphics Mode, make the following changes.

```
105 CALL GRAPHICS(2,2)
130 CALL DCOLOR(7,12)
140 CALL CHAR(144,A$,145,B$)
150 CALL VCHAR(12,16,144)
170 CALL VCHAR(12,16,145)
```

If a program stops for a breakpoint, all characters are reset to their standard patterns. When the program ends normally or because of an error, all characters are reset.

The following example works in all graphics modes.

```
100 CALL CLEAR
110 CALL GRAPHICS(X,Y)
120 CALL CHAR(144,"FFFFFFFFFFFFFFFF")
130 CALL CHAR(42,"OFOFOFOFOFOFOFO")
140 CALL HCHAR(12,17,42)
150 CALL VCHAR(14,17,144)
160 FOR DELAY=1 TO 500
170 NEXT DELAY
RUN
```

The X and Y in line 110 must be replaced with the number of the graphics mode to be designated.

CHARPAT -Subprogram

CHARPAT

Format

```
CALL CHARPAT(character-code,string-variable[,...])
```

Cross Reference

CHAR

Description

The CHARPAT subprogram enables you to ascertain the current character definitions of specified characters.

Character-code is a numeric-expression with a value from 0 to 255, specifying the number of the character of which you want the current definition.

The pattern describing the character definition is returned in the specified string-variable. The pattern is in the form of a 16-digit hexadecimal code. See CHAR for an explanation of the pattern used for character definition.

See Appendix B for a list of available characters.

Example

```
100 CALL CHARPAT(33,C$)
```

Sets C\$ equal to "0010101010001000", the pattern identifier for character 33, the exclamation point.

CHARSET -Subprogram--Set Characters**CHARSET**

Format

CALL CHARSET ([startchar][-][endchar])

Cross Reference

CHAR, COLOR, CCOLOR

Description

The CHARSET subprogram restores default character definitions and colors.

CHARSET, if not followed by (), restores the default character definitions to characters 32-95, inclusive.

In Graphics(1), CHARSET restores the default colors to all 256 characters.

If followed by parenthesis, CALL CHARSET(startchar-endchar) will restore only the specified characters.

Example

COMMAND	ACTION
CALL CHARSET	Restores characters 32 to 96 inclusive.
CALL CHARSET(X-Y)	Restores characters X to Y inclusive.
CALL CHARSET(-Y)	Restores all characters starting at 1 through Y inclusive.
CALL CHARSET(X-)	Restores all characters from X through 255 inclusive.

See Appendix B for a list of available characters.

CHR\$ -Function--Character

CHR\$

Format

CHR\$(character-code)

Type

String

Cross Reference

ASC

Description

The CHR\$ function returns the character corresponding to the ASCII character code specified by the value of the character-code.

CHR\$ is the inverse of the ASC function.

Character-code is a numeric-expression with a value from 0 to 32767 inclusive, specifying the number of the character you wish to use. If the value of character-code is greater than 255, it is repeatedly reduced by 256 until it is less than 256. If the value of the character-code is not an integer, it is rounded to the nearest integer.

Examples

```
100 PRINT CHR$(72)
Prints H.
```

```
100 X$=CHR$(33)
Sets X$ equal to !.
```

Program

For a complete listing of all ASCII characters and their corresponding ASCII values, run the following program.

```
100 CALL CLEAR
110 IMAGE ### ## ### ##
120 FOR A=32 TO 127
130 PRINT USING 110:A,CHR$(A);
140 NEXT A
```

CINT

CINT

Format

CINT(numeric-expression)

Cross Reference

DEFvartype, CDBL, CSNG, CREAL

Description

Converts a number to integer precision.

CAUTION: mixed mode arithmetic is not allowed.

Arithmetic modes:

REAL: real numbers and integers.

BINARY: integers, single-precision, double-precision.

Mixing real numbers with either single- or double-precision will cause a mixed arithmetic mode error.

CIRCLE -Subprogram

CIRCLE

Format

```
CALL CIRCLE(x,y),radius,[,color[,start,end[,aspect ratio]]]
```

Description

Draws an ellipse on the screen with center at (x,y). A circle is drawn when the aspect ratio equals 1.

X,Y are relative coordinates of the screen. The coordinates 0,0 represent the extreme upper left hand corner of the screen.

X represents the vertical coordinate or the dot-column count to the right of the left hand edge of the screen.

Y represents the horizontal coordinate or the dot/row count from the top of the screen counting increasing towards the bottom.

Radius is the major axis of the ellipse.

Start, and end, are the beginning and ending angles in radians. Must be in the range of $-2*PI$ to $2*PI$.

Aspect ratio is the ratio of the X-radius to the Y-radius in terms of coordinates.

If aspect ratio is less than 1, radius is the x-radius and is measured in pixels in the horizontal direction.

If aspect is greater than 1, radius is the y-radius and is measured in pixels in the vertical direction.

When X-radius is equal to Y-radius, then aspect ratio is equal to 1 and a circle is drawn.

Example

```
CIRCLE(150,100),50
```

CLEAR -Subprogram

CLEAR

Format
CALL CLEAR

Cross Reference
DCOLOR, DELSPRITE

Description
The CLEAR subprogram erases the screen.

CLEAR places a space character (ASCII code 32) in every screen position.

The CLEAR subprogram has no effect on sprites. Use the DELSPRITE subprogram to remove sprites.

Programs

When the following program is run, the screen is cleared before the PRINT statements are performed.

```
100 CALL CLEAR
110 PRINT "HELLO THERE!"
120 PRINT "HOW ARE YOU?"
RUN
--screen clears
HELLO THERE!
HOW ARE YOU?
```

If the space character (ASCII code 32) has been redefined by the CALL CHAR subprogram, the screen is filled with the new character when CALL CLEAR is performed.

```
100 CALL CHAR(32,"0103070F1F3F7FFF")
110 CALL CLEAR
120 GOTO 120
RUN
--Screen is filled with *
(Press CLEAR to stop the program.)
```

The following program first fills and then clears the entire screen.

```
100 CALL GRAPHICS(1,2)
110 CALL HCHAR(1,2,72,768)
120 FOR DELAY=1 TO 500::NEXT DELAY
130 CALL CLEAR
140 GOTO 140
RUN
(Press CLEAR to stop the program.)
```

CLOSE

CLOSE

Format

```
CLOSE #file-number[:KILL]
```

Cross Reference

KILL, OPEN, DELETE

Description

The CLOSE instruction closes the specified file. When you close a file, you discontinue the association (between your program and the file) that you established in an OPEN instruction.

If #file-number is omitted, then basic will close all open files, however, the KILL option is not allowed without a specific #file-number.

You can use CLOSE as either a program statement or a command.

The file-number is a numeric-expression whose value specifies the number of the file as assigned in its OPEN instruction.

The KILL option, which can be used only with certain devices, deletes the file after closing it. For more information about using the KILL option with a particular device, refer to the owner's manual that comes with that device.

After the CLOSE instruction is performed, the closed file cannot be accessed by an instruction because the computer no longer associates that file with a file-number. You can then reassign the file-number to another file.

Closing Files Without the CLOSE Instruction

To protect the data in your files, the computer closes all open files when it reaches the end of your program or when it encounters an error (either in Command or Run mode).

Open files are also closed when you do one of the following:

Edit your program (add, delete, or change a program statement).

Enter the BYE, MERGE, NEW, OLD, RUN, or SAVE command.

Open files are not closed when you stop program execution by pressing CLEAR or when your program stops at a breakpoint set by a BREAK instruction.

Examples

Diskette File

```
100 OPEN #24:"DSK1.MYDATA",INTERNAL,UPDATE,FIXED
200 CLOSE #24
RUN
```

The CLOSE statement for a diskette requires no further action on your part.

CLS

CLS

Format

CLS

Description

You may use CLS either as a program statement or a command.

CLS clears the screen or active viewport created such as with the CALL MARGIN statement, and returns the cursor to the home position.

Examples

```
100 CALL GRAPHICS(2,1)
110 CALL MARGIN(1,24,1,40)
120 CALL HCHAR(1,1,ASC("A"),960)
130 CALL MARGIN(5,10,5,10)
140 CLS
150 GOTO 150
RUN
(Press CLEAR to stop the program.)
```

Program will fill the screen with character 65, the letter A, then it creates a window 5 rows by 5 columns. The CLS statement clears this window leaving the remainder of the screen filled with the letter "A".

NOTE: An alternate method of clearing the active "window" in this case would have been to substitute line 140 with:

```
140 DISPLAY AT(1,1)ERASE ALL:""
```

CALL CLEAR or CALL GRAPHICS(n) will clear the entire screen.

COINC -Subprogram--Coincidence**COINC**

Format

Two Sprites

CALL COINC(#sprite-number1,#sprite-number2,tolerance,numeric-variable)

A Sprite and a Screen Pixel

CALL COINC(#sprite-number,pixel-row,pixel-column,tolerance,
numeric-variable)

All Sprites

CALL COINC(ALL,numeric-variable)

Cross Reference

SPRITE

Description

The COINC subprogram enables you to ascertain if sprites are coincident (in conjunction) with each other or with a specified screen pixel.

The exact conditions that constitute a coincidence vary depending on whether you are testing for the coincidence of two sprites, a sprite and a screen pixel, or all sprites.

If the sprites are moving very quickly, COINC may occasionally fail to detect a coincidence.

Two Sprites

Two sprites are considered to be coincident if the upper-left corners of the sprites are within a specified number of pixels (tolerance) of each other.

The values of the numeric-expressions sprite-number1 and sprite-number2 specify the numbers of the two sprites as assigned in the SPRITE subprogram.

A coincidence exists if the distance between the pixels in the upper-left corners of the two sprites is less than or equal to the value of the numeric-expression tolerance.

The distance between two pixels is said to be within tolerance if the difference between pixel-rows and the difference between pixel-columns are both less than or equal to the specified tolerance. Note that this is not the same as the distance indicated by the DISTANCE subprogram.

COINC returns a value in the numeric-variable indicating whether or not the specified coincidence exists. The value is -1 if there is a coincidence or 0 if there is no coincidence.

A Sprite and a Screen Pixel

A sprite is considered to be coincident with a screen pixel if the upper-left corner of the sprite is within a specified number of pixels (tolerance) of the screen pixel or if any pixel in the sprite occupies the screen pixel location.

The `sprite-number` is a numeric-expression whose value specifies the number of the sprite assigned in the `SPRITE` subprogram.

The `pixel-row` and `pixel-column` are numeric-expressions whose values specify the position of the screen pixel.

A coincidence exists if the distance between the pixel in the upper-left corner of the sprite and the screen pixel is less than or equal to the value of the numeric-expression `tolerance`. (Note that a coincidence also exists if any pixel in the sprite occupies the screen pixel location.)

The distance between two pixels is said to be within tolerance if the difference between `pixel-rows` and the difference between `pixel-columns` are both less than or equal to the specified tolerance. Note that this is not the same as the distance indicated by the `DISTANCE` subprogram.

`COINC` returns a value in the numeric-variable indicating whether or not the specified coincidence exists. The value is `-1` if there is a coincidence or `0` if there is no coincidence.

All Sprites

The `ALL` option tests for the coincidence of any of the sprites.

For the `ALL` option, sprites are considered to be coincident if any pixel of any sprite occupies the same screen pixel location as any pixel of any other sprite.

`COINC` returns a value in the numeric-variable indicating whether or not a coincidence exists. The value is `-1` if there is a coincidence or `0` if there is no coincidence.

Program

```
100 CALL CLEAR:: S$="0103070F1F3F7FFF"
120 CALL CHAR(244,S$)::CALL CHAR(250,S$)
140 CALL SPRITE(#1,244,7,50,50)
150 CALL SPRITE(#2,250,5,44,42)
160 CALL COINC(#1,#2,10,C)
170 PRINT C
180 CALL COINC(ALL,C)
190 PRINT C
RUN
-1
0
```

Line 160 shows a coincidence because the upper-left corners of the sprites are within 10 pixels of each other.

Line 180 shows no coincidence because the shaded areas of the sprites do not occupy the same screen pixel location. (Shaded areas are compared only if you specify the `ALL` option.)

COLOR --Subprogram**COLOR**

Format

Pattern Mode

```
CALL COLOR(character-set,foreground-color,background-color[,...])
```

Sprites

```
CALL COLOR(#sprite-number,foreground-color[,...])
```

Cross Reference

CHAR, DCOLOR, GRAPHICS, SCREEN, SPRITE

Description

The **COLOR** subprogram enables you to specify the colors of characters or sprites.

The types of parameters you specify in a call to the **COLOR** subprogram depend on whether you are assigning colors to characters or to sprites.

In general, each character has two colors. The color of the pixels that make up the character itself is the foreground-color; the color of the pixels that occupy the rest of the character position on the screen is the background-color.

When you enter MYARC Advanced BASIC, the foreground-color of all the characters is black; the background-color of all characters is transparent. These default colors are restored when your program ends (either normally or because of an error), stops at a breakpoint, or changes graphics mode.

If a color is transparent, the color actually displayed is the color specified by the **SCREEN** subprogram.

See Appendix F for a listing of available colors and their respective codes.

Pattern Mode and Bit Mapped Modes

In these modes (i.e., Graphics(1,1), (2,2), (2,3), (3,2), (3,3)), the 256 available characters are divided into 32 sets of 8 characters each. When you assign a color combination to a particular set, you specify the colors of all 8 characters in that set.

The character-set is a numeric-expression whose value specifies the number (0-31) of the 8-character set.

Foreground-color and background-color are numeric/expressions whose values specify colors that can be assigned from among the 16 available colors.

See Appendix D for available characters and character sets in Pattern Mode.

Text Modes

An error occurs if you use the COLOR subprogram to assign character colors in either Text Mode (i.e., Graphics(2,1) or Graphics(3,1). Use the SCREEN subprogram to assign character colors in Text Mode. Sprites are not displayed in text mode.

Graphics(1,2) and (1,3)

In these modes, you can use COLOR only to assign colors to sprites; any other use of the COLOR subprogram causes an error. Use the DCOLOR subprogram to specify character and graphics colors in High-Resolution Mode.

Sprites

A sprite is assigned a foreground-color when it is created with the SPRITE subprogram. The background-color of a sprite is always transparent.

To re-assign colors to sprites you must use the sprite parameters, no matter what graphics mode the computer is in.

The sprite-number is a numeric-expression whose value specifies the number of a sprite as assigned by the SPRITE subprogram.

Foreground-color is a numeric-expression whose value specifies a color that can be assigned from among the 16 available colors.

Examples

```
100 CALL COLOR(#5,16)
```

Sets sprite number 5 to have a foreground-color of 16 (white). The background-color is always 1 (transparent).

This example is valid in all graphics modes. (Remember that sprites have no effect in Text Modes.)

```
100 CALL COLOR(#7,INT(RND* 16+1))
```

Sets sprite number 7 to have a foreground-color chosen randomly from the 16 colors available. The background-color is 1 (transparent).

This example is valid in all graphics modes.

Program

This program sets the foreground-color of characters 48-55 to 5 (dark blue) and the background-color to 12 (light yellow).

```
100 CALL CLEAR
110 CALL GRAPHICS(1)
120 CALL COLOR(3,5,12)
130 DISPLAY AT(12,16):CHR$(48)
140 GOTO 140
(Press CLEAR to stop the program.)
```

CONTINUE**CONTINUE**

Format
CONTINUE
CON

Cross Reference
BREAK

Description

The CONTINUE command restarts a program which has been stopped by a breakpoint. It may be entered whenever a program has stopped running because of a breakpoint caused by the BREAK command or statement or pressing Control + Break keys (CLEAR.) However, you cannot use the CONTINUE command if you have edited a program line. CONTINUE may be abbreviated as CON.

When a breakpoint occurs, the standard character set and standard colors are restored. Sprites cease to exist. CONTINUE does not restore user-defined characters that have been reset or any colors. Otherwise, the program continues as if no breakpoint had occurred.

COS --Function--Cosine

COS

Format
COS(numeric-expression)

Type
REAL

Cross Reference
ATN, SIN, TAN

Description
The COS function returns the cosine of the angle whose measurement in radians is the value of the numeric-expression.

The value of the numeric-expression cannot be less than
-1.5707963269514E10 or greater than 1.5707963266374E10.

To convert the measure of an angle from degrees to radians, multiply by $\pi/180$.

Program

The following program gives the cosine for each of several angles.

```
100 A=1.047197551196
110 B=60
120 C=45*PI/180
130 PRINT COS(A);COS(B)
140 PRINT COS(B*PI/180)
150 PRINT COS(C)
RUN
.5 -.9524129804
.5
.7071067812
```

CREAL

CREAL

Format

CREAL(numeric-expression)

Cross Reference

DEFvartype, CDBL, CINT, CSNG

Description

Converts a number to single-precision.

CAUTION: mixed mode arithmetic is not allowed.

Arithmetic modes:

REAL: real numbers and integers.

BINARY: integers, single-precision, double-precision.

Mixing real numbers with either single- or double-precision will cause a mixed arithmetic mode error.

CSNG

CSNG

Format

CSNG (numeric-expression)

Cross Reference

DEFvartype, CINT, CDBL, CREAL

Description

Converts a number to single-precision.

CAUTION: mixed mode arithmetic is not allowed.

Arithmetic modes:

REAL: real numbers and integers.

BINARY: integers, single-precision, double-precision.

Mixing real numbers with either single- or double-precision will cause a mixed arithmetic mode error.

DATA

DATA

Format

DATA data-list

Cross Reference

READ, RESTORE

Description

The DATA statement enables you to store constants within your program. You can assign the constants to variables by using a READ statement.

The data-list consists of one or more constants separated by commas. The constants can be assigned to the variables specified in the variable-list of a READ statement. The assignment is made when the READ statement is executed.

If a numeric variable is specified in the variable-list of a READ statement, a numeric constant must be in the corresponding position in the data-list of the DATA statement. If a string variable is specified in a READ statement, either a string or a numeric constant may be in the corresponding position in the DATA statement. A string constant in a data-list may optionally be enclosed in quotation marks. However, if the string constant contains a comma, a quotation mark, or leading or trailing spaces, it must be enclosed in quotation marks.

A quotation mark within a string constant is represented by two adjacent quotation marks. A null string is represented in a data-list by two adjacent commas, or two commas separated by two adjacent quotation marks.

The order in which the data values appear within the data-list and the order of the DATA statements within a program normally determine the order in which the values are read. Values from each data-list are read sequentially, beginning with the first item in the first DATA statement. If your program includes more than one DATA statement, the DATA statements are read in ascending line-number order (unless you use a RESTORE statement to specify otherwise).

A DATA statement encountered during program execution is ignored.

A DATA statement cannot be part of a multiple-statement line, nor can it include a trailing remark.

Program

The following program reads and prints several numeric and string constants.

```
100 FOR A=1 TO 5
110 READ B,C
120 PRINT B;C
130 NEXT A
140 DATA 2,4,6,7,8
150 DATA 1,2,3,4,5
160 DATA "" "THIS HAS QUOTES""
170 DATA NO QUOTES HERE
180 DATA " NO QUOTES HERE,EITHER"
190 FOR A=1 TO 6
200 READ B$
210 PRINT B$
220 NEXT A
230 DATA 1,NUMBER,MYARC
RUN
2 4
6 7
8 1
2 3
4 5
"THIS HAS QUOTES"
NO QUOTES HERE
  NO QUOTES HERE,EITHER
  1
NUMBER
MYARC
```

Lines 100 through 130 read five sets of data and print their values, two to a line.

DATE/DATE\$

DATE/DATE\$

Format

```
CALL DATE["mm/dd/yy"]
```

```
DATE$
```

Description

DATE\$ can be a function.

CALL DATE can be a statement or a command.

It can be used to set the date or retrieve the current date.

To set the date use the format:

```
CALL DATE("mm/dd/yy")
```

mm is the two-digit equivalent of the current month 01 - 12

dd is the two digit date 01 - 31

yy is the last two digits of the year. Two-digit range = range 01 - 99.

To retrieve the current date, use the function DATE\$.

Example

```
CALL DATE$("01/01/87")
```

This example sets the date to January 1, 1987.

Example

```
PRINT DATE$
```

```
01/01/87
```

```
100 A$=DATE$
```

```
110 PRINT A$
```

```
RUN
```

```
01/01/87
```

Example

```
100 PRINT "TODAY'S DATE IS ";DATE$
```

```
110 INPUT "DO YOU WISH TO CHANGE THE DATE ? ":CHANGE$
```

```
120 IF LEFT$(CHANGE$,1)="Y" OR LEFT$(CHANGE$,1)="y" THEN 130 ELSE END
```

```
130 INPUT "ENTER NEW DATE: ":NEWDATE$
```

```
140 CALL DATE(NEWDATE$)
```

```
150 GOTO 100
```


DCOLOR --Subprogram--Draw Color

DCOLOR

Format

CALL DCOLOR(foreground-color,background-color)

Cross Reference

CIRCLE, COLOR, DRAW, DRAWTO, FILL, GRAPHICS, HCHAR, POINT, RECTANGLE, VCHAR

Description

The DCOLOR subprogram enables you to set the graphics colors.

The graphics colors are used by the CIRCLE, DRAW, DRAWTO, FILL, HCHAR, POINT, RECTANGLE, and VCHAR subprograms in Bit Mapped Graphics and normal Graphics modes.

Foreground-color and background-color are numeric-expressions whose values specify colors that can be assigned from among the 16 available colors. See Appendix F for a list of the available colors.

When you enter MYARC Advanced BASIC, the foreground-color is set to black and the background-color is set to transparent. These default graphics colors are restored only when you change graphics mode. They are not restored when you enter RUN.

DCOLOR is effective only in Bit Mapped and normal Graphics modes. DCOLOR has no effect in Pattern or Text mode.

Programs

The following program sets the foreground-color of graphics to 5 (dark blue) and the background-color to 8 (cyan).

```
100 CALL CLEAR
110 CALL GRAPHICS(2,2)
120 CALL DCOLOR(5,8)
130 CALL HCHAR(8,20,72,3)
```

In the following program, the letters "HHH" are displayed on the screen.

```
100 CALL CLEAR
110 CALL GRAPHICS(2,2)
120 RANDOMIZE
130 CALL DCOLOR(INT(RND*8+1)*2,INT(RND*8+1)*2-1)
140 CALL HCHAR(8,20,72,3)
150 FOR X=1 TO 400
160 NEXT X
170 GOTO 120
(Press CLEAR to stop the program.)
```

Line 130 changes the foreground-color (chosen randomly from the even-numbered colors available) and the background-color (chosen randomly from the odd-numbered colors).

DEF --Define Function

DEF

Format

DEF function-name[(parameter1 [, . . . parameter7])]=expression

Description

The DEF statement enables you to define your own functions. These user-defined functions can then be used in the same way as built-in functions.

The function-name can be any valid variable name that does not appear as a variable name elsewhere in your program.

If the function-name is a numeric variable, the value of the expression must be a number. If the function-name is a string variable, the value of the expression must be a string.

If the function-name is a numeric variable, you can optionally specify its data-type (DEFINT, DEFREAL, DEFSNG, or DEFDBL) by using variable tags.

You can use up to seven parameters to pass values to a function. Parameters must be valid variable names. A variable name used as a parameter cannot be the name of an array. You can use an array element in the expression if the array does not have the same name as a parameter in that statement. The variable names used as parameters in a DEF statement are local to that statement; that is, even if a parameter has the same name as a variable in your program, the value of that variable is not affected.

If a parameter is a numeric variable, you can optionally specify its data-type (DEFINT, DEFREAL, DEFSNG, or DEFDBL) by using variable tags.

A DEF statement must have a lower line number than that of any use of the function-name it defines. A DEF statement is not executed.

A DEF statement can appear anywhere in your program, except that it cannot be part of an IF THEN statement.

DEF Without Parameters

When your program encounters a statement containing a previously defined function-name with no parameters, the expression is evaluated, and the function is assigned the value of the expression at that time.

If you define a function-name without parameters, it must appear without parameters when you use it in your program.

DEF With Parameters

When your program encounters a statement containing a previously defined function-name with parameters, the parameter values are passed to the function in the same order in which they are listed. The expression is evaluated using those values, and the function is assigned the value of the expression at that time. String values can be passed only to string parameters. Numeric values can be passed only to numeric parameters.

If you define a function with parameters, it must appear with the same number of parameters when you use it in your program.

Recursive Definitions

A DEF statement may reference other defined functions (the expression may include previously defined function-names). However, a DEF statement may not be either directly or indirectly recursive (self-referencing).

Direct recursion occurs when you use the function-name in the expression of the same DEF statement. (This would be similar to writing a dictionary definition that included the word you were trying to define.)

Indirect recursion occurs when the expression contains a function-name, and in turn the expression in the DEF statement of that function (or other function subsequently referenced) includes the original function-name. (This would be similar to looking up the dictionary definition of a word, finding that the definition included other words that you needed to look up, and then discovering that the definitions led you directly back to your original word.)

Examples

```
100 DEF PAY(OT)=40*RATE+1.5*RATE*OT
110 RATE=4.00
120 PRINT PAY(3)
RUN
178
```

Defines PAY so that each time it is encountered in a program the pay is figured using the RATE of pay times 40 plus 1.5 times the rate of pay times the overtime hours.

```
100 DEF RND20=INT(RND* 20+1)
```

Defines RND20 so that each time it is encountered in a program an integer from 1 through 20 is given.

```
100 DEF FIRSTWORD$(NAME$)=SEG$(NAME$,1,POS(NAME$," ",1)-1)
```

Defines FIRSTWORD\$ to be the part of NAME\$ that precedes a space.

DEFvartype

DEFvartype

Vartypes: DEFINT, DEFSNG, DEFDBL, DEFREAL, DEFSTR

DEFINT - define as integers
 DEFSNG - define as single-precision binary floating point (32 Bit)
 DEFDBL - define as double-precision binary floating point (64 Bit)
 DEFSTR - define as a string
 DEFREAL - define as double-precision RADIX 99 floating point (64 Bit)

Format: DEFINT I,J,COUNT,LOOPNUM,DIM A(100)
 DEFSNG X,Y,COST,DIM A(100)
 DEFDBL ANGLE,MEASUR,AN,DIM C(50)
 DEFREAL SQRROOT,VALUE,N,DIM D(40)
 DEFSTR NAM,FILENAME,N,F,DIM E(75)

NOTE: DEFREAL ALL is the default mode in MYARC Advanced BASIC.

Cross Reference
 DIM, OPTION BASE, REAL, SUB

Description
 The DEFvartype instruction enables you to declare the data-type of specified variables.

Usually the name given to a variable will identify the type of variable. Example: If a variable name ends in a dollar sign (ie., A\$) then the variable is a string variable. Numeric variables can be identified in MYARC Advanced BASIC in terms of precision by the use of the following symbols as terminators attached to the end of the variable name. %, !, # or @ are termed type declaration tags. A numeric variable without such a tag is a double-precision variable in MYARC Advanced BASIC.

SYMBOL	TYPE OF VARIABLE
\$	STRING VARIABLE
%	INTEGER CONSTANT
!	SINGLE PRECISION
#	DOUBLE PRECISION
@	REAL

Variables can also be declared by use of the DEFvartype statement. The declaration must be present and executable at a lower line number than that of any use of the variable-names that it represent.

A DEFvartype statement must appear at the beginning of a line. Also, any variable defined by that statement must appear later in the program. part of an IF THEN statement.

The variable-list consists of one or more variables separated by commas. The DEFINT, DEFSNG, DEFDBL, and DEFREAL statements allow an ALL option, if this is used then all numeric variables in the program will be defined as the type specified except if they are specifically declared otherwise.

A numeric variable of the INTEGER data-type is a whole number greater than or equal to -32768 and less than 32767.

Integer variables are processed faster and use less memory than do real (or floating) point variables.

CAUTION: mixed mode floating point arithmetic is not allowed.

REAL: real numbers and integers.

BINARY: integers, single-precision, double-precision.

Mixing real numbers with either single- or double-precision will cause a mixed mode arithmetic error.

DEFvartype statements also can be used to declare the types of arrays.

TYPE-DECLARATION-TAGS override DEFvartype statements.

Programs

In the following example, DEFSTR NAM overrides DEFINT ALL such that NAM(5) will be treated as a string.

```
100 DEFINT ALL
110 DEFSTR NAM
120 NAM(5)="MYARC":X%=37.123545:: I=1.2345
130 PRINT NAME;X;I
RUN
MYARC 37.12345 1
```

DELETE

DELETE

Format

DELETE:[startline# - endline#]

Description

100-200 deletes lines 100-line 200 inclusive.

COMMAND	LINES DELETED
DELETE	All lines.
DELETE X	Line number X only.
DELETE X-	Lines from number X to the highest line number, inclusive.
DELETE -X	Lines from the lowest line number to line number X, inclusive.
DELETE X-Y or DELETE X,Y	All lines from line number X to line number Y, inclusive.

If the line-number-range does not include a line number in your program, the following conventions apply:

If line-number-range is higher than any line number in the program, the highest-numbered program line is deleted.

If line-number-range is lower than any line number in the program, the lowest-numbered program line is deleted.

If line-number-range is between lines in the program, only those lines that fall within the range specified will be deleted.

NOTE: For TI 99/4A Programs:

Delete will no longer be used to delete files from DISK STORAGE DEVICE. See KILL, CLOSE, FILES. However, programs that contain a "DELETE" file statement will execute exactly as they did under TI BASIC or TI EXTENDED BASIC. The token used internally will now be occupied by the KILL command. As long as the program is stored in tokenized form (program file, or DV163 merge format), then execution will not be affected. On listing the program, the word "KILL" will be listed instead of "DELETE".

DELETE--no longer applies to files. DELETE applies to line numbers ONLY. To delete files, see KILL.

DELSPRITE --Subprogram--Delete Sprite

DELSPRITE

Format

Delete Specified Sprite

```
CALL DELSPRITE(#sprite-number[,...])
```

Delete All Sprites

```
CALL DELSPRITE(ALL)
```

Cross Reference

CLEAR, SPRITE

Description

The DELSPRITE subprogram enables you to delete one or more sprites. All sprites are deleted when your program ends (either normally or because of an error), stops at a breakpoint, or changes graphics mode.

Delete Specific Sprites

Sprite-number is a numeric-expression whose value specifies the number of the sprite as assigned in the SPRITE subprogram. The sprite can reappear if it is redefined by the SPRITE subprogram, or if the LOCATE subprogram is called.

Delete All Sprites

If you enter the ALL option, all sprites are deleted, and can reappear only if redefined by the SPRITE subprogram.

Examples

```
100 CALL DELSPRITE(#3)
Deletes sprite number 3.
```

```
100 CALL DELSPRITE(#4,#3*C)
Deletes sprite number 4 and the sprite whose number is found by multiplying 3 by C.
```

```
100 CALL DELSPRITE(ALL)
Deletes all sprites.
```


DIM --Dimension**DIM**

Format

DIM array-name(integer1[,... integer7])[,array-name...]

Cross Reference

OPTION BASE

Description

The DIM instruction enables you to dimension (reserve space for) arrays with one to seven dimensions.

You can use DIM as either a program statement or a command.

The array-name must be a valid variable name. It cannot be used as the name of a variable or as the name of another array. An array is either numeric or string, depending on the array-name.

The integer is the upper limit of element numbers in a dimension.

If a program includes an OPTION BASE 1 statement, the first element is element 1, so the number of elements is equal to the integer plus 1.

A string array cannot have more than 16383 elements. For numeric arrays, a DEFINT array cannot have more than 32767 elements, and a floating point array cannot have more than 16383 elements. The number of integers in parentheses following the array-name determines the number of dimensions (1-7) in the array.

You can optionally specify the data-type (DEFvartype) of a numeric array by replacing DIM with the data-type.

An error occurs if you try to dimension a particular array more than once.

Note that you cannot use both instruction formats (DIM and data-type) to dimension the same array.

You cannot use OPTION BASE as a command.

You can dimension as many arrays with one DIM instruction as you can fit in one input line.

If you reference an array without first using a DIM instruction to dimension it, each dimension is assumed to have 11 elements (elements 0-10), or 10 elements (elements 1-10) if your program includes an OPTION BASE 1 statement.

If you use a DIM statement to dimension an array, the DIM statement must have a line number lower than that of any reference to that array. DIM statements are interpreted during pre-scan and are not executed.

A DIM statement can appear anywhere in your program, except as part of an IF THEN statement.

Referencing an Array

To reference a specific element of an array, you must use subscripts. Subscripts are numeric-expressions enclosed in parentheses immediately following the reference to the array-name. An array reference must include one subscript for each dimension in the array. If necessary, the value of a subscript is rounded to the nearest integer.

Reserving Space for Arrays

When you use DIM as a program statement, the computer reserves space for arrays when you enter the RUN instruction, before your program is actually run. If the computer cannot reserve space for an array with the dimensions you specify, the message Memory full in line-number is displayed, and the command does not execute.

When you use DIM as a command, if the computer cannot reserve space for an array with the dimensions you specify, the message Memory Full is displayed and the command does not execute.

Until you place values in an array, each element in a string array is a null string and each element in a numeric array has a value of zero.

Naming Arrays

The rules for naming array variables follow the same pattern as the rules for other type variables, namely if a variable name ends in variable type descriptor then the descriptor defines the variable type.

NOTE: if a DEFSTR statement is executed then a string array name need not end in a \$.

Array variable names ending in % ! # or @ respectively refer to integer, single-precision, double-precision variables or real variables.

Type/declaration tags such as \$, %, !, # or @ take precedence over DEFvartype all declarations.

The following statements will remove arrays from memory:

NEW, OLD, MERGE, RUN (without continue)

CALL MEMSET --sets all elements of an array to a defined value.

Examples

```
100 DIM X$(30)
```

Reserves space in the computer's memory for 31 string members of the array called X\$.

```
100 DIM D(100),B(10,9)
```

Reserves space in the computer's memory for 101 members of the array called D and 110 (11 times 10) members of the array called B.

DISPLAY

DISPLAY

Format

```

DISPLAY [print-list]
DISPLAY [AT(row,column)] [BEEP] [ERASE ALL][CLIP][INVERSE/BLINK]
        [SIZE(numeric-expression)][:print-list]

```

Cross Reference

DISPLAY USING, GRAPHICS, MARGIN, PRINT, BTIME, BCOLOR

Description

The DISPLAY instruction enables you to display numbers and strings on the screen. The numeric- and/or string-expressions in the print-list can be constants and/or variables.

The options available with the DISPLAY instruction make it more versatile for screen output than in the PRINT instruction. You can display data at any screen position, sound a tone when data items are displayed, clear the screen or a portion of the display row before displaying data, and accentuate displayed data by using the INVERSE/BLINK option.

You can use DISPLAY as either a program statement or a command.

The print-list consists of one or more print-items (items to be displayed on the screen) separated by print-separators. See PRINT for an explanation of the print-items and print-separators that make up a print-list.

Options

You can enter the following options, separated by a space, in any order.

AT--The AT option enables you to specify the beginning of the display field. Row and column are relative to the upper-left corner of the screen window defined by the margins. If you do not use the AT option, the display field begins in the far left column of the bottom row of the current screen window. Before a new line is displayed at the bottom of the window, the entire contents of the window (excluding sprites) scroll up one line to make room for the new line. The contents of the top line of the window scroll off the screen and are discarded. If you use the AT option and your print-list includes a TAB function, the TAB location is relative to the beginning of the display field. If you use the AT option and a print-item is too long to fit in the display field, either the extra characters are discarded (if you use the SIZE option) or the print-item is moved to the beginning of the next screen line (if you do not use the SIZE option).

BEEP--The BEEP option sounds a short tone when the data items are displayed.

ERASE ALL--The ERASE ALL option places a space character (ASCII code 32) in every character position in the screen window before displaying the data.

SIZE--The SIZE option is a numeric-expression whose value specifies the number of character positions to be cleared, starting from the beginning of the display field, before the data is displayed. If the numeric-expression is greater than the number of characters remaining in the row (from the beginning of the display field to the right margin), or if you do not use the SIZE option, the display row is cleared from the beginning of the display field to the right margin.

New Options

CLIP--Using the CLIP option, the string represented in the "DISPLAY AT" statement will be clipped at the end of a line rather than wrapping around to the next line, as it does in the default mode. The CLIP option is particularly useful when using "DISPLAY AT" within a window.

BLINK/INVERT--BLINK will cause the line displayed to BLINK on and off. This is only available in GRAPHICS(3,1) mode.

INVERT--Will cause the pixels in each character to invert their colors so the foreground- and background-colors will be inverted. This is only available in GRAPHICS(2,2), (2,3), (3,2), (3,3) modes.

Examples

```
100 DISPLAY AT(5,7):Y
```

Displays the value of Y at the fifth row, seventh column of the screen. It first clears row 5 from column 7 to the right margin.

```
100 DISPLAY ERASE ALL:B
```

Puts the blank character into all positions within the current screen window before displaying the value of B.

```
100 DISPLAY AT(R,C) SIZE(FIELDLEN)BEEP:X$
```

Displays the value of X\$ at row R, column C. First it beeps and blanks FIELDLEN characters.

Program

The following program illustrates a use of DISPLAY. It enables you to position blocks at any screen position to draw a figure or design.

Numbers must be entered as two digits (e.g., 1 would be "01", etc.). Do not press ENTER; the information is accepted as soon as the keys are pressed.

This example is valid only in Pattern Mode.

```
100 CALL CLEAR
110 CALL COLOR(27,5,5)
120 DISPLAY AT(23,1):"ENTER ROW AND COLUMN:"
130 DISPLAY AT (24,1):"ROW:COLUMN:"
140 FOR COUNT=1 TO 2
150 CALL KEY(0,ROW(COUNT),S)
160 IF S =0 THEN 150
170 DISPLAY AT(24,5+COUNT)SIZE(1):STR$(ROW(COUNT)-48)
180 NEXT COUNT
190 FOR COUNT=1 TO 2
200 CALL KEY(0,COLUMN(COUNT),S)
210 IF S =0 THEN 200
220 DISPLAY AT(24,16+COUNT)SIZE(1):STR$(COLUMN(COUNT)-48)
230 NEXT COUNT
240 ROW1=10*(ROW(1)-48)+ROW(2)-48
250 COLUMN1=10*(COLUMN(1)-48)+COLUMN(2)-48
260 DISPLAY AT(ROW1,COLUMN1)SIZE(1):CHR$(244)
270 GOTO 130
```

(Press CLEAR to stop the program.)

DISPLAY USING

DISPLAY USING

Format

DISPLAY [option-list:]USING ;format-string[:print-list]; line-number;

Cross Reference

DISPLAY, IMAGE, PRINT

Description

The DISPLAY USING instruction enables you to define specific formats for numbers and strings you display.

You can use DISPLAY USING as either a program statement or a command.

The format-string specifies the display format. The format-string is a string expression; if you use a string constant, you must enclose it in quotation marks. See IMAGE for an explanation of format-strings.

You can optionally define a format-string in an IMAGE statement, as specified by the line-number.

See DISPLAY under "Options" for an explanation of the options AT, BEEP, ERASE ALL, and SIZE.

See PRINT for an explanation of the print-list and print-options.

The DISPLAY USING instruction is identical to the DISPLAY instruction with the addition of the USING option, except that:

You cannot use the TAB function.

You cannot use any print-separator other than a comma(,), except that the print-list can end with a semicolon (;).

Examples

```
100 N=23.43
```

```
110 DISPLAY AT(10,4):USING"###.##":N
```

Displays the value of N at the tenth row and fourth column, with the format "###.##", after first clearing row 10 from column 4 to the right margin.

```
100 DISPLAY USING "###.##":N
```

Displays the value of N at the 24th row and first column, with the format "###.##".

DISTANCE --Subprogram**DISTANCE**

Format

Two Sprites

```
CALL DISTANCE(#sprite-number1,#sprite-number2,numeric-variable)
```

A Sprite and a Screen Pixel

```
CALL DISTANCE(#sprite-number,pixel-row,pixel-column,numeric-variable)
```

Cross Reference

COINC, SPRITE

Description

The **DISTANCE** subprogram enables you to ascertain the distance between two sprites or between a sprite and a specified screen pixel.

The **DISTANCE** subprogram returns the square of the distance sought. (Note that this is not the same as the distance specified by the "tolerance" in the **COINC** subprogram.)

The square of the distance is the sum of the square of the difference between pixel-rows and the square of the difference between pixel-columns. The distance between the two sprites (or the sprite and the screen pixel) is the square root of the number returned.

If the square of the distance is greater than 32767, the number returned is 32767.

Two Sprites

The distance between two sprites is considered to be the distance between the upper-left corners of the sprites.

Sprite-number1 and **sprite-number2** are numeric-expressions whose values specify the numbers of the two sprites as assigned in the **SPRITE** subprogram.

The number returned to the numeric-variable equals the square of the distance between two sprites.

A Sprite and a Screen Pixel

The distance between a sprite and a screen pixel is considered to be the distance between the upper-left corner of the sprite and the specified pixel.

Sprite-number is a numeric-expression whose value specifies the number of the sprite as assigned in the **SPRITE** subprogram.

The **pixel-row** and **pixel-column** are numeric-expressions whose values specify the position of the screen pixel.

The number returned to the numeric-variable equals the square of the distance between the sprite and the screen pixel.

Examples

```
100 CALL DISTANCE(#3,#4,DIST)
```

Sets DIST equal to the square of the distance between the upper-left corners of sprite #3 and sprite #4.

```
100 CALL DISTANCE(#4,18,89,D)
```

Sets D equal to the square of the distance between the upper-left corner of sprite #4 and position 18,89.

DRAW --Subprogram

DRAW

Format

```
CALL DRAW(line-type,pixel-row1,pixel-column1,pixel-row2,pixel-column2
[,pixel-row3,pixel-column3,pixel-row4,pixel-column4[,...]])
```

Cross Reference

CIRCLE, DCOLOR, DRAWTO, FILL, GRAPHICS, POINT, RECTANGLE

Description

The DRAW subprogram enables you to draw or erase lines between specified pixels.

The value of the numeric-expression line-type specifies the action taken by the DRAW subprogram.

TYPE	ACTION
1	Draws a line of the foreground-color specified by the DCOLOR subprogram. This is accomplished by turning on each pixel in the specified line.
0	Erases a line. This is accomplished by turning off each pixel in the specified line.
2	Reverses the status of each pixel on the specified line. (If a pixel is on, it is turned off; if a pixel is off, it is turned on.) This effectively reverses the color of the specified line.

Pixel-row and pixel-column are numeric-expressions whose values specify the pixels to be connected by the line. You must specify at least two pixels to define the beginning and end points of a line.

Pixel-row must have a value from 1 to 192. Pixel-column must have a value from 1 to 256.

You can optionally draw more lines by specifying additional pairs of pixels. The lines are not connected; each line extends from the first pixel of the pair to the second pixel of the pair. You must specify an even number of pixels.

The last pixel you specify becomes the current position used by the DRAWTO subprogram.

DRAW cannot be used in Pattern or Text modes of display. An error results if you use DRAW in Pattern or Text Modes.

In Graphics(1,2) and (1,3) modes, the computer divides each pixel-row into 32 groups of 8 pixels each. (This is most obvious when you assign a background-color other than cyan or transparent.) The computer can assign 1 foreground-color and 1 background-color, from among the 16 available colors, to each 8-pixel group.

In the Bit-Mapped modes, each pixel is independent of every other pixel on the screen.

Programs

The following program draws a large triangle on the right of the screen.

```
100 CALL GRAPHICS(3)
110 CALL CLEAR
120 CALL DRAW(1,19,185,97,115)
130 CALL DRAW(1,19,185,97,255)
140 CALL DRAW(1,97,115,97,255)
150 GOTO 150
(Press CLEAR to stop the program.)
```

The next program uses a FOR-NEXT loop to draw a pattern of lines.

```
100 CALL CLEAR
110 CALL GRAPHICS(3)
120 CALL SCREEN(6)
130 FOR X=1 TO 255 STEP 5
140 CALL DRAW(1,1,X,128,256-X)
150 NEXT X
160 GOTO 160
(Press CLEAR to stop the program.)
```

DRAWTO --Subprogram

DRAWTO

Format

CALL

DRAWTO(line-type,pixel-row,pixel-column[,pixel-row2,pixel-column2[,...]])

Cross Reference

CIRCLE, DCOLOR, DRAW, FILL, GRAPHICS, POINT, RECTANGLE

Description

The DRAWTO subprogram enables you to draw or erase lines between the current position and the specified pixels.

Line-type is a numeric-expression whose value specifies the action taken by the DRAWTO subprogram.

TYPE	ACTION
1	Draws a line of the foreground-color specified by the DCOLOR subprogram. This is accomplished by turning on each pixel in the specified line.
0	Erases a line. This is accomplished by turning off each pixel in the specified line.
2	Reverses the status of each pixel on the specified line. (If a pixel is on, it is turned off; if a pixel is off, it is turned on.) This effectively reverses the color of the specified line.

The line drawn by DRAWTO extends from the pixel in the current position to the pixel specified by the values of the numeric-expressions pixel-row and pixel-column, which becomes the new current position.

You can optionally draw more lines by specifying additional sets of pixels. A line is drawn to each specified pixel from the new current position (the previously specified pixel).

Pixel-row must have a value from 1 to 192, pixel-column must have a value from 1 to 256.

The current position is the last pixel specified the last time the DRAW or the DRAWTO subprogram was called. When you enter MYARC Advanced BASIC, the current position is the intersection of pixel-row 1 and pixel-column 1.

This default current position is restored only when you change graphics mode.

DRAWTO cannot be used in Pattern or Text modes of display. An error results if you use DRAWTO in Pattern or Text Modes.

In Graphics(1,2) and (1,3) modes, the computer divides each pixel-row into 32 groups of 8 pixels each. (This is most obvious when you assign a background-color other than cyan or transparent.) The computer can assign 1 foreground-color and 1 background-color (from among the 16 available colors), to each 8-pixel group.

Program

The following program uses DRAWTO to create a pattern across the top of the screen.

```
100 CALL GRAPHICS(3)
110 CALL CLEAR
120 A=20::B=20
130 CALL DRAW(1,A,B,A,B)
140 FOR X=1 TO 10
150 B=B+20
160 CALL DRAWTO(1,A,B)
170 CALL DRAWTO(1,A+20,B-20)
180 CALL DRAWTO(1,A+20,B)
190 CALL DRAWTO(1,A,B-20)
200 NEXT X
210 GOTO 210
(Press CLEAR to stop the program.)
```

END

END

Format

END

Cross Reference

STOP

Description

The END statement stops the execution of your program.

In addition to terminating program execution, END causes the computer to perform the following operations:

It closes all open files.

It restores the default character definitions of all characters.

It restores the default foreground color (black) and background color (transparent) to all characters.

It restores the default screen color (cyan).

It deletes all sprites.

It resets the sprite magnification level to 1.

The graphics colors (see DCOLOR) and current position (see DRAWTO) are not affected.

An END statement is not necessary to stop your program; the program automatically stops after the highest numbered line is executed.

END can be used interchangeably with the STOP statement, except that you cannot use STOP after a subprogram.

EOF

EOF

Format
 EOF(file-number)

Type
 DEFINT

Cross Reference
 ON ERROR

Description
 The EOF function returns a value indicating whether there are records remaining in a specified file.

The file-number is a numeric expression whose value specifies the number of the file as assigned in its OPEN instruction.

The value returned by the EOF function depends on the current file position. EOF always treats a file as if it were being accessed sequentially, even if it has been opened for relative access.

VALUE	MEANING
0	Not end-of-file.
(+)1	Logical end-of-file: No records remaining.
-1	Physical end-of-file: No records remaining, and no space available for more records (storage medium full).

The EOF function cannot be used with an audio cassette.

For more information about using EOF with a particular device, refer to the owner's manual that comes with that device.

Examples

```
100 PRINT EOF(3)
```

Prints a value according to whether you are at the end of the file opened as #3.

```
100 IF EOF(27)<>0 THEN 1150
```

Transfers control to line 1150 if you are at the end of the file opened as #27.

```
100 IF EOF(27) THEN 1150
```

Transfers control to line 1150 if you are at the end of the file opened as #27.

ERR --Subprogram--Error

ERR

Format

CALL ERR(error-code,error-type[,error-severity,[line-number]])

Cross Reference

ON ERROR

Description

The ERR subprogram enables you to analyze the conditions that caused a program error.

ERR is normally called from a subroutine accessed by an ON ERROR statement.

The ERR subprogram returns the error-code and error-type, and optionally the error-severity and line-number, of the most recent "uncleared" program error.

An error is "cleared" when another program error occurs or when the program ends. A RETURN statement in a subroutine accessed by an ON ERROR statement also clears the error.

ON ERROR will not trap an error caused by the RUN command.

ERR returns a two- or three-digit number to the numeric variable error-code. See Appendix J for a list of error codes and the conditions that cause them to be displayed.

An error-code of 130 indicates an input/output (I/O) error.

An error-code of 0 indicates that no error has occurred.

The error-type is a numeric variable.

When an I/O error occurs, the value returned in error-type is the number (as assigned in an OPEN instruction) of the file in which the error occurred.

A negative error-type indicates that the error occurred during program execution.

An error-type of 0 indicates that no error has occurred.

Options

The value returned to the numeric variable error-severity is always nine.

The value returned to the numeric variable line-number is the line number of the program statement that was executing when the error occurred.

Examples

```
100 CALL ERR(A,B)
```

Sets A equal to the error-code and B equal to the error-type of the most recent error.

```
100 CALL ERR(W,X,Y,Z)
```

Sets W equal to the error-code, X equal to the error-type, Y equal to the error-severity, and Z equal to the line-number of the most recent error.

Program

The following program illustrates a use of CALL ERR.

```
100 ON ERROR 130
110 CALL SCREEN(18)
120 STOP
130 CALL ERR(W,X,Y,Z)
140 PRINT W;X;Y;Z
150 RETURN NEXT
RUN
79 -1 9 110
```

An error is caused in line 110 by an improper screen-color number. Because of line 100, control is transferred to line 130. Line 140 prints the values obtained. The 79 indicates that a bad value was provided, the -1 indicates that the error occurred during program execution, the 9 is the error-severity, and the 110 indicates that the error occurred in line 110.

EXP --Function--Exponential

EXP

Format

EXP(numeric-expression)

Type

REAL

Cross Reference

LOG

Description

The EXP function returns the value of e raised to the power of the value of the numeric-expression.

EXP is the inverse of the LOG function.

The value of e is 2.718281828459.

Examples

100 Y=EXP(7)

Assigns to Y the value of e raised to the seventh power, which yields 1096.6331584290.

100 L=EXP(4.394960467)

Assigns to L the value of e raised to the 4.394960467 power, which yields 81.0414268887.

FILES

FILES

Format

CALL FILES(pathname)

Cross Reference

DOS Manual, Pathnames, Directories, OPEN, CLOSE, KILL

Description

You can use CALL FILES either as a program statement or a command.

Displays the names of the files and directories on a disk. If pathname is specified, BASIC lists all files that match that pathname. Default is all files and directories in the current directory on the current drive.

To halt listing, depress any key. To continue the listing, press another, or the same key.

Examples

CALL FILES("DSK1")

Displays files in drive 1.

CALL FILES("RD")

Displays files on RD (ramdisk).

CALL FILES("DSK.UTILITIES")

Searches all drives for the disk named "UTILITIES" and displays files.

FILL --Subprogram

FILL

Format

CALL FILL(pixel-row,pixel-column)

Cross Reference

CIRCLE, DCOLOR, DRAW, DRAWTO, GRAPHICS, POINT, RECTANGLE

Description

The FILL subprogram enables you to fill in the area surrounding a specified pixel with a specified color.

Pixel-row and pixel-column are numeric-expressions whose values specify the pixel that you want to surround with a color or pattern.

Character-code is a numeric-expression with a value from 0-215 specifying the character with which to fill the area surrounding the specified pixel.

Pixel-row must have a value from 1 to 192, pixel-column must have a value from 1 to 256.

The color of the pattern that surrounds the specified pixel is the foreground-color specified by the DCOLOR subprogram. If you have not called the DCOLOR subprogram, the default fill color is black.

The area surrounding the specified pixel is filled with the fill pattern until a screen edge or a foreground pixel (a pixel that is turned on) is encountered.

The boundaries of the area to be filled can be defined by lines drawn with the CIRCLE, DRAW, DRAWTO, POINT, RECTANGLE subprograms.

FILL cannot be used in Pattern or Text modes. An error results if you use FILL in Pattern or Text Modes.

In Graphics(1,2) and (1,3) modes the computer divides each pixel-row into 32 groups of 8 pixels each. The computer can assign a foreground-color and a background-color (from among the 16 available colors) to each 8-pixel group.

Program

The following program divides the upper portion of the screen into four horizontal columns and uses FILL to color them.

```
100 CALL CLEAR
110 CALL GRAPHICS(3)
120 CALL DRAW(1,48,0,48,256)
130 CALL DRAW(1,96,0,96,256)
140 CALL DRAW(1,144,0,144,256)
150 CALL DCOLOR(7,8)
160 CALL FILL(43,1)
170 CALL DCOLOR(11,8)
180 CALL FILL(90,1)
190 CALL DCOLOR(3,8)
200 CALL FILL(138,1)
210 CALL DCOLOR(6,8)
220 CALL FILL(188,1)
230 GOTO 230
(Press CLEAR to stop the program.)
```

FOR TO

FOR TO

Format

FOR control-variable=initial-value TO limit[STEP increment]

Cross Reference

NEXT

Description

The FOR TO instruction is used with the NEXT instruction to form a FOR-NEXT loop, which you can use to control a repetitive process.

You can use FOR TO as either a program statement or a command.

FOR-NEXT Loop Execution

When a FOR TO instruction is executed, the initial-value is assigned to the control-variable. The computer executes instructions until it encounters a NEXT instruction (the group of instructions between the FOR TO and NEXT instructions are known as a "FOR-NEXT loop"). However, if the initial-value is greater than the limit (or, if you specify a negative increment, if the initial-value is less than the limit) the FOR-NEXT loop is not executed.

When the NEXT instruction is encountered, the increment is added to the control-variable; if you do not specify an increment, the control-variable is incremented by 1. Note that if the increment is negative, the value of the control-variable is decreased.

The control-variable in the NEXT instruction must be the same as the control-variable in the FOR TO instruction. The new value of the control-variable is then compared to the limit. If you specify a positive increment (or if you do not specify an increment), the FOR-NEXT loop is repeated if the control-variable is less than or equal to the limit. If you specify a negative increment, the FOR-NEXT loop is repeated if the control-variable is greater than or equal to the limit.

If the condition for repeating the FOR-NEXT loop is met, control passes to the instruction immediately following the FOR TO instruction. If the condition is not met, the FOR-NEXT loop terminates (control passes to the statement immediately following the NEXT statement).

Specifications

The value of the numeric-expression control-variable is re-evaluated each time the NEXT instruction is executed. If you change its value while a FOR-NEXT loop is executing, you may affect the number of times the loop is repeated. A FOR-NEXT loop executes much faster if the control-variable has been declared as a DEFINT than it does if the control-variable is REAL.

The control-variable cannot be an element of an array.

The initial-value is a numeric-expression.

The value of the numeric-expression limit is not re-evaluated during the execution of a FOR-NEXT loop. If you change its value while a FOR-NEXT loop is executing, you do not affect the number of times the loop is repeated.

The value of the optional numeric-expression increment is not re-evaluated during the execution of a FOR-NEXT loop. If you change its value while a FOR-NEXT loop is executing, you do not affect the number of times the loop is repeated. The increment cannot be zero.

Nested FOR-NEXT Loops

FOR-NEXT loops may be "nested"; that is, one FOR-NEXT loop may be contained wholly within another. You must observe the following conventions:

Each FOR TO instruction must be paired with a NEXT instruction.

Each nested loop must use a different control-variable.

If a FOR-NEXT loop contains any portion of another FOR-NEXT loop, it must contain all of that FOR-NEXT loop. If a FOR-NEXT loop contains only part of another FOR-NEXT loop, an error occurs, and the message NEXT without FOR is displayed. If the FOR-NEXT loop is part of a program, the computer also displays the line-number where the error occurred.

FOR TO as a Program Statement

After you enter the RUN command, but before your program is actually run, the computer verifies that you have equal numbers of FOR TO and NEXT statements. If the numbers are not equal, the message FOR-NEXT nesting is displayed and the program is not run.

You can exit a FOR-NEXT loop by using a GOTO, ON GOTO, or IF THEN statement. If you use one of these statements to enter a loop, you could cause an error or create an infinite loop.

A FOR TO statement cannot be part of an IF THEN statement.

FOR TO as a Command

If you use FOR TO as a command, it must be part of a multiple-statement line. A NEXT instruction must also be part of the same line.

After you press ENTER to execute the command, but before the command is actually executed, the computer verifies that you have equal numbers of FOR TO and NEXT instructions. If the numbers are not equal, the message FOR-NEXT nesting is displayed and the command is not executed.

Examples

```
100 FOR A=1 TO 5 STEP 2
110 PRINT A
120 NEXT A
```

Executes the statements between this FOR and NEXT A three times, with A having values of 1, 3, and 5. After the loop is finished, A has a value of 7.

```
100 FOR J=7 TO -5 STEP -.5
110 PRINT J
120 NEXT J
```

Executes the statements between this FOR and NEXT J 25 times, with J having values of 7, 6.5, 6, ..., -4, -4.5, and -5. After the loop is finished, J has a value of -5.5.

Program

The following program illustrates a use of the FOR-TO-STEP statement. There are three FOR-NEXT loops, with control-variables of CHAR, ROW, and COLUMN.

```
100 CALL CLEAR
110 D=0
120 FOR CHAR=33 TO 63 STEP 30
130 FOR ROW=1+D TO 21+D STEP 4
140 FOR COLUMN=1+D TO 29+D STEP 4
150 CALL VCHAR(ROW,COLUMN,CHAR)
160 NEXT COLUMN
170 NEXT ROW
180 D=2
190 NEXT CHAR
200 GOTO 200
```

(Press CLEAR to stop the program.)

FREESPACE --Function**FREESPACE**

Format
FREESPACE

Type
REAL

Description

The FREESPACE function returns a number representing, in bytes, the amount of memory space available for MYARC Advanced BASIC programs and data.

The value of the numeric-expression must be zero. Other values are reserved for possible future use.

Garbage Collection

Before FREESPACE returns a value, the computer executes an activity called "garbage collection."

All "inactive" strings are deleted. Strings become inactive when they are not associated with a variable. A string may be created by the computer for its internal use; it becomes inactive when no longer needed.

All "active" strings (strings that are still associated with variables) are moved to a contiguous area at the low end of memory. This leaves all the available memory in one large, contiguous block.

The computer occasionally performs garbage collection by itself, i.e., when no memory is available because of an excess number and size of inactive strings.

Example

```
PRINT FREESPACE
```

Prints a value that indicates the amount of available memory.

GCHAR --Subprogram--Get Character

GCHAR

Format

Pattern and Text Modes

CALL GCHAR(row,column,numeric-variable)

High-Resolution Mode

CALL GCHAR(pixel-row,pixel-column,numeric-variable)

Cross Reference

GRAPHICS, HCHAR, VCHAR

Description

The GCHAR subprogram enables you to ascertain the character code of a character on the screen or the status of a screen pixel.

The meaning of the value returned to the specified numeric-variable varies according to the graphics mode.

Pattern and Text Modes

Row and column are numeric-expressions whose values specify a character position on the screen.

The value of row must be greater than or equal to 1 and less than or equal to 24.

The value of column must be greater than or equal to 1. In Pattern mode, column must be less than or equal to 32; in Text mode, column must be less than or equal to 40.

GCHAR is not affected by margin settings. Row and column are relative to the upper-left corner of the screen, not to the corner of the window defined by the margins.

The character code of the character at the specified position is returned to the numeric-variable. See Appendix B for a list of ASCII character codes.

High-Resolution Mode

The pixel-row and pixel-column are numeric-expressions whose values specify a screen pixel position.

The value of the numeric-expression pixel-row and -column must be greater than or equal to 1. In High-Resolution Mode, pixel-row must be less than or equal to 192. See Appendix K for Graphics Modes ranges.

The value of the numeric-expression pixel-column must be greater than or equal to 1 and less than or equal to the value of pixel columns on screen. See Appendix K.

The color of the specified screen pixel is given by the value returned to the numeric-variable.

Examples

```
100 CALL GCHAR(12,16,X)
```

Assigns to X the ASCII code of the character at row 12, column 16 in Pattern and Text modes.

```
100 CALL GCHAR(R,C,K)
```

Assigns to K the ASCII code of the character that is in row R, column C in Pattern and Text modes.

GOSUB --Go to a Subroutine

GOSUB

Format

GOSUB line-number
GO SUB

Cross Reference

ON GOSUB, RETURN

Description

The GOSUB statement transfers program control to the specified subroutine. A subroutine frequently is used to perform a specific operation several times in the same program.

The line-number is a numeric-expression whose value specifies the program statement at which the subroutine begins.

Use a RETURN statement to return program control to the statement immediately following the GOSUB statement that called the subroutine.

To avoid unexpected results, it is recommended that you exercise care if you use GOSUB to transfer control to or from a subprogram or into a FOR- NEXT loop.

Subroutines may be recursive (self-referencing). To avoid constructing infinite loops, it is recommended that you exercise care when using recursive subroutines.

Nested Subroutines

Subroutines may be "nested"; that is, within a subroutine you can use GOSUB to transfer control to another subroutine. Because RETURN restores program control to the statement immediately following the most recently executed GOSUB, it is important to exercise care when using nested subroutines.

For example, you might use GOSUB in your main program to transfer control to a subroutine. When the computer encounters a RETURN in the second subroutine the GOSUB in the first subroutine. Then, when a RETURN is encountered in the first subroutine, program control returns to the statement following the GOSUB in your main program.

Example

100 GOSUB 200

Transfers control to statement 200. That statement and the ones up to RETURN are executed and then control returns to the statement after the calling statement.

Program

The following program illustrates a use of GOSUB. The subroutine at line 260 figures the factorial of the value of NUMB. The whole program figures the solution to the equation

$$\text{NUMB} = X! / (Y! * (X-Y)!)$$

where the exclamation point means factorial. This formula is used to figure certain probabilities. For instance, if you enter X as 52 and Y as 5, you'll find that the number of possible five-card poker hands is 2,598,960. Both numbers entered must be positive integers less than or equal to 69.

```

100 CALL CLEAR
110 INPUT "ENTER X AND Y: ":X,Y
120 IF X<Y THEN 110
130 IF X>69 OR Y>69 THEN 110
140 IF X<0 THEN PRINT "NEGATIVE":GOTO 110 ELSE NUMB=X
150 GOSUB 260
160 NUMERATOR=NUMB
170 IF Y<0 THEN PRINT "NEGATIVE":GOTO 110 ELSE NUMB=Y
180 GOSUB 260
190 DENOMINATOR=NUMB
200 NUMB=X-Y
210 GOSUB 260
220 DENOMINATOR=DENOMINATOR*NUMB
230 NUMB=NUMERATOR/DENOMINATOR
240 PRINT "NUMBER IS";NUMB
250 STOP
260 REM CALCULATE FACTORIAL
270 IF NUMB<2 THEN NUMB=1::GOTO 320
280 MULT=NUMB-1
290 NUMB=NUMB*MULT
300 MULT=MULT-1
310 IF MULT>1 THEN 290
320 RETURN

```

GOTO**GOTO**

Format

GOTO line-number
GO TO

Cross Reference

ON GOTO

Description

The GOTO statement unconditionally transfers program control to the specified program statement.

The line-number is a numeric-expression whose value specifies the program statement to which unconditional program control is transferred.

To avoid unexpected results, it is recommended that you exercise care if you use GOTO to transfer control to or from a subroutine or into a FOR-NEXT loop.

Program

The following program shows the use of GOTO in line 160. Any time that line is reached, the program executes line 130 next and proceeds from that new point.

```
100 REM ADD 1 THROUGH 100
110 ANSWER=0
120 NUMB=1
130 ANSWER=ANSWER+NUMB
140 NUMB=NUMB+1
150 IF NUMB>100 THEN 170
160 GOTO 130
170 PRINT "THE ANSWER IS";ANSWER
RUN
THE ANSWER IS 5050
```

GRAPHICS --Subprogram**GRAPHICS**

Format

CALL GRAPHICS(graphics-mode1, graphics-mode2)

Cross Reference

CHAR, CIRCLE, COLOR, DCOLOR, DRAW, DRAWTO, FILL, MARGIN, POINT, RECTANGLE, SCREEN

Description

The GRAPHICS subprogram enables you to select the graphics-mode that offers you the combination of text and graphics capabilities that best suits the particular needs of your program.

Graphics-mode is defined by a pair of numbers, the first of which defines the screen width (i.e., 1=32 characters, 2=40 characters, 3=80 characters). the second defines the mode the display is currently operating at (i.e., text or bit-mapped). See Appendix K for a more detailed description of each graphics mode.

When you enter MYARC Advanced BASIC, the computer is in Pattern Mode.

Whenever you use the CALL GRAPHICS subprogram, the computer does the following:

Clears the entire screen.

Restores the default character definitions of all characters.

Restores the default foreground-color (black) and background-color (transparent) to all characters.

Restores the default graphics foreground-color (black) and background-color (transparent).

Restores the default screen color (cyan).

Deletes all sprites.

Resets all sprites.

Resets the sprite magnification level to 1.

Restores the default current position (pixel-row 1, pixel-column 1).

Turns off all sound.

Pattern Mode

In Pattern Mode, the screen is considered to be a grid 24 characters high and 32 characters wide. Each character is 8 pixels high and 8 pixels wide. The 256 available characters are divided into 32 sets of 8 characters each. You can use the COLOR subprogram to assign a foreground- and a background-color, from among the 16 available colors, to each character set.

In Pattern Mode, you have access to sprites.

The DCOLOR subprogram has no effect in Pattern Mode. If you use a CIRCLE, DRAW, DRAWTO, FILL, POINT or RECTANGLE subprogram, the error message Graphics mode error in line-number is displayed.

Text Modes

In Text Modes, the screen is considered to be a grid 24 characters high and 40 characters wide (Graphics(2,1)) or 26 c_h_ and 80 c_w_ (Graphics(3,1)). Each character is 8 pixels high and 6 pixels wide.

You can use the SCREEN subprogram to assign one foreground- and one background-color from among the 16 available colors. The colors you select are assigned to all 256 characters.

In Text Mode, you do not have access to sprites (the SPRITE subprogram has no effect in Text Mode). Using the COLOR subprogram to assign colors to sprites has no effect.

The DCOLOR subprogram has no effect in Text Mode. If you use a CIRCLE, DRAW, DRAWTO, FILL, POINT or RECTANGLE subprogram, the error message Graphics mode error in line-number is displayed.

Graphics(1,2) and (1,3)

In these modes, the screen is considered to be a grid 192 pixels high and 256 pixels wide.

You can use the DCOLOR subprogram to assign colors to the graphics you display.

Use the COLOR subprogram only to assign colors to sprites; any other use of the COLOR subprogram causes an error.

You can use the DCOLOR subprogram to assign color to the graphics you display.

Use the COLOR subprogram only to assign colors to sprites; any other use of the COLOR subprogram causes an error.

In these modes, you have access to sprites.

In order to maintain compatibility with MYARC Extended BASIC II, CALL GRAPHICS(1), (2), and (3) will be supported as follows:

```
CALL GRAPHICS (1) = CALL GRAPHICS(1,1)
CALL GRAPHICS (2) = CALL GRAPHICS(2,1)
CALL GRAPHICS (3) = CALL GRAPHICS(1,2)
```

All programs using these older calls to graphics will run with no modification.

In these modes the computer divides each pixel-row into 32 groups of 8 pixels. The computer can assign a foreground-color and a background-color (from among the 16 available colors) to each 8-pixel group.

Bit-Mapped Graphics Modes

In bit-mapped graphics modes, each pixel on the screen is totally independent from any other. Each character of text is 8 pixels high and 6 pixels wide.

Graphics Mode	Screen Dimension (Pixel)	Screen Dimension (Text)
2,2	256 x 212	40 x 26
2,3	256 x 212	40 x 26
3,2	512 x 212	80 x 26
3,3	512 x 212	80 x 26

Example

```
100 CALL GRAPHICS (3)
```

As a statement, changes the graphics mode to High-Resolution during program execution until execution stops or until another statement changes the Graphics Mode to something else.

HCHAR --Subprogram--Horizontal Character

HCHAR

Format

CALL HCHAR(row,column,character-code[,number of repetitions])

Cross Reference

DCOLOR, GCHAR, GRAPHICS, VCHAR

Description

The HCHAR subprogram enables you to place a character on the screen and repeat it horizontally.

Row and column are numeric-expressions whose values specify the position on the screen where the character is displayed.

The value of row must be greater than or equal to 1, and must be less than or equal to the total number of rows available on the screen. The value of column must be greater than or equal to 1 and must be less than or equal to the total number of columns available on the screen.

HCHAR is not affected by margin settings.

Character-code is a numeric-expression with a value from 0-255, specifying the number of the character. See Appendix B for a list of ASCII character codes.

The optional number-of-repetitions is a numeric-expression whose value specifies the number of times the character is repeated horizontally. If the repetitions extend past the end of a row they continue from the first character of the next row. If the repetitions extend past the end of the last row they continue from the first character of the first row.

If you use HCHAR to display a character on the screen, and then later use CHAR, COLOR, or DCOLOR to change the appearance of that character, the result depends on the Graphics Mode.

In Pattern and Text Modes, the displayed character changes to the newly specified pattern and/or color(s).

In other modes the displayed character remains unchanged.

Examples

```
100 CALL HCHAR(12,16,33)
```

Places character 33 (an exclamation point) in row 12, column 16.

```
100 CALL HCHAR(1,1,ASC("!"),768)
```

Places an exclamation point in row 1, column 1, and repeats it 768 times, which fills the screen in Pattern Mode.

```
100 CALL HCHAR(R,C,K,T)
```

Places the character with an ASCII code specified by the value of K in row R, column C and repeats it T times.

HEX\$

HEX\$

Format

HEX\$(numeric-expression)

Description

Returns hexadecimal string equivalent to numeric-expression.

This command functions on integer values only.

Example

A\$ = HEX\$(-1)::PRINT A\$

The computer prints:

FFFF

IF THEN ELSE

IF THEN ELSE

Format

```
IF relational-expression THEN line-number1 [ELSE line-number2]
   numeric-expression          statement1      statement2
```

Description

The IF THEN statement enables you to transfer program control to a specified program statement, or to execute a statement or series of statements, based on the status of a condition you specify.

The condition tested by the IF THEN statement can be either a relational-expression or a numeric-expression.

A relational-expression is "true" if it accurately describes the relationship between the variables it references; otherwise, it is "false."

A numeric-expression is "false" if it has a value of zero; otherwise, it is "true."

The action specified following THEN or ELSE can be either a line-number or a statement.

If the conditional requirement is met and you specify a line-number, program control is transferred to the program statement located at that line-number.

If the conditional requirement is met and you specify a statement, the specified statement is executed. The statement may be either a single program statement or a series of program statements separated by a double colon (::) statement separator symbol.

If the tested condition is "true," the computer performs the action specified following THEN.

If the tested condition is "false" and you use the ELSE option, the computer performs the action specified following ELSE. Note: A statement separator symbol (::) must not immediately precede ELSE, as this causes a syntax error.

If the tested condition is "false" and you do not use the ELSE option, there are three possibilities.

IF THEN is followed by a statement, program execution proceeds with the next program line.

IF THEN is followed by a line-number only, program execution proceeds with the next program line.

IF THEN is followed by a line-number and a statement separator, program execution proceeds with the statements after the statement separator. Note: In this case, the statement separator symbol functions as an implied ELSE.

An IF THEN statement cannot contain a DEF, DIM, FOR, NEXT, OPTION BASE, SUB, or SUBEND instruction.

Examples

```
100 IF X>5 THEN GOSUB 300 ELSE X=X+5
```

If X is greater than 5, then 300 is executed. When the subroutine is ended control returns to the line following this line. If X is 5 or less, X is set equal to X+5 and control passes to the next line.

```
100 IF Q THEN C=C+1::GOTO 500 ELSE L=L/C::GOTO 300
```

If Q is not zero, then C is set equal to C+1 and control is transferred to line 500. If Q is zero, the L is set equal to L/C and control is transferred to line 300.

```
100 IF A$="Y" THEN COUNT=COUNT+1::DISPLAY AT(24,1):"HERE WE GO AGAIN!":GOTO 300
```

If A\$ is not equal to "Y", then control passes to the next line. If A\$ is equal to "Y", then COUNT is incremented by 1, a message is displayed, and control is transferred to line 300.

```
100 IF HOURS =40 THEN PAY=HOURS*WAGE ELSE PAY=HOURS*WAGE+.5*WAGE*(HOURS-40)::OT=1
```

If HOURS is less than or equal to 40, then PAY is set equal to HOURS*WAGE and control passes to the next line. If HOURS is greater than 40, then PAY is set equal to HOURS*WAGE+.5*WAGE*(HOURS-40), OT is set equal to 1, and control passes to the next line.

Program

The following program illustrates a use of IF THEN ELSE. It accepts up to 1000 numbers and then prints them in order from smallest to largest.

```
100 CALL CLEAR
110 DIM VALUE(1000)
120 PRINT "ENTER VALUES TO BE SORTED.":"ENTER '9999' TO END ENTRY."
130 FOR COUNT=1 TO 1000
140 INPUT VALUE(COUNT)
150 IF VALUE(COUNT)=9999 THEN 170
160 NEXT COUNT
170 COUNT=COUNT-1
180 PRINT "SORTING."
190 FOR SORT1=1 TO COUNT
200 FOR SORT2=COUNT TO SORT1+1
210 IF VALUE(SORT1)>VALUE(SORT2) THEN
    TEMP=VALUE(SORT1)::VALUE(SORT1)=VALUE(SORT2)::VALUE(SORT2)=TEMP
220 NEXT SORT2
230 NEXT SORT1
240 FOR SORTED=1 TO COUNT
250 PRINT VALUE(SORTED)
260 NEXT SORTED
```

IMAGE

IMAGE

Format

IMAGE format-string

Cross Reference

DISPLAY USING, PRINT USING

Description

The IMAGE statement enables you to specify the format in which numbers or strings are printed or displayed by a PRINT USING or DISPLAY USING statement.

The format-string is a string constant.

A format-string containing a quotation mark or leading or trailing spaces must be enclosed in quotation marks. A format-string included in a PRINT USING or DISPLAY USING statement (rather than as part of an image statement) must be enclosed in quotation marks.

Any character can be part of a format-string. Certain combinations of characters are interpreted as format-fields, as described below.

An IMAGE statement is not executed.

An IMAGE statement cannot be part of a multiple-statement line.

Format-Fields

A format-string can consist of one or more format-fields, each specifying the format of one print-item. Format-fields can be separated by any character except a decimal point or a pound sign.

A format-field may consist of the following characters:

A pound sign (#) is replaced by a character from a print-item in the print-list of a PRINT USING or DISPLAY USING instruction. Allow one pound sign for each digit or character; allow one pound sign for the minus sign if necessary. If you do not allow as many pound signs as are necessary to represent the print-item, each pound sign is replaced by an asterisk (*). If you use more pound signs than are necessary to represent the print-item, each pound sign is replaced by a space. Added spaces precede a number (which right-justifies the number); added spaces follow a string (which left-justifies the string).

To indicate that a number is to be given in scientific notation, circumflexes (^) must be given for the E and power numbers. There must be four or five circumflexes, and 10 or fewer characters (minus sign, pound signs, and decimal point) when using the E format.

The decimal point separates the whole and fractional portions of numbers, and is printed where it appears in the IMAGE statement.

All other letters, numbers, and characters are printed exactly as they appear in the IMAGE statement.

Format-string may be enclosed in quotation marks. If it is not enclosed in quotation marks, leading and trailing spaces are ignored. However, when used directly in PRINT...USING or DISPLAY...USING, it must be enclosed in quotation marks.

Each IMAGE statement may have space for many images, separated by any character except a decimal point. If more values are given in the PRINT USING or DISPLAY USING statement than there are images, then the images are reused, starting at the beginning of the statement.

If you wish, you may put format-string directly in the PRINT...USING or DISPLAY USING statement immediately following USING. However, if a format-string is used often, it is more efficient to refer to an IMAGE statement.

Examples

```
100 IMAGE $####.###
110 PRINT USING 100:A
```

IMAGE \$####.### allows printing of any number from -999.999 to 9999.999. The following illustrates how some sample values would be printed or displayed:

VALUE	APPEARANCE
-999.999	\$-999.999
-34.5	\$ -34.500
0	\$ 0.000
12.4565	\$ 12.457
6312.991	\$6312.999
99999999	\$*****

```
100 IMAGE ANSWERS ARE ### AND ##.##
110 PRINT USING 100:A,B
```

Allows printing of two numbers. The first may be from -99 to 999 and the second may be from -9.99 to 99.99. The following illustrates how some sample values would be printed or displayed:

VALUES	APPEARANCE
-99 -9.99	ANSWERS ARE -99 AND -9.99
-7 -3.459	ANSWERS ARE -7 AND -3.46
0 0	ANSWERS ARE 0 AND .00
14.8 12.75	ANSWERS ARE 15 AND 12.75
795 852	ANSWERS ARE 795 AND *****
-984 64.7	ANSWERS ARE *** AND 64.70

```
300 IMAGE DEAR ####
310 PRINT USING 300:X$
```

Allows printing a four-character string. The following illustrates how some sample values would be printed or displayed:

VALUES	APPEARANCE
JOHN	DEAR JOHN,
TOM	DEAR TOM ,
RALPH	DEAR ****,

Programs

The following program illustrates a use of IMAGE. It reads and prints seven numbers and their totals.

```

100 CALL CLEAR
110 IMAGE $####.##
120 IMAGE " ####.##"
130 DATA 233.45,-147.95,8.4,37.263,-51.299,85.2,464
140 TOTAL=0
150 FOR A=1 TO 7
160 READ AMOUNT
170 TOTAL=TOTAL+AMOUNT
180 IF A=1 THEN PRINT USING 110:AMOUNT ELSE PRINT USING 120:AMOUNT
190 NEXT A
200 PRINT "-----"
210 PRINT USING "$####.##":TOTAL

```

```

RUN
$ 233.45
-147.95
  8.40
 37.26
-51.30
 85.20
464.00
-----
$ 629.06

```

Lines 110 and 120 set up the images. They are the same except for the dollar sign in line 110. To keep the blank space where the dollar sign was, the format-string in line 120 is enclosed in quotation marks.

Line 180 prints the values using the IMAGE statements.

Line 210 shows that the format can be put directly in the PRINT USING statement.

The amounts are printed with the decimal points aligned.

The following program shows the effect of using more values in the PRINT USING statement than there are images in the IMAGE statement.

```

100 IMAGE ###.##,###.##
110 PRINT USING 100:50.34,50.34,37.26,37.26
RUN
50.34, 50.3
37.26, 37.3

```

INIT --Subprogram--Initialize**INIT**

Format

CALL INIT[(numeric-expression)]

Cross Reference

LINK, LOAD

Description

The INIT subprogram reserves memory space to enable the computer to run assembly-language subprograms.

In addition to allocating memory space, INIT removes any assembly-language subprograms that were previously loaded into memory.

The value of the optional numeric-expression specifies the number of bytes of memory you want to reserve for assembly-language subprograms.

If you do not enter a numeric-expression, the computer reserves 8K (8192) bytes of memory.

If the value of the numeric-expression is 0, the computer reserves no memory for assembly-language subprograms and releases all memory previously allocated.

The maximum amount of memory space that you can allocate by using INIT is 49,152 bytes.

If you do not call INIT before the first time you use the LOAD subprogram to load an assembly-language subprogram from an external device into memory, the computer reserves 8K bytes of memory.

Although it is not necessary to call INIT in your program, you may wish to do so to remove previously loaded subprograms from memory. Lowering or eliminating the amount of memory reserved for assembly-language subprograms leaves more memory free to be used by MYARC Advanced BASIC.

Examples

CALL INIT

Allocates 8K bytes of memory space.

CALL INIT(200)

Allocates 200 bytes of memory space.

CALL INIT(0)

Releases all memory previously allocated.

INPUT**INPUT**

Format

Keyboard Input

```
INPUT [input-prompt:]variable-list
```

File Input

```
INPUT #file-number[,REC record-number]
```

Cross Reference

ACCEPT, EOF, LINPUT, OPEN, REC, TERMCHAR

Description

The INPUT statement suspends program execution to enable you to enter data from the keyboard. INPUT can be used to retrieve data from an external device.

The variable-list consists of one or more variables separated by commas. Values are assigned to the variables in the variable-list in the order they are input. A value assigned to a numeric variable must be a number; a value assigned to a string variable may be a string or a number.

Variables are assigned values sequentially in the variable-list. A value can be assigned to a variable, and then that variable can be used as a subscript later in the same variable-list.

Input from the Keyboard

If you do not specify a file-number, the program pauses to accept input from the keyboard.

If you enter an input-prompt, it appears at the beginning of the input field, followed immediately by the flashing cursor.

The input-prompt is a string-expression; if you use a string constant, you must enclose it in quotation marks.

If you do not enter an input-prompt, a question mark (?) appears at the beginning of the input field, followed by a space. The flashing cursor appears in the character position following the space.

The input field begins in the far left column of the bottom row of the screen window defined by the margins. You can enter up to 157 characters from the keyboard; however, an exceptionally long entry may not be processed correctly by the computer.

The values entered to the variable-list of one INPUT statement must be separated by commas. You must enter the same number of values as there are variables in the variable-list.

A string value entered from the keyboard can optionally be enclosed in quotation marks. However, a string containing a comma, a quotation mark, or

leading or trailing spaces must be enclosed in quotation marks. A quotation mark within a string is represented by two adjacent quotation marks.

You normally press ENTER to complete keyboard input; however, you can also use Alt 7(AID), Alt 9(BACK), Alt 5(BEGIN), CLEAR, Alt 6(PROC'D), DOWN ARROW, or UP ARROW. You can use the TERMCHAR function to determine which of these keys was pressed to exit from the previous INPUT, LINPUT, or ACCEPT instruction.

Note that pressing CLEAR during keyboard input normally causes a break in the program. However, if your program includes an ON BREAK NEXT statement, you can use CLEAR to exit from an input field.

The computer sounds a short tone to signal that it is ready to accept keyboard input.

Examples

100 INPUT X
Allows the input of a number.

100 INPUT X\$,Y
Allows the input of a string and a number.

100 INPUT "ENTER TWO NUMBERS: ":A,B
Displays the prompt ENTER TWO NUMBERS and then allows the entry of two numbers.

100 INPUT A(J),J
First evaluates the subscript of A and then accepts data into that element of the array A. Then a value is accepted into J.

100 INPUT J,A(J)
First accepts data into J and then accepts data into the Jth element of the array A.

Program

The following program illustrates a use of INPUT from the keyboard.

```

100 CALL CLEAR
110 INPUT "ENTER YOUR FIRST NAME: ":FNAME$
120 INPUT "ENTER YOUR LAST NAME: ":LNAME$
130 INPUT "ENTER A THREE DIGIT NUMBER: ":DOLLARS
140 INPUT "ENTER A TWO DIGIT NUMBER: ":CENTS
150 IMAGE OF $###.## AND THAT IF YOU
160 CALL CLEAR
170 PRINT "DEAR ";FNAME$;",";": :
180 PRINT "      THIS IS TO REMIND YOU"
190 PRINT "THAT YOU OWE US THE AMOUNT"
200 PRINT USING 150:DOLLARS+CENTS/100
210 PRINT "IF YOU DO NOT PAY US, YOU WILL SOON"
220 PRINT "RECEIVE A LETTER FROM OUR"

```

```
230 PRINT "ATTORNEY, ADDRESSED TO"  
240 PRINT FNAME$;" ";LNAME$;"!": :  
250 PRINT TAB(15);"SINCERELY," : : :TAB(15);"I. DUN YOU": : : :  
260 GOTO 260  
(Press CLEAR to stop the program.)
```

Lines 110 through 140 allow the person using the program to enter data, as requested with the input-prompts.

Lines 170 through 250 construct a letter based on the input. (Be certain to enter the colons exactly as indicated, because they control line spacing.)

Input from a File

If you include a file-number, input is accepted from the specified device.

The file-number is a numeric-expression whose value specifies the number of the file as assigned in its OPEN instruction.

If necessary, file-number is rounded to the nearest integer.

If you use the REC option, the record-number is a numeric-expression whose value specifies the number of the record from which you want to input to the variable-list. The records in a file are numbered sequentially, starting with zero. The REC option can be used only with a file opened for RELATIVE access.

If necessary, record-number is rounded to the nearest integer.

You can accept input only from files opened in INPUT or UPDATE mode. DISPLAY files must have fewer than 161 characters in each record to be used with an INPUT statement; however, an exceptionally long record may not be processed correctly by the computer.

If there are more variables in the variable-list than there are values in the current record, the computer proceeds as follows:

In the case of INTERNAL FIXED records, null strings are assigned to the remaining variables, causing a program error if any of the remaining variables are numeric.

For other records, the computer reads the next record in the file, and uses its values to complete the variable-list.

If there are more values in the current record than are necessary to fill the variable-list, the remaining values are discarded. However, if the variable-list ends with a comma, the computer is placed in an input-pending condition. The remaining values are assigned to the variables in the variable-list of the next INPUT statement unless that statement includes the REC option, in which case the remaining values are discarded.

Examples

```
100 INPUT #1:X$
```

Puts into X\$ the next value available in the file that was opened as #1.

```
100 INPUT #23:X,A,LL$
```

Puts into X, A, and LL\$ the next three values from the file that was opened as #23 with data in INTERNAL format.

```
100 INPUT #11,REC 44:TAX
```

Puts into TAX the first value of record number 44 of the file that was opened as #11 with RELATIVE file organization.

```
100 INPUT #3:A,B,C,
```

```
110 INPUT #3:X,Y,Z
```

Puts into A, B, and C the next three values from the file opened as #3. The comma after C creates an input-pending condition, and because the INPUT statement in line 110 has no REC clause, the computer assigns to X, Y, and Z data values beginning where the previous INPUT statement stopped.

Program

The following program illustrates a use of the INPUT statement. It opens a file on disk drive 1 called TEST and writes 5 records to the file. It then goes back and reads the records and displays them on the screen.

```
100 OPEN #1:"DSK1.TEST",SEQUENTIAL,INTERNAL,OUTPUT,FIXED 64
```

```
110 FOR A=1 TO 5
```

```
120 PRINT #1:"THIS IS RECORD",A
```

```
130 NEXT A
```

```
140 CLOSE #1
```

```
150 CALL CLEAR
```

```
160 OPEN #1:"DSK1.TEST",SEQUENTIAL,INTERNAL,INPUT,FIXED 64
```

```
170 PRINT
```

```
180 FOR B=1 TO 5
```

```
190 INPUT #1:A$,C
```

```
200 PRINT A$;C
```

```
210 NEXT B
```

```
220 CLOSE #1
```

```
RUN
```

```
THIS IS RECORD 1
```

```
THIS IS RECORD 2
```

```
THIS IS RECORD 3
```

```
THIS IS RECORD 4
```

```
THIS IS RECORD 5
```

INT --Function--Integer**INT**

Format

INT(numeric-expression)

Type

Real

Description

The INT function returns the largest integer not greater than the value of the numeric-expression.

If the value of the numeric-expression is an integer, INT returns the value of the numeric-expression itself. If the numeric-expression is not an integer, INT returns the largest integer not greater than the numeric-expression.

Examples

```
100 PRINT INT(3.4)
Prints 3.
```

```
100 X=INT(3.9)
Sets X equal to 3.
```

```
100 P=INT(3.9999999999)
Sets P equal to 3.
```

```
100 DISPLAY AT(3,7):INT(4.0)
Displays 4 at the third row, seventh column of the current screen window.
```

```
100 N=INT(-3.9)
Sets N equal to -4.
```

```
100 K=INT(-3.00000001)
Sets K equal to -4.
```

JOYST --Subprogram--Joystick

JOYST

Format

CALL JOYST(key-unit,x,y)

Description

The JOYST subprogram enables you to ascertain the position of either of the Joystick Controllers.

The numeric-expression key-unit can have a value of 1 or 2, specifying the joystick you are testing.

The position of the specified joystick is returned in the numeric variables x and y as follows:

POSITION	X	Y
Center	0	0
Up	0	(+)4
Upper Right	(+)4	(+)4
Right	(+)4	0
Lower Right	(+)4	-4
Down	0	-4
Lower Left	-4	-4
Left	-4	0
Upper Left	-4	(+)4

If the specified joystick is not connected to the computer, x and y are both returned as 0.

Example

```
100 CALL JOYST(1,X,Y)
```

Returns values in X and Y according to the position of joystick number 1.

Program

The following program illustrates a use of the JOYST subprogram. It creates a sprite and then moves it around according to the input from a joystick.

```
100 CALL CLEAR
110 CALL SPRITE(#1,33,5,96,128)
120 CALL JOYST(1,X,Y)
130 CALL MOTION(#1,-Y*4,X*4)
140 GOTO 120
(Press CLEAR to stop the program.)
```

KEY

KEY

Expanded usage of the KEY command has been incorporated into the MYARC 9640.

Using the familiar command CALL KEY, the KEY subprogram is invoked. This KEY subprogram has been enlarged to also cover MYARC Advanced BASIC.

In addition, using the newly added KEY (not CALL KEY) commands, you can now change or tailor the functions performed by individual program function keys in various ways to accommodate your own programming needs. Three different constructs are used to change and/or utilize your redefined keys.

CALL KEY --subprogram

Format

CALL KEY(key-unit,key,status)

Description

The KEY subprogram enables you to transfer one character from the keyboard directly to a program.

CALL KEY can sometimes replace an INPUT statement, especially for the input of a single character.

The numeric-expression key-unit can have a value from 0 to 6, as explained below.

The character code of the key pressed is returned in the numeric variable key. If no key is pressed, a value of 0 is returned.

See Appendix B for a list of the available characters.

The keyboard status is returned in the numeric variable status as explained below.

Because the character represented by the key pressed is not displayed on the screen, the information already on the screen is not disturbed.

Key-Unit Options

The value you specify for the key-unit determines what portion of the keyboard is active and how the key pressed is interpreted.

KEY-UNIT	RESULT
0	Console keyboard, in mode previously specified by CALL KEY.
1	Only the left side of the keyboard is active.
2	Only the right side of the keyboard is active.

- 3 Places keyboard in standard TI 99/4 mode. (Most Command Module software use this mode.) Both upper- and lower-case alphabetical characters are returned by the computer as upper-case only, and the function keys (BACK, BEGIN, etc.) return codes 1 through 15. No control characters are active.
- 4 Remaps the keyboard in the PASCAL mode. Both upper- and lower-case alphabetical character codes are returned by the computer, and the function keys return codes from 129 through 143. The control character codes are 1 through 31.
- 5 Places the keyboard in 99/4A BASIC mode. Both upper- and lower-case alphabetical character codes are returned by the computer. The function key codes are 1 through 15, and the control key codes are 128 through 159 (and 187.)
- 6 Places the keyboard in MYARC Advanced BASIC mode. See Appendix M, Additional Extended ASCII Codes for Keyboard Mode 6.

Status

The value returned as the status can be interpreted as follows:

- 1 The same key was pressed as was returned the last time KEY was called.
- 0 No key was pressed.
- 1 A different key was pressed than was returned the last time KEY was called.

Example

```
100 CALL KEY(0,K,S)
```

Returns in K the ASCII code of any key pressed on the keyboard except SHIFT, CTRL, FCTN, and CAPS, and in S a value indicating whether a key was pressed.

Program

The following program illustrates a use of the KEY subprogram. It creates a sprite and then enables you to move it around by using the arrow keys (E, S, D, and X) without pressing FCTN. Note that line 130 returns to line 120 if no key has been pressed.

To stop the sprite's movement, press any key (except the arrow keys) on the left side of the keyboard.

```
100 CALL CLEAR
110 CALL SPRITE(#1,33,5,96,128)
120 CALL KEY(1,K,S)
130 IF S=0 THEN 120
140 IF K=5 THEN Y=-4
```

```
150 IF K=0 THEN Y=4 160 IF K=2 THEN Y=-4
170 IF K=3 THEN X=4
180 IF K=1 THEN X,Y=0
190 IF K>5 THEN X,Y=0
200 CALL MOTION(#1,Y,X)
210 GOTO 120
(Press CLEAR to stop the program.)
```

KEY COMMANDS FOR REDEFINING FUNCTION KEYS

KEY

Format

KEY numeric expression, string expression

Description

The KEY numeric expression, string expression command allows you to redefine the associated string of a specified function key. Upon invoking BASIC, each function key has an associated string function as given in TABLE 1, pages 17 and 19 in the MYARC 9640 Manual. The purpose of this command is to allow you to redefine the default for any specified function key.

Numeric expression defines the function key number that is being redefined. Valid function key numbers are 1-16.
(Note: most present-day keyboards have 10 function keys.)

String expression defines the string that is to be returned when the function key is pressed.

Either in the imperative mode (cursor blinking), or when a program is asking for input while running, pressing the function key will return its associated string.

Format

ON KEY (numeric expression) GOSUB line number

KEY(numeric expression) ON/OFF

Description

The ON KEY (numeric expression) GOSUB line number and the KEY(numeric expression) ON/OFF commands enable a running program to be halted and execution transferred to a predefined subprogram when a function key is pressed.

To successfully allow the program to transfer to the desired subroutine, you must first tell MYARC Advanced BASIC which function key is to transfer control to where, and also to tell the basic interpreter and operating system to start looking for the occurrence of a function key depression.

The numeric expression must be a valid function key number from 1 to 16. The line number tells the basic interpreter where the subroutine is to start once

the function key is pressed. However, you must also inform the operating system to look for, and to tell the basic interpreter that a particular function key has been pressed. This is performed by KEY(numeric expression) ON command.

When this statement is executed after an ON KEY (numeric expression) GOSUB line number for that particular function key, the function key is now considered "armed". Once armed, the basic interpreter will check with the operating system at the end of every line in the program, to determine if the function key has been pressed.

If yes, program control will be transferred to the line number given, just as if the main program had executed a GOSUB. The subroutine can end in either a RETURN, to return control to the program where it had been before being interrupted or, to a RETURN line number where execution returns to the line number given. In either case program execution will continue as if nothing had happened.

However, there are two cases when an armed function key is pressed but execution will not be interrupted:

1. If the program is executing the subroutine of a previously pressed function key and a RETURN has not yet occurred.
2. If you are in a user-defined subprogram where it would not make sense given the context.

In both cases however, the function key press will be queued for execution when possible.

In order to have the basic interpreter not check for function key presses (i.e., to unarm the function key), execute the command

```
KEY(numeric expression) OFF
```

Another way is to give a line number of 0 as a GOSUB destination.

Remember, the action of an armed function key is only valid during a running program. If you are in the basic imperative mode or if the program is asking for input from the keyboard, a function key press will produce the default or user-defined string as a response.

EXAMPLE

The following example shows how to use the KEY command to display the time on the bottom left of the screen whenever the F1 key is pressed.

```
100 ON KEY (1) GOSUB 200
110 KEY(1) ON
120 GOTO 120
200 DISPLAY AT(24,1):TIME$
210 RETURN
```

KILL**KILL**

Format

KILL file-specification

Cross Reference

CLOSE

Description

The KILL instruction removes a file from an external storage device. Although the file is not physically erased, the space it occupies becomes available for you to store another file in the future.

You can use KILL as either a program statement or a command.

The file-specification indicates the name of the file to be deleted. The file-specification is a string-expression; if you use a string constant, you must enclose it in quotation marks.

You can also remove files stored on some external devices by using the KILL option in the CLOSE instruction.

For more information about the options available with a particular device, refer to the owner's manual that comes with that device.

Example

KILL "DSK1.MYFILE"

Deletes the file named MYFILE from the diskette in disk drive 1.

Program

The following program illustrates a use of KILL.

```
100 INPUT "NAME OF FILE TO BE DELETED: ":X$
110 KILL X$
```

NOTE: For TI 99/4A PROGRAMS

Delete will no longer be used to delete files from disk storage device (see KILL, CLOSE, FILES). However programs that contain a "DELETE" file statement will execute exactly as they did under TI BASIC or TI EXTENDED BASIC. The token used internally will now be occupied by the KILL command. As long as the program is stored in tokenized form (program file, or DV163 merge format) the execution will not be affected. On listing the program the word "KILL" will be listed instead of "DELETE".

LEFT\$

LEFT\$

Format

LEFT\$(string\$,numvar)

Cross Reference

SEG\$, RIGHT\$, POS

Description

LEFT\$() returns the leftmost portion of the string represented by string\$ of length numvar.

The LEFT\$ function creates a new string but does not destroy the original string.

LEFT\$(A\$,5) is equivalent to SEG\$(A\$,1,5) if A\$ is at least 5 characters long.

If the string is shorter than the length specified, the string LEFT\$ function will pad the string with blank spaces rather than return an error condition

LEFT\$ can be used with numerical data if the number is first converted to a string using the STR\$(n) function.

Example

```
100 B$=LEFT$("1234",3)
110 PRINT B$
120 C$=VAL(LEFT$(STR$(-1234),4)
130 PRINT C$
RUN
123
-123
```

LEFT\$ can also be used to make a program user friendly by separating first from last names, checking the first character of a response etc.

Example

```
100 INPUT "What is your full name please ":NAME$
110 SP=POS(NAME$," ",1)
120 FIRST$=LEFT$(NAME$,SP-1)
130 INPUT FIRST$&" IS THE CAPITOL OF THE UNITED STATES BROOKLYN ?":ANSWER$
135 A$=LEFT$(ANSWER$,1)
140 IF A$="Y" OR A$="y" THEN PRINT "I'M SORRY ";FIRST$;" that is not
    correct":GOTO 170
150 IF A$="N" OR A$="n" THEN PRINT "* THAT IS RIGHT":STOP
160 PRINT "TYPE YES or NO as a response please ":GOTO 130
RUN
What is your full name please ? ABRAHAM LINCOLN
ABRAHAM IS THE CAPITOL OF THE UNITED STATES BROOKLYN ? NO
THAT IS RIGHT
```

LEN --Function--Length**LEN**

Format

LEN(string-expression)

Type

DEFINT

Description

The LEN function returns the number of characters in the string specified by the string-expression.

If the string-expression is a null string, LEN returns a zero.

Remember that a space is a valid character and is considered to be part of the length of a string.

Examples

```
100 PRINT LEN("ABCDE")
```

Prints 5.

```
100 X=LEN("THIS IS A SENTENCE.")
```

Sets X equal to 19.

```
100 DISPLAY LEN("")
```

Displays 0.

```
100 DISPLAY LEN(" ")
```

Displays 1.

```
100 A$="DAVID"
```

```
110 DISPLAY LEN(A$)
```

Displays 5 when A\$ equals DAVID.

LET

LET

Format

[LET]variable-list=expression

Description

The LET instruction, often called the "assignment" instruction, enables you to assign values to variables.

You can use LET as either a program statement or a command.

The variable-list consists of one or more variables separated by commas. Do not mix numeric and string variables in the same variable-list. However, you can include both DEFINT and REAL numeric variables in the same variable-list.

The value of expression is assigned to all variables in the variable-list. If the variable-list contains numeric variables, the expression must be a numeric-expression. If the variable-list contains string variables, the expression must be a string-expression.

The word LET can be optionally omitted from instruction.

Examples

100 T=4

Assigns to T the value 4.

100 X,Y,Z=12.4

Assigns to X, Y, and Z the value 12.4.

100 A=3<5

Assigns -1 to A because it is true that 3 is less than 5.

100 B=12<7

Assigns 0 to B because it is not true that 12 is less than 7.

100 L\$,D\$,B\$="B"

Assigns to L\$, D\$, and B\$ the string constant "B".

Program

The following program illustrates a use of LET.

```
100 K=1
110 K,A(K)=3
120 PRINT K;A(1)
130 PRINT A(3);A(K)
RUN
3 3
0 0
```

In line 100, the variable K is assigned the value 1.

In line 110, the variable K and the array element A(K) are assigned the value of 3. Note that when line 110 is executed, the subscript K is not assigned a new value, but has the same value it had before the line was executed. Therefore, A(K) is an expression equivalent to A(1), referring to the same element of the array.

In line 120, the values of K and A(1) are printed.

When line 130 is executed, K has a value of 3; therefore, A(K) is now an expression equivalent to A(3). Both expressions have a value of 0 (the default value) because no value has been assigned to this element of array.

LINK --Subprogram**LINK**

Format

```
CALL LINK(subprogram-name[,parameter-list])
```

Cross Reference

```
INIT, LOAD, SUB
```

Description

The LINK subprogram enables you to transfer control from a MYARC Advanced BASIC program to an assembly-language subprogram.

The subprogram-name is an entry point in an assembly-language subprogram that you have previously loaded into memory with the LOAD subprogram. The subprogram-name is a string-expression; if you use a string constant, it must be enclosed in quotation marks.

The optional parameter-list consists of one or more parameters, separated by commas, that are to be passed to the assembly-language subprogram. The contents of the parameter-list depend on the particular subprogram you are accessing.

The rules for passing parameters to an assembly-language subprogram are the same as the rules for passing parameters to a MYARC Advanced BASIC subprogram (see SUB).

Example

```
100 CALL LINK("START",1,3)
```

Links the MYARC Advanced BASIC program to the assembly-language subprogram START, and passes the values 1 and 3 to it.

LINPUT --Line Input**LINPUT**

Format

Keyboard Input

LINPUT [input-prompt:]string-variable

File Input

LINPUT #file-number[,REC record-number]:string-variable

Cross Reference

ACCEPT, EOF, INPUT, OPEN, TERMCHAR

Description

The LINPUT statement suspends program execution to enable you to enter a line of unedited data from the keyboard. LINPUT can be used also to retrieve an unedited record from an external device.

LINPUT assigns an entire line, a file record, or the remaining portion of a file record (if there is an input-pending condition) to the string-variable.

See INPUT for an explanation of keyboard- and file-input, and input options.

No editing is performed on the input data. All characters (including commas, quotation marks, colons, semicolons, and leading and trailing spaces) are assigned to the string-variable as they are encountered.

The maximum value that can be input from the keyboard is 255 characters.

LINPUT is frequently used instead of INPUT when the input data may include a comma. (A comma is not accepted as input by the INPUT statement, except as part of a string enclosed in quotation marks.)

To use LINPUT for file input the file must be in DISPLAY format.

You normally press ENTER to complete keyboard input; however, you can also use AID, BACK, BEGIN, CLEAR, PROC'D, DOWN ARROW, or UP ARROW. You can use the TERMCHAR function to determine which of these keys was pressed to exit from the previous ACCEPT, INPUT, or LINPUT instruction.

Note that pressing CLEAR during keyboard input normally causes a break in the program. However, if your program includes an ON BREAK NEXT statement, you can use CLEAR to exit from an input field.

Examples

```
100 LINPUT L$
```

Assigns to L\$ anything typed before ENTER is pressed.

```
100 LINPUT "NAME: "NM$
```

Displays NAME: and assigns to NM\$ anything typed before ENTER is pressed.


```
100 LINPUT #1,REC M:L$(M)
```

Assigns to L\$(M) the value that was in record M of the file that was opened as #1 with RELATIVE DISPLAY file organization.

Program

The following program illustrates a use of LINPUT. It reads a previously existing file and displays only the lines that contain the word "THE."

```
100 OPEN #1:"DSK1.TEXT1",INPUT,FIXED 80,DISPLAY
110 IF EOF(1) THEN CLOSE #1 :: STOP
120 LINPUT #1:A$
130 X=POS(A$,"THE",1)
140 IF X>0 THEN PRINT A$
150 GOTO 110
```

NOTE:

Remember to press the two keys, Control + Break whenever the Manual refers to "CLEAR".

LIST

LIST

Format

List to the screen

LIST [line-number-range]

List to a File (or Device)

LIST "file-specification"[:line-number-range]

Cross Reference

LLIST

Description

The LIST command displays the program (or a portion of it) currently in memory. You can also use LIST to output the program listing to an external device.

The optional line-number-range specifies the portion of the program to be listed. If you do not enter a line-number-range, the entire program is listed. The program lines are always listed in ascending order.

If you enter a file-specification, the program listing is output to the specified file or device. The file-specification, a string constant, must be enclosed in quotation marks.

The program listing is output as a SEQUENTIAL file in DISPLAY format with VARIABLE records (see OPEN); the file-specification option can be used only with devices that accept these options. For more information about listing a program on a particular device, refer to the owner's manual that comes with that device. If you do not enter a file-specification, the program listing is displayed on the screen.

You can stop the listing at any time by pressing CLEAR. Pressing any other key (except SHIFT, ALT, or CTRL) causes the listing to pause until you press a key again.

The LIST command only works with peripherals that support DISPLAY/VARIABLE type records.

The Line-Number-Range

A line-number-range can consist of a single line number, a single line number followed by a hyphen, a single line number preceded by a hyphen, or a range of line numbers.

COMMAND	LINES LISTED
LIST	All lines.
LIST X	Line number X only.
LIST X-	Lines from number X to the highest line number, inclusive.
LIST -X	Lines from the lowest line number to line number X, inclusive.
LIST X-Y or LIST X Y	All lines from line number X to line number Y, inclusive.

If the line-number-range does not include a line number in your program, the following conventions apply:

If line-number-range is higher than any line number in the program, the highest-numbered program line is listed.

If line-number-range is lower than any line number in the program, the lowest-numbered program line is listed.

If line-number-range is between lines in the program, the next higher numbered program line is listed.

Examples

LIST
Lists the entire program in memory on the display screen.

LIST 100
Lists line 100.

LIST 100-
Lists line 100 and all after it.

LIST -200
Lists all lines up to and including line 200.

LIST 100-200
Lists all lines from 100 through 200.

LLIST

LLIST

Format

LLIST[linenum1][-][linenum2][,width]

Cross Reference

LIST, LPRINT

Description

Same line format as LIST except that LLIST automatically sends list to default print device.

COMMAND	LINES LISTED
LLIST	All lines.
LLIST X	Line number X only.
LLIST X-	Lines from number X to the highest line number, inclusive.
LLIST -X	Lines from the lowest line number to line number X, inclusive.
LLIST X-Y or LLIST X Y	All lines from line number X to line number Y, inclusive.
LLIST X-Y,132	All lines from line number X to line number Y, inclusive are printed using a page width of 132 characters.

If the line-number-range does not include a line number in your program, the following conventions apply:

If line-number-range is higher than any line number in the program, the highest-numbered program line is listed.

If line-number-range is lower than any line number in the program, the lowest-numbered program line is listed.

If line-number-range is between lines in the program, the next higher numbered program line is listed.

Width is the number of characters across the page the default is 80.

If the page width depends upon an escape code or control code sequence, then that sequence must be sent to the print device using the LPRINT command.

The default device can be changed by changing the name of LPT. (See defaults.)

LPT="PIO"

Examples

LLIST

Prints the entire program in memory on the display screen.

LLIST 100

Prints line 100.

LLIST 100-

Prints line 100 and all after it.

LLIST -200

Prints all lines up to and including line 200.

LLIST 100-200

Prints all lines from 100 through 200.

LLIST 100-200,132

Prints all lines from 100 through 200 on a page width of 132 characters.

LOAD --Subprogram

LOAD

Format

File Only

```
CALL LOAD(file-specification-list)
```

Data Only

```
CALL LOAD(address,byte-list[,"",address,byte-list[,...]])
```

File and Data

```
CALL LOAD(file-specification-list,address,byte-list[,...])
```

```
CALL LOAD(address,byte-list,file-specification-list[,...])
```

Cross Reference

INIT, LINK, PEEK, PEEKV, POKEV, VALHEX

Description

The LOAD subprogram enables you to load assembly-language subprograms into memory. You can also use LOAD to assign values directly to specified CPU (Central Processing Unit) memory addresses. You can use the POKEV subprogram to assign values to VDP (Video Display Processor) memory.

To load an assembly-language subprogram, specify a file-specification-list; to assign values to CPU memory, specify an address and a byte-list (an address must always be followed by a byte-list).

You must enter at least one parameter. The first parameter you specify can be either a file-specification-list or an address.

If you wish to follow an address and byte-list with another address and byte-list, enter a file-specification-list or a null string (two-adjacent quotation marks) as a separator.

The optional file-specification-list consists of one or more file-specifications separated by commas. A file-specification is a string-expression; if you use a string constant, you must enclose it in quotation marks.

Each file-specification names an assembly-language object (program) file to be loaded into memory. The specified file can include subprogram names, so that the subprograms can be executed by the LINK subprogram.

The object file to be loaded must be in DISPLAY format with FIXED records (see OPEN). For more information about the file options available with a particular device, refer to the owner's manual that comes with that device.

You can optionally load bytes of data to a specified CPU memory address. The address specifies the first address where the data is to be loaded; if the byte-list specifies more than one byte of data, the bytes are assigned to sequential memory addresses starting with the address you specify.

The numeric-expression address must have a value from -32768 to 32767 inclusive.

You can specify an address from 0 to 32767 inclusive by specifying the actual address.

You can specify an address from 32768 to 65535 inclusive by subtracting 65536 from the actual address. This will result in a value from -32768 to -1 inclusive.

If you know the hexadecimal value of the address, you can use the VALHEX function to convert it to a decimal numeric-expression, eliminating the possible need for calculations.

If necessary, the address is rounded to the nearest integer.

The byte-list consists of one or more bytes of data, separated by commas, that are to be loaded into CPU memory starting with the specified address.

Each byte in the byte-list must be a numeric-expression with a value from 0 to 32767. If the value of a byte is greater than 255, it is repeatedly reduced by 256 until it is less than 256. If necessary, a byte is rounded to the nearest integer.

Note that you must use the INIT subprogram to reserve memory space before you use LOAD to load a subprogram.

If you call the LOAD subprogram with invalid parameters or load an object file with absolute (rather than relocatable) addresses, the computer may function erratically or cease to function entirely. If this occurs, turn off the computer, wait several seconds, then turn the computer back on again.

The Loader

LOAD uses a "relocatable linking" loader.

Because it is "relocatable," you cannot use LOAD to specify a memory address at which you want to load a file. However, the file you are loading may specify an absolute load address if it includes an AORG directive.

Because it is "linking", the object files specified in the file-specification-list can reference each other.

LOCATE --Subprogram**LOCATE**

Format

```
CALL LOCATE(#sprite-number,pixel-row,pixel-column[,...])
```

Cross Reference

```
DELSPRITE, SPRITE
```

Description

The LOCATE subprogram enables you to change the location of one or more sprites.

The sprite-number is a numeric-expression whose value specifies the number of a sprite as assigned by the SPRITE subprogram.

The pixel-row and pixel-column are numeric-expressions whose values specify the screen pixel location of the pixel at the upper-left corner of the sprite.

LOCATE can cause a sprite that has been deleted with DELSPRITE sprite-number to reappear.

Program

The following program illustrates a use of the LOCATE subprogram.

```
100 CALL CLEAR
110 CALL SPRITE(#1,33,7,1,1,25,25)
120 YLOC=INT(RND* 150+1)
130 XLOC=INT(RND* 200+1)
140 FOR DELAY=1 TO 300 :: NEXT DELAY
150 CALL LOCATE(#1,YLOC,XLOC)
160 GOTO 120
(Press CLEAR to stop the program.)
```

Line 110 creates a sprite as a fairly quickly moving red exclamation point.

Line 140 locates the sprite at a location randomly chosen in lines 120 and 130.

Line 150 repeats the process.

Also see the third example of the SPRITE subprogram.

LOG --Function--Natural Logarithm

LOG

Format

LOG(numeric-expression)

Type

REAL

Cross Reference

EXP

Description

The LOG function returns the natural logarithm of the value of the numeric-expression. LOG is the inverse of the EXP function.

The value of the numeric-expression must be greater than zero.

Examples

```
100 PRINT LOG(3.4)
```

Prints the natural logarithm of 3.4, which is 1.223775432.

```
100 X=LOG(EXP(7.2))
```

Sets X equal to the natural logarithm of e raised to the 7.2 power, which is 7.2.

```
100 S=LOG(SQR(T))
```

Sets S equal to the natural logarithm of the square root of the value of T.

Program

The following program returns the logarithm of any positive number in any base.

```
100 CALL CLEAR
110 INPUT "BASE: ":B
120 IF B =1 THEN 110
130 INPUT "NUMBER: ":N
140 IF N =0 THEN 130
150 LG=LOG(N)/LOG(B)
160 PRINT "LOG BASE";B;"OF";N;"IS";LG
170 PRINT
180 GOTO 110
(Press CLEAR to stop the program.)
```

LPR

LPR

Format
CALL LPR(x,y)

Cross Reference
POINT, DRAW, DRAWTO, LINE, PSET or preset or the current position of the mouse cursor.

Description
Last Point Referenced returns the coordinates of the last point referenced by the graphics commands.

LPT

LPT

Syntax
LPT=device name string

Cross Reference
DOS Manual, DEFAULTS, LCOPY, LTRACE, LLIST

Description
You can use LPT either as a program statement or a command.

LPT is used to modify the name of the default print device.

Example

LPT="PIO"
LPT="RS232.BA=9600,DA=8"

The default print device is accessed from BASIC in the command mode or within a program by use of the following commands: LCOPY, LTRACE, LLIST.

LTRACE

LTRACE

Cross Reference
TRACE, BREAK

Description

LTRACE is used exactly as TRACE except the output is directed towards the default print device rather than the screen.

LTRACE is a valuable aid because it is not affected by screen clearing commands such as:

CALL CLEAR, CLS, DISPLAY, ERASE ALL, CALL GRAPHICS() etc.

MAGNIFY --Subprogram**MAGNIFY**

Format

CALL MAGNIFY(numeric-expression)

Cross Reference

CHAR, SPRITE

Description

The MAGNIFY subprogram enables you to specify whether all sprites are single- or double-sized and whether they are unmagnified or magnified.

The value of the numeric-expression specifies the size and magnification "level" of all sprites. (You cannot specify the level of an individual sprite.)

LEVEL	CHARACTERISTICS
1	Single-sized, unmagnified
2	Single-sized, magnified
3	Double-sized, unmagnified
4	Double-sized, magnified

The screen position of the pixel in the upper-left corner of a sprite is considered to be the position of that sprite. That pixel remains in the same screen position regardless of changes to the magnification level.

When you enter MYARC Advanced BASIC, sprites are single-sized and unmagnified (level 1). When your program ends (either normally or because of an error), stops at a breakpoint, or changes graphics mode, the sprite magnification level is restored to 1.

Single-Sized Sprites

A single-sized sprite is defined only by the character you specify when the sprite is created.

Double-Sized Sprites

A double-sized sprite is defined by four consecutive characters, including the character that you specify when the sprite is created.

If the number of the character you specify is a multiple of 4, that character is the first of the four characters that comprise the sprite's definition. If the character number is not a multiple of 4, the next lower character that is a multiple of four is the first character of the sprite.

The first of the four characters defines the upper-left quarter of the sprite, the second character defines the lower-left quarter of the sprite, the third defines the upper-right quarter of the sprite, and the last of the four characters defines the lower-right quarter of the sprite.

Unmagnified Sprites

An unmagnified sprite occupies only the number of characters on the screen specified by the characters that define it.

A single-sized unmagnified sprite occupies 1 character position on the screen; a double-sized unmagnified sprite occupies 4 character positions.

Magnified Sprites

A magnified sprite expands to twice the height and twice the width of an unmagnified sprite. The expansion occurs down and to the right; the pixel in the upper-left corner of the sprite remains in the same screen position.

A magnified sprite has 4 times the area of an unmagnified sprite. When you magnify a sprite, each pixel of the unmagnified sprite expands to 4 pixels of the magnified sprite.

A single-sized magnified sprite occupies 4 character positions on the screen; a double-sized magnified sprite occupies 16 character positions.

Program

The following program illustrates a use of the MAGNIFY subprogram.

A little figure (single-sized, unmagnified) appears near the center of the screen. In a moment, it becomes twice as big (single-sized, magnified), covering four character positions. In another moment, it is replaced by the upper-left corner of a larger figure (single-sized, magnified), still covering four character positions. Then the full figure appears (double-sized, magnified), covering sixteen character positions. Finally it is reduced in size to four character positions (double-sized, unmagnified).

```

100 CALL CLEAR
110 CALL CHAR(148,"1898FF3D3C3CE404")
120 CALL SPRITE(#1,148,5,92,124)
130 GOSUB 230
140 CALL MAGNIFY(2)
150 GOSUB 230
160 CALL CHAR(148,"0103C3417F3F070707077E7C40000080C0C080
FCFEE2E3E0E0E06060606070")
170 GOSUB 230
180 CALL MAGNIFY(4)
190 GOSUB 230
200 CALL MAGNIFY(3)
210 GOSUB 230
220 STOP
230 REM DELAY
240 FOR DELAY=1 TO 500
250 NEXT DELAY
260 RETURN

```

Line 110 defines character 148.

Line 120 sets up sprite using character 148. By default the magnification factor is 1.

Line 140 changes the magnification factor to 2.

Line 160 redefines character 148. Because the definition is 64 characters long, it also defines characters 149, 150, and 151.

Line 180 changes the magnification factor to 4.

Line 200 changes the magnification factor to 3.

MARGIN --Subprogram**MARGIN**

Format

CALL MARGIN(left,right,top,bottom)

Cross Reference

ACCEPT, CLEAR, DISPLAY, DISPLAY USING, GRAPHICS, INPUT, LINPUT, PRINT, PRINT USING

Description

The MARGIN subprogram enables you to define screen margins. The margins you specify define a screen window that affects the operation of several instructions.

Left, right, top, and bottom are numeric-expressions whose values specify the margins.

The margins cannot "overlap"; that is, the position of the top margin must be higher on the screen than the bottom margin, and the position of the left margin must be farther left on the screen than the right margin.

When creating a screen window, you must leave the window large enough to allow entry of a command.

The valid range for margin location varies according to the graphics mode. Acceptable values for the margins in each mode are found in Appendix K.

The upper-left corner of the window defined by the margins is considered to be the intersection of row 1 and column 1 by the ACCEPT, DISPLAY, and DISPLAY USING instructions that use the AT option.

The lower-left corner of the window is considered to be the beginning of the input line by the ACCEPT, INPUT, and LINPUT instructions.

The lower-left corner of the window is considered to be the beginning of the print line by the DISPLAY, DISPLAY USING, PRINT, and PRINT USING instructions.

When the ACCEPT, INPUT, LINPUT, or PRINT USING instructions cause scrolling, scrolling occurs only in the window.

The CLEAR, GCHAR, HCHAR, and VCHAR subprograms are not affected by the margin settings.

In all modes, the margins can extend to the edges of the screen.

When you enter MYARC Advanced BASIC, the left margin is set to 3 and the right margin to 30. The top and bottom margins are set to 1 and 24 respectively. When a program running in High-Resolution Mode ends, these default margin settings are restored.

Examples

```
100 CALL MARGIN(3,30,1,24)
```

Sets all four margins to the default value in Pattern and High-Resolution Modes.

```
100 CALL MARGIN(1,40,1,24)
```

Sets the left, right, top and bottom margins to the extreme edges of the screen in the 40 column Text Mode (Graphics(2,1)).

MAX --Function--Maximum**MAX**

Format

MAX(numeric-expression1,numeric-expression2)

Type

Numeric (REAL or DEFINT)

Cross Reference

MIN

Description

The MAX function returns the larger value of two numeric-expressions.

MAX is the opposite of the MIN function.

If the values of the numeric-expressions are equal, MAX returns that value.

Examples

```
100 PRINT MAX(3,8)
Prints 8.
```

```
100 F=MAX(3E12,1800000)
Sets F equal to 3E12.
```

```
100 G=MAX(-12,-4)
Sets G equal to -4.
```

```
100 A=7::B=-5
110 L=MAX(A,B)
Sets L equal to 7 when A=7 and B=-5.
```

MEMSET

MEMSET

Format

```
CALL MEMSET(array-variable(),expression)
```

Cross Reference

DIM, SWAP

Description

The MEMSET statement will set all elements of the designated numeric or string array to the value of the expression.

Example

```
100 DIM A$(2,2),C(400)
110 CALL MEMSET(A$(),"B")
120 PRINT A$(2,1)
130 CALL MEMSET(C(),234)
140 PRINT C(0);C(400)
RUN
B
234 234
```

MERGE

MERGE

Format

```
MERGE["]file-specification["]
```

Cross Reference

SAVE

Description

The MERGE command combines a program from an external storage device with the program currently in memory. MERGE is frequently used to combine several previously written program segments into one program.

The file-specification is a string constant that indicates the name of the program on the external device. The file-specification can optionally be enclosed in quotation marks.

The lines of the external program are inserted in line-number order among the lines of the program in memory. If a line number in the external program duplicates a line number in the program in memory, the new line replaces the old line.

The MERGE command does not clear breakpoints.

A program on an external device can be merged only if it was saved with the MERGE option of the SAVE command.

Example

```
MERGE DSK1.SUB
```

Merges the program SUB into the program currently in memory.

Program

Listed below is an example of how to merge programs. If the following program is saved on DSK1 as BOUNCE with the merge option, it can be merged with other programs.

```
100 CALL CLEAR
110 RANDOMIZE
140 DEF RND50=INT(RND* 50-25)
150 GOSUB 10000
10000 FOR AA=1 TO 100
10010 QQ=RND50
10020 LL=RND50
10030 CALL MOTION(#1,QQ,LL)
10040 NEXT AA
10050 RETURN
SAVE "DSK1.BOUNCE",MERGE
NEW
```

Place the following program into the computer's memory.

```
120 CALL CHAR(96,"18183CFFFF3C1818")
130 CALL SPRITE(#1,96,7,92,128)
150 GOSUB 500
160 STOP
```

Now merge BOUNCE with the above program.

MERGE DSK1.BOUNCE

The program that results from merging BOUNCE with the above program is shown here.

```
LIST
100 CALL CLEAR
110 RANDOMIZE
120 CALL CHAR(96,"18183CFFFF3C1818")
130 CALL SPRITE(#1,96,7,92,128)
140 DEF RND50=INT(RND* 50-25)
150 GOSUB 10000
160 STOP
10000 FOR AA=1 TO 100
10010 QQ=RND50
10020 LL=RND50
10030 CALL MOTION(#1,QQ,LL)
10040 NEXT AA
10050 RETURN
```

Note that line 150 is from the program that was merged (BOUNCE), not from the program that was in memory.

MIN --Function--Minimum**MIN**

Format

MIN(numeric-expression1,numeric-expression2)

Type

Numeric

Cross Reference

MAX

Description

The MIN function returns the smaller value of two numeric-expressions. MIN is the opposite of the MAX function.

If the values of the numeric-expressions are equal, MIN returns that value.

Examples

100 PRINT MIN(3,8)

Prints 3.

100 F=MIN(3E12,1800000)

Sets F equal to 1800000.

100 G=MIN(-12,-4)

Sets G equal to -12.

100 A=7::B=-5

110 L=MIN(A,B)

Sets L equal to -5 when A=7 andB=-5.

MOD --Function

MOD

Format

MOD(numvar1,numvar2)

Description

MOD computes the arithmetic remainder (MODulo) from the expression numvar1,numvar2. The remainder is then rounded up or down to the nearest integer.

Example

```
10 FOR I=1 TO 1000
20 R = MOD(I,20)
30 PRINT I,R
40 NEXT I RUN
```

The above program prints to the screen the modulo base 20 of all integers between 1 and 1000.

MOTION --Subprogram**MOTION**

Format

```
CALL MOTION(#sprite-number,vertical-velocity,horizontal-velocity[,...])
```

Cross Reference

SPRITE

Description

The MOTION subprogram enables you to change the velocity of one or more sprites.

The sprite-number is a numeric-expression whose value specifies the number of a sprite as assigned by the SPRITE subprogram.

The vertical- and horizontal-velocity are numeric-expressions whose values range from -128 to 127. If both values are zero, the sprite is stationary. The speed of a sprite is in direct linear proportion to the absolute value of the specified velocity.

A positive vertical-velocity causes the sprite to move toward the bottom of the screen; a negative vertical-velocity causes the sprite to move toward the top of the screen.

A positive horizontal-velocity causes the sprite to move to the right; a negative horizontal-velocity causes the sprite to move to the left.

If neither the vertical- nor horizontal-velocity are zero, the sprite moves at an angle in a direction and at a speed determined by the velocity values.

When a moving sprite reaches an edge of the screen, it disappears. The sprite reappears in the corresponding position at the opposite edge of the screen.

Program

The following program illustrates a use of the MOTION subprogram.

```
100 CALL CLEAR
110 CALL SPRITE(#1,33,5,92,124)
120 FOR XVEL=-16 TO 16 STEP 2
130 FOR YVEL=-16 TO 16 STEP 2
140 DISPLAY AT(12,11):XVEL;YVEL
150 CALL MOTION(#1,YVEL,XVEL)
160 NEXT YVEL
170 NEXT XVEL
```

Line 110 creates a sprite.

Lines 120 and 130 set values for the motion of the sprite.

Line 150 sets the sprite in motion.

Lines 160 and 170 complete the loops that set the values for the motion of the sprite.

MOUSE --Commands

MOUSE

The MYARC 9640 supports the industry standard MS mouse interface. Software within the operating system is used to position the mouse on the screen and detect mouse key depressions. The mouse itself is implemented as sprite 1 and therefore sprite 1 should not be used elsewhere in the program when using the mouse. In order to easily interface to these low level routines, MYARC Advanced BASIC implements a standard set of mouse commands. An example program is given in Appendix L illustrating the use of these commands.

MOUSE ON

Turns ON mouse interrupt. Mouse buttons are checked at the start of each BASIC statement.

If a mouse button is pressed, program execution is branched to an "ON MOUSE" subroutine or subprogram if the particular mouse key pressed was "armed".

MOUSE OFF

Turns OFF mouse interrupt checking.

MOUSE STOP

Delays action of the mouse button until MOUSE ON statement is encountered. The MOUSE ON interrupt is put on hold until a MOUSE ON command is later executed. Branching then takes place immediately if a mouse button was depressed.

ON MOUSE GOSUB

Line number 1, line number 2, or line number 3 is executed to start the event trapping. The program line number of a sub routine is executed when its corresponding button is pressed. Mouse button number 1 (the left most) corresponds to line number 1. Line number 2 corresponds to button 2, and line number 3 to button 3.

CALLS MKEY(button#,status)

Status monitors the status of mouse button #.

-1 button was pressed only once

0 button is not being or has not been pressed

1 button was pressed once since last mousep(n) call

CALL MKEYLAST(r,c)

Returns value of row and column during last MKEY(,).

CALL MLOC(dr,dc)

Returns the location of the row and column when a mouse button was last pressed.

CALL MREL(dr,dc)
Returns information of row and column when mouse button was released.

CALL MOUSE(Y,X)
Monitors status of mouse button X. If button X is pressed, Y equals 1.
If button X is not pressed, Y equals 0.

CALL MOUSEDRAG(ON)
Solid line showing motion between mouse presses.

CALL MOUSEDRAG(OFF)
Reverse mouse drag ON command.

CALL HIDEMOUSE
Eliminates mouse cursor.

CALL SEEMOUSE
Displays mouse cursor.

NEW

NEW

Format
NEW

Description

The NEW command erases the program currently in memory, so that you can enter a new program.

The NEW command restores the computer to the condition it was in when you selected MYARC Advanced BASIC from the main selection list with the following exceptions:

Memory allocated by the INIT subprogram is not returned to the memory area available to MYARC Advanced BASIC.

Assembly-language subprograms loaded by the LOAD subprogram remain in memory.

NEW restores all other default values, closes any open files, erases all variable values and names, and cancels any BREAK or TRACE commands in effect.

NEXT**NEXT**

Format
NEXT control-variable

Cross Reference
FOR TO

Description
The NEXT instruction marks the end of a FOR-NEXT loop.

You can use NEXT as either a program statement or a command.

The control-variable is the same control-variable that appears in the corresponding FOR TO instruction.

The NEXT instruction is always paired with a FOR TO instruction to form a FOR-NEXT loop (see FOR TO).

A NEXT statement cannot be part of an IF THEN statement.

If NEXT is used as a command, it must be part of a multiple-statement line. A FOR TO instruction must precede it on the same line.

Program

The following program illustrates a use of the NEXT statement in lines 130 and 140.

```

100 TOTAL=0
110 FOR COUNT=10 TO 0 STEP -2
120 TOTAL=TOTAL+COUNT
130 NEXT COUNT
140 FOR DELAY=1 TO 100::NEXT DELAY
150 PRINT TOTAL,COUNT;DELAY
RUN
30          -2 101

```

NUMBER**NUMBER**

Format

NUMBER [initial-line-number][,increment]
 NUM

Description

The NUMBER command puts the computer in Number Mode, so that it automatically generates line numbers for your program.

If you enter an initial-line-number, the first line number displayed is the one you specify. If you do not specify an initial-line-number, the computer starts with line number 100.

Succeeding line numbers are generated by adding the value of the numeric-expression increment to the previous line number. To specify increment only (without specifying an initial-line-number), you must precede the increment with a comma. The default increment is 10.

If a line number generated by the NUMBER command is the number of a line already in the program in memory, the existing program line is displayed with the line number. To indicate that the displayed line is an existing program line, the prompt symbol (>) that normally appears to the left of the line number is not displayed. When the computer displays an existing program line, you can either edit the line or press ENTER to leave the line unchanged.

If you enter a program line that contains an error, the appropriate error message is displayed, and the same line number appears again, enabling you to retype the line correctly.

If the next line number to be generated is greater than 32767, the computer leaves Number Mode.

To leave Number Mode, press ESC. If the computer is displaying only a line number (that is, a line number not followed by any characters), you can leave Number Mode by pressing ENTER, UP ARROW, DOWN ARROW.

Special Editing Keys in Number Mode

In Number Mode, you can use the editing keys whether you are changing existing program lines or entering new ones.

LEFT ARROW --Pressing LEFT ARROW moves the cursor one character position to the left. When the cursor moves over a character, it does not change or delete it.

RIGHT ARROW --Pressing RIGHT ARROW moves the cursor one character position to the right. When the cursor moves over a character, it does not change or delete it.

INS --Pressing INS enables you to insert characters at the cursor position. Characters that you type are inserted until you press one of the other special editing keys. The character at the cursor position and all characters to the right of the cursor move to the right as you type. You may lose characters if they move so far to the right that they are no longer in the program line.

DEL --Pressing DEL deletes the character in the cursor position. All characters to the right of the cursor move to the left.

ERASE (Ctrl C) --Pressing ERASE erases the program line currently displayed (including the line number). The program line is erased only from the screen, not from memory.

REDO (Alt + F8) --Pressing REDO causes the program line or other text most recently input to be displayed. This line can be especially helpful if you make an error while editing a program line, causing the computer not to accept it. Pressing REDO displays the original line so that you can make corrections without having to retype the entire line. When you press REDO, the computer leaves Number Mode and enters Edit Mode.

ESC (Ctrl +Break) --Pressing ESC causes the computer to leave Number Mode. If you were entering a new program line, it is not accepted. If you were changing an existing program line, any changes that you made are ignored.

ENTER --If you press ENTER when the computer is displaying only a line number (that is, a line number not followed by any characters), the computer leaves Number Mode. If the line number is the number of an existing program line, that program line is not changed or deleted.

If you press ENTER when the computer is displaying a line number followed by a program line, that line is accepted and the next line number is generated. The displayed line may be a new line that you have entered, an existing program line that you have not changed, or an existing program line that you have edited.

UP ARROW --UP ARROW works exactly the same as ENTER in Number Mode.

DOWN ARROW --DOWN ARROW works exactly the same as ENTER in Number Mode.

Example

In the following, what you type is UNDERLINED. Press ENTER after each line. NUM instructs the computer to number starting at 100 with increments of 10.

```

NUM
100 X=4
110 Z=10
120
NUM 110
110 Z=11
120 PRINT (Y+X)/Z
130
NUM 105,5
105 Y=7
110 Z=11
115
LIST
100 X=4
105 Y=7
110 Z=11
120 PRINT (X+Y)/Z

```

NUM 110 instructs the computer to number starting at 110 with increments of 10. Change line 110 to Z=11.

NUM 105,5 instructs the computer to number starting at line 105 with increments of 5. Line 110 already exists.

OLD

OLD

Format

OLD ["]file-specification["]

Cross Reference

SAVE

Description

The OLD command loads a program from an external storage device into memory.

The file-specification indicates the name of the program to be loaded from the external device. The file-specification, a string constant, can optionally be enclosed in quotation marks.

The program to be loaded can be one of the following:

A saved MYARC Advanced BASIC program.

A file in DISPLAY VARIABLE 80 format, created by the LIST command or a text editing or word processing program.

A specially prepared assembly-language program that executes automatically when it is loaded.

Before the program is loaded, all open files are closed. The program currently in memory is erased after the program begins to load. For more information see "Loading an Existing Program".

Protected and Unprotected Programs

To execute an unprotected MYARC Advanced BASIC program that has been loaded into memory, enter the RUN command when the cursor appears. You can use the LIST command to display the program or any portion of the program.

If the program was saved using the PROTECTED option of the SAVE command, it starts executing automatically when it is loaded. When the program ends (either normally or because of an error) or stops at a breakpoint, it is erased from memory.

Examples

OLD CSI

Displays instructions and then loads into the computer's memory a program from a cassette recorder.

OLD "DSK1.MYPROG"

Loads into the computer's memory the program MYPROG from diskette in disk drive one.

OLD DSK.DISK3.UPDATE85

Loads into the computer's memory the program UPDATE85 from the diskette named DISK3.

ON BREAK

ON BREAK

Format

ON BREAK STOP

ON BREAK NEXT

Cross Reference

BREAK

Description

The ON BREAK statement enables you to specify the action you want the computer to take when either a breakpoint is encountered or CLEAR is pressed.

If you enter the STOP option, or if your program does not include an ON BREAK statement, program execution stops when a breakpoint is encountered or CLEAR is pressed.

If you enter the NEXT option, program execution continues normally (with the next program statement) when a breakpoint is encountered or CLEAR is pressed. If you press CLEAR while the computer is performing an input or an output operation with certain external devices, an error condition occurs, causing the program to halt. When the NEXT option is in effect, pressing CTL-ALT-DEL is the only way to interrupt your program. However, by doing so, you perform a "reboot" of the system therefore erasing the program in memory and causing you to exit from MYARC Advanced BASIC without closing any open files, possibly causing the loss of data in those files.

ON BREAK does not affect a breakpoint that occurs when a BREAK statement with no line-number-list is encountered in a program.

Program

The following program illustrates the use of ON BREAK.

```
100 CALL CLEAR
110 BREAK 150
120 ON BREAK NEXT
130 BREAK
140 FOR A=1 TO 50
150 PRINT "CLEAR IS DISABLED."
160 NEXT A
170 ON BREAK STOP
180 FOR A=1 TO 50
190 PRINT "NOW IT WORKS."
200 NEXT A
```

Line 110 sets a breakpoint at line 150.

Line 120 sets breakpoint handling to go to the next line.

A breakpoint occurs at line 130 despite line 120, because no line number has been specified after BREAK. Enter CONTINUE.

No breakpoint occurs at line 150 because of line 120; CLEAR has no effect during the execution of lines 140 through 160 because of line 120. Line 170 restores the normal use of CLEAR.

ON ERROR**ON ERROR**

Format

ON ERROR STOP

ON ERROR line-number

Cross Reference

ERR, GOSUB, RETURN

Description

The ON ERROR statement enables you to specify the action you want the computer to take if a program error occurs.

If you enter the STOP option, or if your program does not include an ON ERROR statement, program execution stops when a program error occurs.

If you enter a line-number, a program error causes program control to be transferred to the subroutine that begins at the specified line-number. A RETURN statement in the subroutine returns control to a specified program statement.

When an error transfers control to a subroutine, the line-number option is cancelled. If you wish to restore it, your program must execute an ON ERROR line-number statement again.

The ON ERROR line-number statement does not transfer control when the error is caused by a RUN statement.

Program

The following program illustrates a use of ON ERROR.

```

100 CALL CLEAR
110 DATA "A","4","B","C"
120 ON ERROR 190
130 FOR G=1 TO 4
140 READ X$
150 X=VAL(X$)
160 PRINT X;"SQUARED IS";X*X
170 NEXT G
180 STOP
190 REM ERROR SUBROUTINE
200 ON ERROR 230
210 X$="5"
220 RETURN
230 REM SECOND ERROR
240 CALL ERR(CODE,TYPE,SEVER,LINE)
250 PRINT "ERROR";CODE;" IN LINE";LINE
260 RETURN 170

```

Line 120 causes any error to pass control to line 190.

Line 130 begins a loop. An error occurs in line 150 and control passes to line 190.

Line 200 causes the next error to pass control to line 230.

Line 210 changes the value of X\$ to an acceptable value. Line 220 returns control to the line in which the error occurred (line 150).

The second time an error occurs, the SECOND ERROR subroutine is called because of line 200. Line 240 obtains specific information about the error by using CALL ERR. Line 250 reports the nature of the error, and line 260 returns control to line 170 of the main program, which begins the next iteration of the loop.

When the third error occurs, the message Bad Argument in 150 is displayed because the program does not specify what action to take if another error occurs. Program execution ceases.

ON GOSUB**ON GOSUB**

Format

ON numeric-expression;GOSUB; line-number-list
GOSUB

Cross Reference

GOSUB, RETURN

Description

The ON GOSUB statement enables you to transfer conditional program control to one of several subroutines.

The value of the numeric-expression determines to which of the line numbers in the line-number-list program control is transferred.

If the value of the numeric-expression is 1, program control is transferred to the subroutine that begins at the program statement specified by the first line number in the line-number-list; if the value of the numeric-expression is 2, program control is transferred to the subroutine that begins at the program statement specified by the second line number in the line-number-list; and so on.

If necessary, the value of the numeric-expression is rounded to the nearest integer. The value of the numeric-expression must be greater than or equal to 1 and less than or equal to the number of line numbers in the line-number-list.

The line-number-list consists of one or more line numbers separated by commas. Each line number specifies a program statement at which a subroutine begins.

Use a RETURN statement to return program control to the statement immediately following the ON GOSUB statement that called the subroutine.

To avoid unexpected results, it is recommended that you exercise special care if you use ON GOSUB to transfer control to or from a subprogram or into a FOR-NEXT loop.

Examples

```
100 ON X GOSUB 1000,2000,300
```

Transfers control to 1000 if X is 1, 2000 if X is 2, and 300 if X is 3.

```
100 ON P-4 GOSUB 200,250,300,800,170
```

Transfers control to 200 if P-4 is 1 (P is 5), 250 if P-4 is 2, 300 if P-4 is 3, 800 if P-4 is 4, and 170 if P-4 is 5.

Program

The following program illustrates a use of ON GOSUB.

```
100 CALL CLEAR
110 DISPLAY AT(11,1):"CHOOSE ONE OF THE FOLLOWING:"
120 DISPLAY AT(13,1):"1 ADD TWO NUMBERS."
130 DISPLAY AT(14,1):"2 MULTIPLY TWO NUMBERS."
140 DISPLAY AT(15,1):"3 SUBTRACT TWO NUMBERS."
150 DISPLAY AT(16,1):"4 EXIT PROGRAM."
160 DISPLAY AT(20,1):"YOUR CHOICE:"
170 DISPLAY AT(22,2):"FIRST NUMBER."
180 DISPLAY AT(23,1):"SECOND NUMBER."
190 CALL MARGIN(3,30,1,24)
200 ACCEPT AT(20,14)VALIDATE(DIGIT):CHOICE
210 IF CHOICE<1 OR CHOICE>4 THEN 200
220 IF CHOICE=4 THEN STOP
230 ACCEPT AT(22,16)VALIDATE(NUMERIC):FIRST
240 ACCEPT AT(23,16)VALIDATE(NUMERIC):SECOND
250 CALL MARGIN(3,30,1,8)
260 ON CHOICE GOSUB 280,300,320
270 GOTO 190
280 DISPLAY AT(3,1)ERASE ALL:FIRST;"PLUS";SECOND;"EQUALS";FIRST+SECOND
290 RETURN
300 DISPLAY AT(3,1)ERASE ALL:FIRST;"TIMES";SECOND;"EQUALS";FIRST*SECOND
310 RETURN
320 DISPLAY AT(3,1)ERASE ALL:FIRST;"MINUS";SECOND;"EQUALS";FIRST-SECOND
330 RETURN
```

Line 260 determines where to go according to the value of CHOICE.

ON GOTO**ON GOTO**

Format

```
ON numeric-expression GOTO line-number-list
GOTO
```

Cross Reference

GOTO

Description

The ON GOTO statement enables you to transfer unconditional program control to one of several program statements.

The value of the numeric-expression determines to which of the line numbers in the line-number-list program control is transferred. If the value of the numeric-expression is 1, program control is transferred to the program statement specified by the first line number in the line-number-list; if the value of the numeric-expression is 2, program control is transferred to the program statement specified by the second line number in the line-number-list; and so on.

If necessary, the value of the numeric-expression is rounded to the nearest integer. The value of the numeric-expression must be greater than or equal to 1 and less than or equal to the number of line numbers in the line-number-list.

The line-number-list consists of one or more line numbers separated by commas. Each line number specifies a program statement.

To avoid unexpected results, it is recommended that you exercise care if you use ON GOTO to transfer control to or from a subroutine or a subprogram or into a FOR-NEXT loop.

Examples

```
100 ON X GOTO 1000,2000,300
```

Transfers control to 1000 if X is 1, 2000 if X is 2, and 300 if X is 3. The equivalent statement using an IF-THEN-ELSE statement is IF X=1 THEN 1000 ELSE IF X=2 THEN 2000 ELSE IF X=3 THEN 300 ELSE PRINT "ERROR!":STOP.

```
100 ON P-4 GOTO 200,250,300,800,170
```

Transfers control to 200 if P-4 is 1 (P is 5), 250 if P-4 is 2, 300 if P-4 is 3, 800 if P-4 is 4, and 170 if P-4 is 5.

Program

The following program illustrates a use of ON GOTO. Line 260 determines where to go according to the value of CHOICE.

```
100 CALL CLEAR
110 DISPLAY AT(11,1):"CHOOSE ONE OF THE FOLLOWING:"
120 DISPLAY AT(13,1):"1 ADD TWO NUMBERS."
130 DISPLAY AT(14,1):"2 MULTIPLY TWO NUMBERS."
140 DISPLAY AT(15,1):"3 SUBTRACT TWO NUMBERS."
150 DISPLAY AT(16,1):"4 EXIT PROGRAM."
160 DISPLAY AT(20,1):"YOUR CHOICE:"
170 DISPLAY AT(22,2):"FIRST NUMBER:"
180 DISPLAY AT(23,1):"SECOND NUMBER:"
190 CALL MARGIN(3,30,1,24)
200 ACCEPT AT(20,14)VALIDATE(DIGIT):CHOICE
210 IF CHOICE<1 OR CHOICE>4 THEN 200
220 IF CHOICE=4 THEN STOP
230 ACCEPT AT(22,16)VALIDATE(NUMERIC):FIRST
240 ACCEPT AT(23,16)VALIDATE(NUMERIC):SECOND
250 CALL MARGIN(3,30,1,8)
260 ON CHOICE GOTO 270,290,310
270 DISPLAY AT(3,1)ERASE ALL:FIRST;"PLUS";SECOND;"EQUALS";FIRST+SECOND
280 GOTO 190
290 DISPLAY AT(3,1)ERASE ALL:FIRST;"TIMES";SECOND;"EQUALS";FIRST*SECOND
300 GOTO 190
310 DISPLAY AT(3,1)ERASE ALL:FIRST;"MINUS";SECOND;"EQUALS";FIRST-SECOND
320 GOTO 190
```

ON WARNING**ON WARNING**

Format

```
ON WARNING PRINT
                STOP
                NEXT
```

Description

The ON WARNING statement enables you to specify the action you want the computer to take if a warning condition occurs during the execution of your program.

A warning, a condition caused by invalid input or output, does not normally cause program execution to be terminated.

If you enter the PRINT option, or if your program does not include an ON WARNING statement, the computer displays a warning message when a warning condition occurs during program execution.

If you enter the STOP option, program execution stops when a warning condition occurs during program execution.

If you enter the NEXT option, program execution continues normally when a warning condition occurs and no warning message is displayed. Normally, execution continues beginning with the next program statement; however, if the cause of the warning is an invalid response to an INPUT statement, program execution continues beginning with that same INPUT statement.

You may have multiple ON WARNING statements in the same program.

Program

The following program illustrates a use of ON WARNING.

```
100 CALL CLEAR
110 ON WARNING NEXT
120 PRINT 120,5/0
130 ON WARNING PRINT
140 PRINT 140,5/0
```



```
150 ON WARNING STOP
160 PRINT 160,5/0
170 PRINT 170
RUN
120          9.99999E+**
140
* WARNING
  NUMERIC OVERFLOW IN 140
    9.99999E+**
160
* WARNING
  NUMERIC OVERFLOW IN 160
```

Line 110 sets warning handling to go to the next line. Line 120 therefore prints the result without any message.

Line 130 sets warning handling to the default, printing the message and then continuing execution. Line 140 therefore prints 140, then the warning, and then continues.

Line 150 sets warning handling to print the warning message and then stop execution. Line 160 therefore prints 160 and the warning message and then stops.

OPEN

OPEN

Format

```
OPEN #file-number:file-specification[ file-organization[ size]]
[,file-type][,open-mode][,record-type[ record-length]]
```

Cross Reference

CLOSE, INPUT, PRINT

Description

The OPEN instruction establishes an association between the computer and an external device, enabling you to store, retrieve, and process data.

The file-number is a numeric-expression having a value between 1 and 255. The file-number is assigned to the external file or device indicated by the file-specification so that input/output processing instructions may refer to the file by its file-number. While a file is open, its file-number cannot be assigned to another file. However, you may have more than one file open to a device at one time. File-number 0 always refers to the keyboard and screen of your computer, and is always open. You cannot open or close file-number 0.

If necessary, the file-number is rounded to the nearest integer.

The file-specification is a string-expression; if you use a string constant, you must enclose it in quotation marks.

Options

The following options may be entered in any order.

The file-organization specifies whether records are to be accessed sequentially or randomly. Enter SEQUENTIAL for sequential access, or RELATIVE for random access. Records in a sequential-access file are read or written in sequence from beginning to end. Records in a random-access (relative-record) file can be accessed in any order (they can be processed randomly or sequentially.)

If you do not specify a file-organization, it is assumed to be SEQUENTIAL.

You can optionally specify the initial size of the file. Size is a numeric-expression, the value of which specifies the initial number of records in the file. Note: The size option cannot be used with all peripherals.

The file-type specifies the format of data in the file.

INTERNAL--The computer transfers data in binary format. This is the most efficient method of sending data.

DISPLAY--The computer transfers data in ASCII format. DISPLAY

files can only use FIXED records of 64 or 128. If no file-type is specified in OPEN, the default is DISPLAY.

DISPLAY type files require a special kind of output record. Each element in the PRINT field must be separated by a comma enclosed in quotation marks. The comma serves as a field separator in the file. The omission of this comma causes an I/O error. Note: This is not the same as a print separator, which must be inserted between an element in the PRINT field and the field separator.

The open-mode specifies the input/output operations that can be performed on the file.

INPUT--The computer can only read data from the file.

OUTPUT--The computer can only write data to the file.

UPDATE--The computer can both read from and write to the file.

APPEND--The computer can only write data and only at the end of the file; records already in the file cannot be accessed.

If you open an existing file for OUTPUT, the data items you write to the file replace those currently in the file.

If you do not specify an open-mode, it is assumed to be an UPDATE.

The record-type specifies whether the records in the file are FIXED (all of the same length) or VARIABLE (of various lengths).

SEQUENTIAL files can have FIXED or VARIABLE records. If you do not specify the record-type of a SEQUENTIAL file, it is assumed to be VARIABLE.

RELATIVE files must have FIXED records. If you do not specify the record-type of a RELATIVE file, it is assumed to be FIXED.

You can optionally specify the length of records in the file. Record-length is a numeric-expression, the value of which specifies the fixed size (for FIXED records) or maximum size (for VARIABLE records) of each record.

If you do not specify a record-length, its value is supplied by the peripheral.

If you open a file that does not exist, a file is created with the options you specify. If you open a file that does exist, the options you specify must be the same as the options that you specified when you created the file, except that a file with FIXED records can be opened as either SEQUENTIAL or RELATIVE, regardless of the file-organization that you specified when you created the file.

For more information about the options available with a particular device, refer to the owner's manual that comes with that device.

Examples

```
100 OPEN #1:"CS1",OUTPUT,FIXED
```

Opens a file on cassette. The file is SEQUENTIAL, with data stored in DISPLAY format. The file is opened in OUTPUT mode with FIXED length records of 64 bytes.

```
300 OPEN #23:"DSK.MYDISK.X",RELATIVE 100,INTERNAL,FIXED,UPDATE
```

Opens a file named "X". The file is on the diskette named MYDISK in whichever drive that diskette is located. The file is RELATIVE, with data kept in INTERNAL format with FIXED length records of 80 bytes. The file is opened in UPDATE mode and starts with 100 records made available for it.

```
100 OPEN #234:A$,INTERNAL
```

Where A\$ equals "DSK2.ABC", assumes a file on the diskette in drive 2 with a name of ABC. The file is SEQUENTIAL, with data kept in INTERNAL format. The file is opened in UPDATE mode with VARIABLE length records that have a maximum length of 80 bytes.

Program

The following program illustrates a use of the SIZE option in an OPEN statement.

```
100 OPEN #1:"DSK1.LARGE",RELATIVE
110 PRINT #1,REC 100:0
120 CLOSE #1
130 OPEN #1:"DSK1.LARGE",SEQUENTIAL,FIXED
200 CLOSE #1
```

Line 100 opens a RELATIVE file on diskette.

Line 110 writes to the 100th record, thereby reserving space for 100 contiguous records.

Line 120 closes the file.

Line 130 reopens the file, this time with SEQUENTIAL file organization.

Line 200 closes the file.

OPTION BASE**OPTION BASE**

Format

```
OPTION BASE 0
           1
```

Cross Reference

DIM

Description

The **OPTION BASE** statement enables you to set the lower limit of array subscripts.

You can use the **OPTION BASE** statement to specify a lower array-subscript limit of either 0 or 1. If your program does not include an **OPTION BASE** statement, the lower limit is set to 0.

The **OPTION BASE** statement applies to every array in your program. You can have only one **OPTION BASE** statement in a program.

If you do not set the lower array-subscript limit to 1, the computer reserves memory for element 0 of each dimension of each array. To avoid reserving unnecessary memory, it is recommended that you set the lower limit to 1 if your program does not use element 0.

The **OPTION BASE** statement must have a lower line number than any **DIM** statement or any reference to an array in your program. The **OPTION BASE** statement is evaluated during pre-scan and is not executed.

The **OPTION BASE** statement cannot be part of an **IF THEN** statement.

Example

```
100 OPTION BASE 1
```

Sets the lowest allowable subscript of all arrays to one.

OUT

CALL OUT

Format

CALL OUT(port,databyte[,databyte...])

Cross Reference

INP, INPUT\$

Description

You may use CALL OUT either as a program statement or a command.

Sends a databyte to a port.

Port may be an integer from 1 to 65535.

The dataByte may be any integer between 0 and 255.

Data is received and sent internally through various components within the computer, known as ports.

The OUT statement is used to obtain direct control of a device such as the keyboard, sound, etc.

OUT is the complement function to the INP command.

PATTERN --Subprogram

PROGRAM

Format

CALL PATTERN(#sprite-number,character-code[...])

Cross Reference

CHAR, MAGNIFY, SPRITE

Description

The PATTERN subprogram enables you to change the pattern of one or more sprites.

The sprite-number is a numeric-expression whose value specifies the number of the sprite as assigned in the SPRITE subprogram.

Character-code is a numeric-expression with a value from 0-255, specifying the character number of the character you want to use as the pattern for a sprite.

If you use the MAGNIFY subprogram to change to double-sized sprites, the sprite definition includes the character specified by the character-code and three additional characters (see MAGNIFY.)

Program

The following program illustrates a use of the Pattern subprogram.

```

100 CALL CLEAR
110 CALL COLOR(12,16,16)
120 FOR A=19 TO 24
130 CALL HCHAR(A,1,120,32)
140 NEXT A
150 A$="01071821214141FFFF4141212119070080E09884848282FFFF8282848498E000"
160 B$="01061820305C4681814246242C180700806018342462428181623A0C0418E000"
170 C$="0106182C2446428181465C3020180700806018040C3A6281814262243418E000"
180 CALL CHAR(244,A$,248,B$,252,C$)
190 CALL SPRITE(#1,244,5,130,1,0,8)
200 CALL MAGNIFY(3)
210 FOR A=244 TO 252 STEP 4
220 CALL PATTERN(#1,A)
230 FOR DELAY=1 TO 15 :: NEXT DELAY
240 NEXT A
250 GOTO 210
(Press CLEAR to stop the program.)

```

Lines 110 through 140 build a floor.

Lines 150 through 180 define characters 244 through 255.

Line 190 creates a sprite in the shape of a wheel and starts it moving to the right.

Line 200 makes the sprite double-sized.

Lines 210 through 250 make the spokes of the wheel appear to move as the character displayed is changed.

PEEK --Subprogram--Peek at CPU RAM

PEEK

Format

```
CALL PEEK(address,numeric-variable-list[,"",address,
numeric-variable-list[,...]])
```

Cross Reference

LOAD, PEEKV, POKEV, VALHEX

Description

The PEEK subprogram enables you to ascertain the contents of specified CPU memory addresses.

You can use the PEEKV subprogram to ascertain the contents of VDP memory.

The address is a numeric-expression whose value specifies the first CPU (Central Processing Unit) memory address at which you want to peek.

The address must have a value from -32768 to 32767 inclusive.

You can specify an address from 0 to 32767 inclusive by specifying the actual address.

You can specify an address from 32768 to 65535 inclusive by subtracting 65536 from the actual address. This will result in a value from -32768 to -1 inclusive.

If you know the hexadecimal value of the address, you can use the VALHEX function to convert it to a decimal numeric-expression, eliminating the need for manual calculations.

If necessary, the address is rounded to the nearest integer.

The numeric-variable-list consists of one or more numeric-variables separated by commas. Bytes of data starting from the specified CPU memory address are assigned sequentially to the numeric-variables in the numeric-variable-list.

One byte, with a value from 0 to 255 inclusive, is returned to each specified numeric-variable.

You can specify multiple addresses and numeric-variable-lists by entering a null string (two adjacent quotation marks) as a separator between a numeric-variable-list and the next address.

If you call the PEEK subprogram with invalid parameters, the computer may function erratically or cease to function entirely. If this occurs, turn off the computer, wait several seconds, and then turn the computer back on again.

Examples

```
100 CALL PEEK(8192,X1,X2,X3,X4)
```

Returns the values in memory locations 8192, 8193, 8194, and 8195 in the variables X1, X2, X3, and X4, respectively.

```
100 CALL PEEK(22433,A,B,C,"",-4276,X,Y,Z)
```

Returns the values in locations 22433, 22434, and 22435 in A, B, C, respectively; and the values in locations 61260, 61261, and 61263 in X, Y, and Z, respectively.

```
100 CALL PEEK(VALHEX("4F55"),V1,V2,V3)
```

Uses VALHEX to ascertain the decimal equivalent of the hexadecimal number 4F55, which is 20309. Then the values in locations 20309, 20310, and 20311 are returned in V1, V2, and V3, respectively.

Program

The following program returns in A the number of the highest numbered sprite (#15) currently in use. A zero is returned to B, because no sprites are defined after the DELSPRITE statement.

```
100 CALL CLEAR
110 CALL SPRITE(#15,33,7,100,100,0,0)
120 CALL PEEK(VALHEX("837A"),A)
130 CALL DELSPRITE(ALL)
140 CALL PEEK(VALHEX("837A"),B)
150 PRINT A,B
```

PEEKV --Subprogram--Peek at VDP RAM

PEEKV

Format

```
CALL PEEKV(address,numeric-variable-list[,"",address,
numeric-variable-list[,...]])
```

Cross Reference

LOAD, PEEK, POKEV, VALHEX

Description

The PEEKV subprogram enables you to ascertain the contents of specified VDP memory addresses. You can use the PEEK subprogram to ascertain the contents of CPU memory.

The address is a numeric-expression whose value specifies the first VDP (Video Display Processor) memory address at which you want to peek.

The address must have a value from 0 to 16383 inclusive.

If you know the hexadecimal value of the address (0000-3FFF), you can use the VALHEX function to convert it to a decimal numeric-expression.

If necessary, the address is rounded to the nearest integer.

The numeric-variable-list consists of one or more numeric-variables separated by commas. Bytes of data starting from the specified VDP memory address are assigned sequentially to the numeric-variables in the numeric-variable-list.

One byte, with a value from 0 to 255 inclusive, is returned to each specified numeric-variable.

You can specify multiple addresses and numeric-variable-lists by entering a null string (two adjacent quotation marks) as a separator between a numeric-variable-list and the next address.

If you call the PEEKV subprogram with invalid parameters, the computer may function erratically. If this occurs, turn off the computer, wait several seconds, then turn the computer back on.

Example

```
100 CALL PEEKV(6300,A1,A2,A3)
```

Returns the values in locations 6300, 6301, and 6302 in A1, A2, and A3, respectively.

Programs

The following program illustrates a use of the PEEKV subprogram.

```
100 CALL CLEAR
110 CALL POKEV(32* 16+12,66)
120 CALL PEEKV(32* 16+12,A)
130 PRINT A
```

Line 110 pokes a "B" into a location that causes it to appear in the middle of the screen. Line 120 peeks at that location, and assigns the value found there (66) to the variable A.

The next program starts a sprite moving diagonally across the screen. Line 120 assigns the values of the row and column coordinates of the sprite to Y and X, respectively.

```
100 CALL CLEAR
110 CALL SPRITE(#1,33,5,100,100,25,25)
120 CALL PEEKV(VALHEX("300"),X,Y)
130 DISPLAY AT(24,1):Y;X
140 GOTO 120
(Press CLEAR to stop the program.)
```

PI --Function--Pi**PI**

Format
PI

Type
REAL

Description

The PI function returns the value of pi.

The value of pi is 3.14159265359.

Example

```
100 VOLUME=4/3*PI*6^3
```

Sets VOLUME equal to four-thirds times pi times six cubed, which is the volume of a sphere with a radius of six.

POINT --Subprogram**POINT****Format**

CALL POINT(pixel-type,pixel-row,pixel-column[,pixel-row,pixel-column2[,...]])

Cross Reference

CIRCLE, DCOLOR, DRAW, DRAWTO, FILL, GCHAR, GRAPHICS, RECTANGLE

Description

The POINT subprogram enables you to place, or erase specific points (pixels) on the screen, one or more at a time.

Pixel-type is a numeric-expression whose value specifies the action taken by the POINT subprogram.

TYPE	ACTION
2	Reverses the status of the specified point (pixel). (If a pixel is on, it is turned off; if a pixel is off, it is turned on). This effectively reverses the color of the specified pixel.
1	Places a point, of the foreground-color specified by the DCOLOR subprogram, at a specified pixel-row and pixel-column. This is accomplished by turning on the pixel at the designated row and column.
0	Erases a point at a specified pixel-row and pixel-column. This is accomplished by turning on the pixel at the designated row and column.

Pixel-row and pixel-column are numeric-expressions whose values represent the screen position where the point will be placed (turned on or off).

You can optionally place more points by specifying additional sets of pixels.

Pixel-row and pixel-column must be within the range of the particular graphics mode of the screen.

The last pixel-row/pixel-column you specify becomes the current position used by the DRAWTO subprogram.

POINT cannot be used in Pattern or Text Modes.

Example

```
100 CALL POINT(1,96,128)
```

Turns on a single pixel in the center of the screen.

POKEV --Subprogram--Poke to VDP RAM

POKEV

Format

CALL POKEV(address,byte-list[,"",address,byte-list[,...]])

Cross Reference

LOAD, PEEK, PEEKV, VALHEX

Description

The POKEV subprogram enables you to assign values directly to specified VDP memory addresses.

You can use the LOAD subprogram to assign values to CPU.

The address is a numeric-expression whose value specifies the first VDP (Video Display Processor) memory address where data is to be poked. If the byte-list specifies more than one byte of data, the bytes are assigned to sequential memory addresses starting with the address you specify.

The address must have a value from 0 to 16383 inclusive.

If you know the hexadecimal value of the address (0000-3FFF), you can use the VALHEX function to convert it to a decimal numeric-expression.

If necessary, the address is rounded to the nearest integer.

The byte-list consists of one or more bytes of data, separated by commas, that are to be poked into VDP memory starting with the specified address.

Each byte in the byte-list must be a numeric-expression with a value from 0 to 32767. If the value of a byte is greater than 255, it is repeatedly reduced by 256 until it is less than 256. If necessary, a byte is rounded to the nearest integer.

You can specify multiple addresses and byte-lists by entering a null string (two adjacent quotation marks) as a separator between a byte-list and the next address.

If you call the POKEV subprogram with invalid parameters the computer may function erratically. If this occurs, turn off the computer, wait several seconds, then turn the computer back on.

Examples

```
100 CALL POKEV(3333,233)
```

Pokes the value 233 into location 3333.

```
100 CALL POKEV(13784,273)
```

Pokes the value 17 (273 reduced by 256 once) into location 13784.

100 CALL POKEV(7343,246,"",VALHEX("2E4F"),433)
Pokes the value 246 into location 7343, and uses VALHEX to ascertain the decimal equivalent of the hexadecimal number 2E4F (11855). The value 177 (433 reduced by 256 once) is then poked into this location.

Program

The following program uses POKEV to display on the screen the characters that correspond to ASCII codes 65 through 208, at the location specified by the value of R* 32+C.

```
100 CALL CLEAR::X=65
110 FOR R=0 TO 23
120 FOR C=0 TO 31 STEP 6
130 CALL POKEV(R* 32+C,X)
140 X=X+1
150 NEXT C
160 NEXT R
```

POS --Function--Position

POS

Format

POS(string-expression,substring,numeric-expression)

Type

DEFINT

Description

The POS function returns the position of the first occurrence of a substring within a specified string.

The string-expression specifies the string within which you are seeking the substring. If you use a string constant, it must be enclosed in quotation marks.

The substring is the segment (of the string-expression) you are trying to locate. The substring is a string-expression; if you use a string constant, it must be enclosed in quotation marks.

The value of the numeric-expression specifies the character position in the string-expression where the search for the substring begins.

If necessary, the value of the numeric-expression is rounded to the nearest integer.

If the substring is present within the string-expression, POS returns the number of the character position (within the string-expression) of the first character of the substring.

If the substring is not present, or if the value of the numeric-expression is greater than the number of characters in the string-expression, POS returns a zero.

Examples

```
100 X=POS("PAN","A",1)
```

Sets X equal to 2 because A is the second letter in PAN.

```
100 Y=POS("APAN","A",2)
```

Sets Y=3 because the A in the third position in APAN is the first occurrence of A in the portion of APAN that was searched.

```
100 Z=POS("PAN","A",3)
```

Sets Z equal to 0 because A was not in the part of PAN that was searched.

```
100 R=POS("PABNAN","AN",1)
```

Sets R equal to 5 because the first occurrence of AN starts with the A in the fifth position in PABNAN.

Program

The following program illustrates a use of POS. Input is searched for spaces, and is then printed with each word on a single line.

```
100 CALL CLEAR
110 PRINT "ENTER A SENTENCE."
120 LINPUT X$
130 S=POS(X$," ",1)
140 IF S=0 THEN PRINT X$::PRINT::GOTO 110
150 Y$=SEG$(X$,1,S)::PRINT Y$
160 X$=SEG$(X$,S+1,LEN(X$))
170 GOTO 130
(Press CLEAR to stop the program.)
```


POSITION --Subprogram**POSITION**

Format

```
CALL POSITION(#sprite-number,numeric-variable1,numeric-variable2[,...])
```

Cross Reference

SPRITE

Description

The POSITION subprogram enables you to ascertain the current position of one or more sprites.

The sprite-number is a numeric-expression whose value specifies the number of the sprite as assigned in the SPRITE subprogram.

The current screen position of a sprite is returned as two numeric-variables representing the pixel-row and pixel-column, respectively, specifying the position of a screen pixel.

The screen position of the pixel in the upper-left corner of a sprite is considered to be the position of that sprite.

Note that a sprite in motion continues to move during and following the execution of the POSITION subprogram. Remember to allow for this continued motion in your program.

Example

```
100 CALL POSITION(#1,Y,X)
```

Returns the position of the upper left corner of sprite #1. Also see the third example of the SPRITE subprogram.

PPR

PPR

Description

Prints current printer device-name to screen or current output device.

PRINT

PRINT

Format

Print to the Screen

```
PRINT [print-list]
```

Print to a File (or Device)

```
PRINT #file-number[,REC record-number][:print-list]
```

Cross Reference

DISPLAY, OPEN, PRINT USING, TAB

Description

The PRINT instruction enables you to display data items on the screen or print them to an external device. You can use PRINT as either a program statement or a command.

The print-list consists of one or more print items (items to be printed or displayed) separated by print separators. A PRINT instruction without a print-list advances the print position to the first position of the next record. This has the effect of printing a blank record, unless the preceding PRINT instruction ended with a print-separator.

The numeric- and/or string-expressions in the print-list can be constants and/or variables.

Print items are the numeric- and string-expressions to be printed. Any function is also a valid print item.

Print separators are the punctuation (commas, semicolons, and colons) between print items specifying the placement of the print items in the print record.

Printing to the Screen

Each print item is displayed in the row of the screen window defined by the margins, starting from the far left column of the window. Before a new line is displayed at the bottom of the window, the entire contents of the window (excluding sprites) scroll up one line to make room for the new line. The contents of the top line of the window scroll off the screen and are discarded.

Each line on the screen is treated as one print record. The record length of the screen is the width of the window.

Printing to a File

If you include an optional file-number, the print-list is sent to the specified device. The file-number is a numeric-expression whose value specifies the number of the file as assigned in its OPEN instruction. You cannot print to a file opened in INPUT mode.

If you do not specify a file-number (or if you specify file-number 0), the print-list is displayed on the screen.

If you use the REC option, the record-number is a numeric-expression whose value specifies the number of the record in which you want to print the print-list. The records in a file are numbered sequentially, starting with zero. The REC option can be used only with a file opened for RELATIVE access.

If you print to a file opened in INTERNAL format with FIXED records, each record is filled with trailing binary zeros, if necessary, to bring it to its specified length. If a record is longer than the record length of the file, it is truncated (extra characters are discarded).

For more information about printing to a particular device, refer to the owner's manual that comes with that device.

Printing Numbers: INTERNAL Files

The amount of memory space allocated to a number printed to a file opened in INTERNAL format varies according to its data-type. A DEFINT is always allocated 3 bytes, whereas a REAL number is always allocated 9 bytes.

Note that if you print a DEFINT value to a file, you cannot access that file on a Home Computer that does not support the INTEGER data-type. You can circumvent this by converting all DEFINT variables and functions to REAL variables before printing them to a file.

Printing Numbers: The Screen and DISPLAY Files

The format of a number printed to the screen or to a file opened in DISPLAY format varies according to the characteristics of the number.

Positive numbers and zero are printed with a leading space (instead of a plus sign); negative numbers are printed with a leading minus sign. All numbers are printed with a trailing space.

Numbers are printed in either decimal form or scientific notation, according to these rules:

All numbers with 10 or fewer digits are printed in decimal form.

REAL numbers with more than 10 digits are printed in scientific notation only if they can be presented with more significant digits in scientific notation than in decimal form. If printed in decimal form, all digits beyond the tenth are omitted.

If a number is printed in decimal form, the following rules apply:

DEFINT numbers and REAL numbers with no decimal portion are printed without decimal points.

REAL numbers are printed with decimal points in the proper position. If the number has more than 10 digits, it is rounded to 10 digits. A zero is not printed by itself to the left of the decimal point. Trailing zeros after the decimal point are omitted.

If number is printed in scientific notation, the following rules apply:

The format is mantissaEexponent.

The mantissa is printed with six or fewer digits, with one digit to the left of the decimal point.

Trailing zeros are omitted after the decimal point of the mantissa.

If there are more than five digits after the decimal point of the mantissa, the fifth digit is rounded.

The exponent is a two-digit number displayed with a plus or minus sign.

If you attempt to print a number with an exponent greater than 99 or less than -99, the computer prints two asterisks (**) following the sign of the exponent.

Printing Strings

A string constant in a print-list must be enclosed in quotation marks. A quotation mark within a string constant is represented by two adjacent quotation marks.

A string printed to a file opened in INTERNAL format has a length one greater than the length of the string.

When a string is printed to the screen or to a file opened in DISPLAY format, no leading or trailing spaces are added to the string.

Print Separators

At least one print separator must be placed between adjacent print items in the print-list. Valid print separators are the semicolon (;), the colon (:), and the comma (,).

A semicolon (;) print separator causes the next print item to print immediately after the current print item.

A colon (:) print separator causes the next print item to print at the beginning of the next record. Consecutive colons used as print separators must be divided by a space. Otherwise, they are treated as a statement separator symbol.

If you print to the screen or to a file opened in DISPLAY format, a comma (,) print separator causes the next print item to print at the beginning of the next "zone." Print records are divided into 14-character zones; the number of zones in a print record varies according to its record length.

If you print to a file opened in INTERNAL format, a comma print separator has the same effect as a semicolon print separator.

If a print separator would have the effect of splitting the next print item between two records, the print item is moved to the beginning of the following record. However, if discarding the trailing space from a numeric print item allows it to fit in the current record, the number is printed in the current record without its trailing space.

If the print-list ends with a print separator, the computer is placed in a print-pending condition. Unless the next PRINT instruction includes the REC option, it is considered to be a continuation of the current PRINT instruction. RESTORE #file-number terminates a print-pending condition.

If the print-list is not terminated by a print separator, the computer considers the current record complete when all the print items in the print-list are printed. The first print-item in the next PRINT instruction begins in the next record.

Examples

```
100 PRINT
```

Causes a blank line to appear on the display screen.

```
100 PRINT "THE ANSWER IS";A
```

Causes the string constant THE ANSWER IS to be printed on the display screen, followed immediately by the value of ANSWER. If ANSWER is positive, there will be a blank for the positive sign after IS.

```
100 PRINT X:Y/2
```

Causes the value of X to be printed on a line and the value of Y/2 to be printed on the next line.

```
100 PRINT #12,REC 7:A
```

Causes the value of A to be printed on the eighth record of the file that was opened as number 12 with RELATIVE file organization. (Record number 0 is the first record.)

```
100 PRINT #32:A,B,C,
```

Causes the values of A, B, and C to be printed on the next record of the file that was opened as number 32. The final comma creates a pending print-condition. The next PRINT statement directed to file number 32 will print on the same record as this PRINT statement unless it specifies a record, or a RESTORE #32 statement is executed, thereby closing the print-pending print condition.

```
100 PRINT #1,REC 3:A,B
```

```
150 PRINT #1:C,D
```

Causes A and B to be printed in record 3 of the file that was opened as number 1. PRINT #1:C,D causes C and D to be printed in record 4 of the same file.

Program

The following program prints out values in various positions on the screen.

```
100 CALL CLEAR
```

```
110 PRINT 1;2;3;4;5;6;7;8;9
```

```
120 PRINT 1,2,3,4,5,6
```

```
130 PRINT 1:2:3
```

```
140 PRINT
```

```
150 PRINT 1;2;3;
```

```
160 PRINT 4;5;6/4
```

```
RUN
```

```
1 2 3 4 5 6 7 8 9
```

```
1          2
```

```
3          4
```

```
5          6
```

```
1
```

```
2
```

```
3
```

```
1 2 3 4 5 1.5
```

PRINT USING

PRINT USING

Format

Print to the Screen

```
PRINT USING format-string[:print-list]
           line-number
```

Print to a File (or Device)

```
PRINT #file-number[,REC record-number],USING format-string[print-list]
                                           line-number
```

Cross Reference

IMAGE, PRINT

Description

The PRINT USING instruction enables you to define specific formats for numbers and strings you print.

You can use PRINT USING as either a program statement or a command.

The format-string specifies the print format. The format-string is a string expression; if you use a string constant you must enclose it in quotation marks. See IMAGE for an explanation of format-strings.

You can optionally define a format-string in an IMAGE statement, as specified by the line-number.

See PRINT for an explanation of the print-list print options.

The PRINT USING instruction is identical to the PRINT instruction with the addition of the USING option, except that:

You cannot use the TAB function.

You cannot use any print separator other than a comma (,), except that the print-list can end with a semicolon (;).

If you use PRINT USING to print to a file, the file must have been opened in DISPLAY format.

Examples

```
100 PRINT USING "###.##":32.5
Prints 32.50.
```

```
100 PRINT USING "THE ANSWER IS ###.##":123.98
Prints THE ANSWER IS 124.0.
```

```
100 PRINT USING 185:37.4,-86.2
185 IMAGE ###.#
Prints the values of 37.4 and -86.2 using the IMAGE statement in line 185.
```


RANDOMIZE**RANDOMIZE**

Format
RANDOMIZE[seed]

Cross Reference
RND

Description
The RANDOMIZE instruction varies the sequence of pseudo-random numbers generated by the RND function.

You can use RANDOMIZE as either a program statement or a command.

The optional seed is a numeric-expression whose value specifies the random number sequence to be generated by RND functions. The first two bytes of the internal representation of the value of the seed determine the random number sequence generated by RND. If the first two bytes of the seed are identical each time you run your program, the same random number sequence is generated. If you do not enter a seed, a different and unpredictable sequence of random numbers is generated by RND each time you run your program.

Program

The following program illustrates a use of the RANDOMIZE statement. It accepts a value for the seed and prints the first 10 values obtained using the RND function.

```
100 CALL CLEAR
110 INPUT "SEED: ";S
120 RANDOMIZE S
130 FOR A=1 TO 10::PRINT A;RND::NEXT A::PRINT
140 GOTO 110
(Press CLEAR to stop the program.)
```


READ

READ

Format

READ variable-list

Cross Reference

DATA, RESTORE

Description

The READ statement enables you to assign constants (stored within your program in DATA statements) to variables.

The variable-list, consisting of one or more variables separated by commas, specifies the numeric and/or string variables that are to be assigned values. When a READ statement is executed, the variables in its variable-list are assigned values from the data-list of a DATA statement. Unless you use a RESTORE statement to specify otherwise, DATA statements are read in ascending line-number order.

If a data-list does not contain enough values to assign to all the variables, the READ statement assigns values from subsequent DATA statements until all the variables have been assigned a value. If there are no more DATA statements, a program error occurs and the message Data error in line-number is displayed.

If a numeric variable is specified in the variable-list, a numeric constant must be in the corresponding position in the data-list of a DATA statement. If a string variable is specified in the variable-list, either a string or a numeric constant can be in the corresponding position in the DATA statement.

See the DATA statement for examples.

REC --Function--Record Number

REC

Format

REC(file-number)

Type

DEFINT

Description

The REC function returns a record number reflecting the position of the next record in the specified file.

The file-number is a numeric-expression whose value specifies the number of the file as assigned in its OPEN instruction.

The REC function returns the number of the record in the specified file that is to be accessed by the next PRINT, INPUT, or LINPUT instruction (the next sequential record). (REC always treats a file as if it were being accessed sequentially, even if it has been opened for relative access.)

The records in a file are numbered sequentially starting with zero.

Example

100 PRINT REC(4)

Prints the position of the next record in the file that was opened as number 4.

Program

The following program illustrates a use of the REC function.

```
100 CALL CLEAR
110 OPEN #1:"DSK1.PROFILE",RELATIVE,INTERNAL
120 FOR A=0 TO 3
130 PRINT #1:"THIS IS RECORD",A
140 NEXT A
150 RESTORE #1
160 FOR A=0 TO 3
170 PRINT REC(1)
180 INPUT #1:A$,B
190 PRINT A$;B
200 NEXT A
210 CLOSE #1
RUN
```

```
0  
THIS IS RECORD 0  
1  
THIS IS RECORD 1  
2  
THIS IS RECORD 2  
3  
THIS IS RECORD 3
```

Line 110 opens a file.

Lines 120 through 140 write four records on the file.

Line 150 resets the file to the beginning.

Lines 160 through 200 print the file position and read and print the values at that position.

Line 210 closes the file.

RECTANGLE --Subprogram

RECTANGLE

Format

```
CALL RECTANGLE(line-type,pixel-row1,pixel-column1,
pixel-row2,pixel-column2,pixel-row3 ,pixel-column3[,...])
```

Cross Reference

CIRCLE, DCOLOR, DRAW, DRAWTO, FILL, GRAPHICS, POINT

Description

The RECTANGLE subprogram enables you to place rectangles of various types and proportions on the screen.

Rectangles may be hollow (only the perimeter of the rectangle is drawn), or solid (both the perimeter and the entire area enclosed by the perimeter is drawn).

Line-type is a numeric-expression whose value specifies the action taken by the RECTANGLE subprogram.

TYPE	ACTION
5	Reverses the status of each pixel of the specified rectangle (solid). (If a pixel is on, it is turned off; if a pixel is off, it is turned on). This effectively reverses the color of the specified rectangle.
4	Draws a rectangle (solid), of the foreground-color specified by the DCOLOR subprogram. This is accomplished by turning on each pixel in the specified rectangle.
3	Erases a rectangle (solid). This is accomplished by turning off each pixel in the specified rectangle.
2	Reverses the status of each pixel in the perimeter of the specified rectangle. (If a pixel is on, it is turned off; if a pixel is off, it is turned on.) This effectively reverses the color of the perimeter.
1	Draws the perimeter of a rectangle, of the foreground-color specified by the DCOLOR subprogram. This is accomplished by turning on each pixel in the specified rectangle.
0	Erases the perimeter of a rectangle. This is accomplished by turning off each pixel in the specified rectangle.

Pixel-row(#), and pixel-column(#), are numeric-expressions whose values represent the screen positions of specific points of the rectangle. There are three points needed to define the rectangle, as shown below.

Pixel-row1 / pixel-column1 specify the TOP LEFT corner of the rectangle.

Pixel-row2 / pixel-column2 specify the TOP RIGHT corner of the rectangle.

Pixel-row3 / pixel-column3 specify the BOTTOM LEFT corner of the rectangle.

All pixel-rows must have a value from 1 to 192. All pixel-columns must have a value from 1 to 256.

Note that the first pixel set (pixel-row1 and pixel-column1) represents the top leftmost point of the rectangle and must have a lower column value than the second pixel set. The second pixel set represents the top rightmost point of the rectangle. In the same manner, the third pixel set, which represents the bottom leftmost point of the rectangle, must have a higher row value than set1 or set2.

If the procedure outlined above is not followed, an error is issued.

You can optionally draw more rectangles by specifying additional sets of pixels. You must specify three sets of pixels for each rectangle.

The bottom-rightmost point of the last rectangle drawn becomes the current position used by the DRAWTO subprogram.

RECTANGLE cannot be used in Pattern or Text Modes.

Program

```
100 CALL GRAPHICS(1,2)
110 CALL RECTANGLE(1,8,80,8,175,134,80)
120 FOR T=1 TO 8 :: CALL RECTANGLE(4,T* 16,100,T* 16,155,T* 16+T-1,100) ::
    NEXT T
130 FOR DELAY=1 TO 2000 :: NEXT DELAY
140 CALL RECTANGLE(3,16,100,16,155,128,100)
150 FOR DELAY=1 TO 2000 :: NEXT DELAY
160 END
```

Line 100 selects a usable graphics mode (and clears the screen).

Line 110 draws a large box on the screen.

Line 120 uses a for-next loop to fill the box with lines of different thickness. (This shows how RECTANGLE could be used to replace DRAW. RECTANGLE is slower, but more versatile.)

Line 130 uses a for-next loop to delay execution of the next statement.

Line 140 clears the lines, but leaves the box to illustrate how RECTANGLE can be used as an eraser.

Line 150 delays the execution of the next statement.
Line 160 ends the program.

REM --Remark

REM

Format
REM remark
! remark

Description

The REM statement enables you to document your program by including explanatory remarks within the program itself.

You can use any character in a remark.

The length of a REM statement is limited only by the length of a program statement.

A REM statement encountered during program execution is ignored by the computer.

Trailing Remarks

In addition to the REM statement, trailing remarks can be added to the ends of lines in MYARC Advanced BASIC, allowing detailed internal documentation of programs. An exclamation mark (!) begins each trailing remark.

Example

100 REM BEGIN SUBROUTINE
Identifies a section beginning a subroutine.

100 FOR X=1 to 16 ! BEGIN LOOP
Identifies a section beginning a FOR-NEXT loop.

RESEQUENCE

RESEQUENCE

Format

```
RESEQUENCE [initial-line-number][,increment]
RES
```

Description

The RESEQUENCE command assigns new line numbers to all lines in the program currently in memory.

If you enter an initial-line-number, the first line number assigned is one you specify. If you do not specify an initial-line-number, the computer starts with line number 100.

Succeeding line numbers are assigned by adding the value of the numeric-expression increment to the previous line number. Note that to specify an increment only (without specifying an initial-line-number), you must precede the increment with a comma. The default increment is 10.

To ensure that your program continues to function properly, all line-number references within your program are changed to reflect the newly assigned line numbers. (Line numbers mentioned in REM statements are not affected.) If an invalid line-number reference (a reference to a line number that does not exist in your program) is encountered, the computer changes the line-number reference to 32767, without displaying any error message or warning.

If the values you enter for the initial-line-number and increment would have the effect of creating a line number greater than 32767, the message Bad line number is displayed and the program is not resequenced.

Examples

```
RES
```

Resequences the lines of the program in memory to start with 100 and number by 10s.

```
RES 1000
```

Resequences the lines of the program to start with 1000 and number by 10s.

```
RES 1000,15
```

Resequences the lines of the program in memory to start with 1000 and number by 15s.

```
RES ,15
```

Resequences the lines of the program in memory to start with 100 and number by 15s.

RESTORE**RESTORE**

Format

Restore Data

RESTORE [line-number]

Restore a File

RESTORE #file-number[,REC record-number]

Cross Reference

DATA, INPUT, PRINT, READ

Description

The RESTORE instruction specifies either the DATA statement to be used with the next READ statement or the record to be accessed by the next file-processing instruction.

RESTORE with DATA and READ Statements

If you enter a line-number, the next READ statement executed assigns values beginning from the data-list in the specified DATA statement.

If the specified line-number is not the line-number of a DATA statement, the computer uses the first DATA statement with a line-number higher than the one you specified.

If there is no higher numbered DATA statement, a program error occurs and the message Data error in line-number is displayed (the line-number is the line number of the READ statement that caused the error).

If you do not enter a line-number or a file-number, the next READ statement executed assigns values beginning from the data-list of the first DATA statement in your program.

If there are no DATA statements in your program, the message Data error in line-number is displayed.

RESTORE with a File

If you enter a file-number, RESTORE repositions the specified file at its first record, record zero (unless you use the REC option). The file-number is a numeric-expression whose value specifies the number of the file as assigned in its OPEN instruction.

If you use the REC option, the record-number is a numeric-expression specifying the number of the record at which you want to position the file. The records in a file are numbered sequentially, starting with zero. The REC option can be used only with a file opened for RELATIVE access.

RESTORE terminates any print- or input-pending conditions.

Examples

100 RESTORE

Sets the next DATA statement to be used to the first DATA statement in the program.

100 RESTORE 130

Sets the next DATA statement to be used to the DATA statement at line 130 or, if line 130 is not a DATA statement, to the next DATA statement after line 130.

100 RESTORE #1

Sets the next record to be used by the next PRINT, INPUT, or LINPUT statement using file #1 to be the first record in the file.

100 RESTORE #4,REC H5

Sets the next record to be used by the next PRINT, INPUT, or LINPUT statement using file #4 to be record H5.

RETURN**RETURN**

Format

With GOSUB and ON GOSUB

RETURN

With ON ERROR

RETURN [NEXT
line-number]

Cross Reference

GOSUB, ON GOSUB, ON ERROR

Description

The RETURN statement causes program control to return to the main program from a subroutine called by a GOSUB, ON GOSUB, or ON ERROR statement.

RETURN with GOSUB and ON GOSUB

When the computer encounters a RETURN statement in a subroutine called by a GOSUB or ON GOSUB statement, program control returns to the statement immediately following the GOSUB or ON GOSUB statement.

No options are allowed with a RETURN statement in a subroutine called by a GOSUB or ON GOSUB statement.

RETURN with ON ERROR

The action taken by the computer when it encounters a RETURN statement in a subroutine called by an ON ERROR statement depends on the RETURN option.

If you specify the NEXT option, program control returns to the statement immediately following the statement that caused the error.

If you specify a line-number, program control is transferred to the specified program statement.

If you do not specify an option, program control returns to the statement that caused the error. The statement is re-executed.

RETURN "clears" the error, so that it can no longer be analyzed by the ERR subprogram.

Programs

The following program illustrates a use of RETURN as used with GOSUB. The program figures interest on an amount of money put into savings.

```
100 CALL CLEAR
110 INPUT "AMOUNT DEPOSITED: ":AMOUNT
120 INPUT "ANNUAL INTEREST RATE: ":RATE
130 IF RATE 1 THEN RATE=RATE* 100
140 PRINT "NUMBER OF TIMES COMPOUNDED"
```

```

150 INPUT "ANNUALLY: "COMP
160 INPUT "STARTING YEAR: ":Y
170 INPUT "NUMBER OF YEARS: ":N
180 CALL CLEAR
190 FOR A=Y TO Y+N
200 GOSUB 240
210 PRINT A,INT(AMOUNT* 100+.5)/100
220 NEXT A
230 STOP
240 FOR B=1 TO COMP
250 AMOUNT=AMOUNT+AMOUNT*RATE/(COMP* 100)
260 NEXT B
270 RETURN

```

The following program illustrates a use of RETURN with ON ERROR.

```

100 CALL CLEAR
110 A=1
120 ON ERROR 160
130 X=VAL("D")
140 PRINT 140
150 STOP
160 REM ERROR HANDLING
170 IF A>4 THEN 220
180 A=A+1
190 PRINT 190
200 ON ERROR 160
210 RETURN
220 PRINT 220 :: RETURN NEXT
RUN

```

```

190
190
190
190
220
140

```

Line 120 causes an error to transfer control to line 160. Line 130 causes an error.

Line 170 checks to see if the error has occurred four times and transfers control to 220 if it has. Line 180 increments the error counter by one. Line 190 prints 190. Line 200 resets the error handling to transfer to line 160. Line 210 returns to the line that caused the error and executes it again.

Line 220, which is executed only after the error has occurred four times, prints 220 and returns to the line following the line that caused the error.

Line 140, the next one after the one that causes the error, prints 140.

See also example of the ON ERROR statement.

RIGHT\$

RIGHT\$

Format

RIGHT\$(string-expression,length)

Cross Reference

LEFT\$, POS, STR\$

Description

RIGHT\$ returns the right-most "length" of characters from the string expression. If the string-expression is shorter than the length, the actual string-expression will be returned.

Example

```
10 A$="MY NAME IS HARRY POTTER"  
20 PRINT RIGHT$(A$,12)  
RUN  
HARRY POTTER
```

RND --Function--Random Number**RND**Format
RNDType
REALCross Reference
RANDOMIZEDescription
The RND function returns a pseudo-random number.

RND returns the next pseudo-random number in the current series of pseudo-random numbers. The number returned is always greater than or equal to 0 and less than 1.

The numbers returned by RND are called "pseudo-random" because they are not generated strictly at random, but are generated as members of predefined series. You can use the RANDOMIZE instruction to make the numbers generated by RND more random.

The same sequence of random numbers is generated by RND each time you run a particular program unless the program includes a RANDOMIZE instruction.

Examples

```
100 COLOR16=INT(RND* 16)+1
Sets COLOR16 equal to some number from 1 through 16.
```

```
100 VALUE=INT(RND* 16)+10
Sets VALUE equal to some number from 10 through 25.
```

```
100 LL(8)=INT(RND*(B-A+1))+A
Sets LL(8) equal to some number from A through B.
```

RPT\$ --Function--Repeat String

RPT\$

Format

RPT\$(string-expression,numeric-expression)

Type

String

Description

The RPT\$ function returns a string consisting of a specified string repeated a specified number of times.

The string-expression specifies the string to be repeated. If you use a string constant, it must be enclosed in quotation marks.

The value of the numeric-expression specifies the number of repetitions of the string-expression.

If the length of the string-expression and the value of the numeric-expression would create a string longer than 255 characters, the excess characters are discarded and the following message is displayed:

```
*WARNING
STRING TRUNCATED
```

Examples

```
100 M$=RPT$("ABCD",4)
Sets M$ equal to "ABCDABCDABCDABCD".
```

```
100 CALL CHAR(244,RPT$("0000FFFF",8))
Defines characters 244 through 247 with the string
"0000FFFF0000FFFF0000FFFF0000FFFF0000FFFF0000FFFF0000FFFF".
```

```
100 PRINT USING RPT$("#",40):X$
Prints the value of X$ using an image that consists of 40 number signs.
```

RUN

RUN

Format

Execute Program in Memory

RUN [line-number]

Execute Program on External Device

RUN file-specification[,Continue]

Description

The RUN instruction causes the computer either to execute the program currently in memory or to both load and execute a program from an external. You can use RUN as either a program statement or a command.

When you use RUN as a program statement, one program can start the execution of another program. This enables you to divide a large program into smaller segments, each of which can be loaded into memory only as needed.

If you specify a line-number, your program starts running at the specified program line.

If you enter a file-specification, your program is first loaded into memory from the specified external device, and then executed starting from the lowest-numbered line in the program. The file-specification is a string expression; if you use a string constant, you must enclose it in quotation marks. If you additionally specify the Continue option, the new program loaded must contain only variables used in the previous program. A syntax error will occur when trying to use a variable not contained in the previous program.

If you do not enter either a line-number or a file-specification, the computer executes the program currently in memory starting with the lowest-numbered line in the program.

Before the program starts running, the computer:

Sets the values of all numeric variables to zero.

Sets the values of all string variables to null strings (strings containing no characters).

Closes all open files.

Restores the default screen color (cyan).

Deletes all sprites.

Resets the sprite magnification level to 1.

Checks for certain program errors.

RUN does not affect the graphics mode, margin settings, graphics colors (see DCOLOR), or current position (see DRAWTO).

Examples

RUN

Causes the computer to begin execution of the program in memory.

RUN 200

```
100 RUN 200
```

Causes the computer to begin execution of the program in memory starting at line 200.

RUN "DSK1.PRG3"

```
100 RUN "DSK1.PRG3"
```

Causes the computer to load and begin execution of the program named PRG3 from the diskette in disk drive 1.

100 A\$="DSK1.MYFILE"

```
110 RUN A$
```

Causes the computer to load and begin execution of the program named MYFILE from the diskette in disk drive 1.

Program

The following program illustrates a use of the RUN command used as a statement. It creates a "menu" and lets the person using the program choose what other program he wishes to run. The other programs should RUN this program rather than ending in the usual way, so that the menu is given again after they are finished.

```
100 CALL CLEAR
110 PRINT "1 PROGRAM 1."
120 PRINT "2 PROGRAM 2."
130 PRINT "3 PROGRAM 3."
140 PRINT "4 END."
150 PRINT
160 INPUT "YOUR CHOICE: "C
170 IF C=1 THEN RUN "DSK1.PRG1"
180 IF C=2 THEN RUN "DSK1.PRG2"
190 IF C=3 THEN RUN "DSK1.PRG3"
200 IF C=4 THEN STOP
210 GOTO 100
```


SAVE

SAVE

Format

SAVE file-specification[,MERGE
PROTECTED]

Cross Reference

MERGE, OLD

Description

The SAVE command copies the program in memory to an external storage device. When you are using SAVE, your program remains in memory, even if an error occurs.

The saved program can later be loaded back into memory with the OLD command.

The file-specifications names the program to be stored. The file-specification, a string constant, optionally can be enclosed in quotation marks.

To specify that your program is to be available for merging with other programs, use the MERGE option. If you use the MERGE option, the program is stored as a SEQUENTIAL file in DISPLAY format with VARIABLE records (see OPEN); MERGE can be used only with devices that accept these options.

For more information about using MERGE with a particular device, refer to the owner's manual that comes with that device.

If you do not use the MERGE option, your program cannot later be merged with another program.

If you use the PROTECTED option, you ensure that the program, when subsequently loaded with the OLD command, cannot be listed, edited, or saved.

A protected program starts executing automatically when it is loaded; when the program ends (either normally or because of an error) or stops at a breakpoint, it is erased from memory. As the PROTECTED option is not reversible, it is recommended that you keep an unprotected version of the program. If you also wish to protect a diskette-based program from being deleted, use the protect feature of the Disk Manager.

SAVE removes any breakpoints you have set in your program.

Examples

SAVE DSK1.PRG1

Saves the program in memory on the diskette in disk drive 1 under the name PRG1.

SAVE DSK1.PRG1,PROTECTED

Saves the program in memory on the diskette in disk drive 1 under the name PRG1. The program may be loaded into memory, but it may not be edited, listed, or resaved.

SAVE DSK1.PRG1,MERGE

Saves the program in memory on the diskette in disk drive 1 under the name PRG1. The program may later be merged with a program in memory by using the MERGE command.

SAY --Subprogram

SAY

Format

CALL SAY(word-string[,direct-string][,...])

Cross Reference

SPGET

Description

The SAY subprogram enables you to instruct the computer to produce speech.

Word-string is a string-expression whose value is any of the words or phrases in the computer's resident vocabulary. If you use a string constant, you must enclose it in quotation marks. Alphabetic characters must be upper-case.

The computer substitutes "UHOH" for a word-string not in the vocabulary.

A speech phrase (more than one word) must be enclosed in pound signs (#). A speech phrase must be predefined; that is, it must be resident in the computer's vocabulary.

A compound is a new word formed by combining two words already in the vocabulary. For example, SOME+THING produces "something" and THERE+FOUR produces "therefore". A compound must not be enclosed in pound signs.

See Appendix H for a list of the computer's resident vocabulary.

Direct-string is a string-expression whose value is the computer's internal representation of a word or phrase. You can use or modify a direct-string returned by the SPGET subprogram.

See Appendix I for information on adding suffixes to direct-strings. You can specify multiple word-strings and direct-strings by alternating them. To specify two consecutive word-strings or direct-strings, enter an extra comma as a separator between them.

Examples

```
100 CALL SAY("HELLO, HOW ARE YOU")
```

Causes the computer to say "Hello, how are you".

```
CALL SAY(,A$,,B$)
```

Causes the computer to say the words indicated by A\$ and B\$, which must have been returned by SPGET.

The following program illustrates a use of CALL SAY with a word-string and three direct-strings.

```
100 CALL SPGET("HOW",X$)
110 CALL SPGET("ARE",Y$)
120 CALL SPGET("YOU",Z$)
130 CALL SAY("HELLO",X$,,Y$,,Z$) .
```

SCREEN --Subprogram**SCREEN**

Format

```
CALL SCREEN([,foreground-color,]background-color)
```

Cross Reference

COLOR, DCOLOR, GRAPHICS

Description

The SCREEN subprogram enables you to change the screen color. The screen color is the color of the border and the color displayed when transparent is specified as the foreground- or background-color of a character or pixel.

In Text Mode, SCREEN enables you to change the color of the displayed characters, as well as the color of the screen.

Background-color is a numeric-expression whose value specifies a screen color from among the 16 available colors.

In Text Modes, foreground-color is a numeric-expression whose value specifies a color from among the 16 available colors, representing the foreground-color of all 256 characters.

If you specify a foreground-color and the computer is not in Text Mode, it has no effect. If the computer is in Text Mode and you do not specify foreground-color, the foreground-color remains unchanged.

When you enter MYARC Advanced BASIC, the background-color is cyan and the foreground-color is black. When your program ends (either normally or because of an error), stops at a breakpoint, or changes graphics mode, the default colors are restored.

The codes for the available colors are listed in Appendix F.

Examples

```
100 CALL SCREEN(8)
```

Changes the screen to cyan, which is the standard screen color.

```
100 CALL SCREEN(2)
```

Changes the screen to black.

Program

The following program uses CALL SCREEN with CALL VCHAR and PRINT in the Text Mode to change the color of a character.

```
100 CALL CLEAR
110 CALL GRAPHICS(2,1)
120 CALL VCHAR(12,12,33,3)
130 CALL SCREEN(5,16)
140 PRINT "DARK BLUE SCREEN WITH WHITE LETTERS"
150 GOTO 150
(Press CLEAR to stop the program.)
```

Line 130 changes the screen to dark blue and the characters to white.

SEG\$ --Function--String Segment**SEG\$**

Format

SEG\$(string-expression, start-position, length)

Type

String

Description

The SEG\$ function returns a specified substring (segment of a string).

The string-expression specifies the string of which you want to specify a substring. If you use a string constant, it must be enclosed in quotation marks.

The start-position is a numeric-expression whose value specifies the character position in the string-expression where the substring begins. The value of the start-position must be greater than zero.

The length is a numeric-expression whose value specifies the length of the substring.

If the start-position is greater than the length of the string-expression, or if the length is zero, SEG\$ returns a null string.

If the specified length is greater than the remaining length of the string-expression (starting from the specified start-position), SEG\$ returns a substring consisting of all characters in the string-expression starting from the start-position to the end of the string-expression.

Examples

```
100 X$=SEG$("FIRSTNAME LASTNAME",1,9)
Sets X$ equal to FIRSTNAME.
```

```
100 Y$=SEG$("FIRSTNAME LASTNAME"11,8)
Sets Y$ equal to LASTNAME.
```

```
100 Z$=SEG$("FIRSTNAME LASTNAME",10,1)
Sets Z$ equal to " ".
```

```
100 PRINT SEG$(A$,B,C)
Prints the substring of A$ starting at the character at position B and extending for C characters.
```

SGN --Function--Signum (Sign)**SGN**

Format

SGN(numeric-expression)

Type

DEFINT

Description

The SGN function returns a number indicating the algebraic sign of the value of the numeric-expression.

If the value of the numeric-expression is negative, SGN returns a -1.

If the value of the numeric-expression is zero, SGN returns a 0.

If the value of the numeric-expression is positive, SGN returns a (+)1.

Examples

```
100 IF SGN(X2)=1 THEN 300 ELSE 400
```

Transfers control to line 300 if X2 is positive and to line 400 if X2 is zero or negative.

```
100 ON SGN(X)+2 GOTO 200,300,400
```

Transfers control to line 200 if X is negative, line 300 if X is zero, and line 400 if X is positive.

SIN --Function--Sine**SIN**

Format
 SIN(numeric-expression)

Type
 REAL

Cross Reference
 ATN, COS, TAN

Description
 The SIN function returns the sine of the angle whose measurement in radians is the value of the numeric-expression.

The value of the numeric-expression cannot be less than -1.5707963267944E10 or greater than 1.5707963267944E10.

To convert the measure of an angle from degrees to radians, multiply pi/180.

Program

The following program gives the sine for each of several angles.

```

100 A=.5235987755982
110 B=30
120 C=45*PI/180
130 PRINT SIN(A);SIN(B)
140 PRINT SIN(B*PI/180)
150 PRINT SIN(C)
RUN
.5 -.9880316241
.5
.7071067812
```


SOUND --Subprogram**SOUND**

Format

```
CALL SOUND(duration,frequency1,volume1[,frequency2,volume2]
[,frequency3,volume3][,frequency4,volume4])
```

Description

The SOUND subprogram enables you to instruct the computer to produce musical tones or noise.

The computer contains three music generators and one noise generator, enabling you to create up to four different sounds at once. You can specify the frequency and volume of each sound independently.

Duration is a numeric-expression whose absolute value specifies the length of the sound in milliseconds (thousandths of seconds). Duration can have an absolute value from 1 to 4250. (A value of 1000 will produce a sound for one second.)

The actual duration produced by the computer may vary by as much as one sixtieth (1/60) of a second from the value you specify.

You can enter only one duration, which applies to all specified sounds (music and noise).

Frequency is a numeric-expression that has different meanings depending on whether you use it to specify one of the music generators or the noise generator.

You must enter at least one frequency.

The frequency of a music generator specifies the frequency of the tone in Hertz (cycles per second). The acceptable values range from 110 to 44733; the upper limit exceeds the range of human hearing.

The actual frequency produced by the computer may vary by as much as ten percent from the value you specify.

See Appendix C for the frequencies of some commonly used tones.

The frequency of the noise generator has a value from -1 to -8, specifying the type of noise produced.

The frequencies from -1 to -3 produce different types of periodic noise. A frequency of -4 produces a periodic noise that varies depending on the frequency value of the third music generator.

The frequencies from -5 to -7 produce different types of white noise. A frequency of -8 produces a white noise that varies depending on the frequency value of the third music generator.

Volume is a numeric-expression whose value is inversely proportional to the loudness of the sound.

You must enter at least one volume.

The volume can be from 0 to 30. Zero is the maximum volume and 30 is silence.

If you call SOUND while the computer is still producing the tones specified in a previous call to the SOUND subprogram, the result depends on the algebraic sign of the duration of the previous call to SOUND. If the duration was positive, the new sound does not begin until the old sound is complete. If the duration was negative, the new sound begins immediately, interrupting the old sound.

Examples

```
100 CALL SOUND(1000,110,0)
Plays A below low C loudly for one second.
```

```
100 CALL SOUND(500,110,0,131,0,196,3)
Plays A below low C and low C loudly, and G below C not as loudly, all for half a second.
```

```
100 CALL SOUND(4250,-8,0)
Plays loud white noise for 4.250 seconds.
```

```
100 CALL SOUND(DUR,TONE,VOL)
Plays the tone indicated by TONE for a duration indicated by DUR, at a volume indicated by VOL.
```

Program

The following program plays the 13 notes of the first octave that is available on the computer.

```
100 X=2^(1/12)
110 FOR A=1 TO 13
120 CALL SOUND(100,110*X^A,0)
130 NEXT A
```

SPEED

SPEED

Format

CALL SPEED(x)

Description

Command to set speed of execution of BASIC program.

- SPEED = 5 Maximum speed
- SPEED = 3 Approx. speed of MYARC Advanced BASIC
- SPEED = 2 Approx. speed of TI Extended BASIC
- SPEED = 1 Minimum speed - TI 99/4A emulation

SPGET --Subprogram--Get Speech**SPGET**

Format

CALL SPGET(word-string,string-variable[,...])

Cross Reference

SAY

Description

The SPGET subprogram enables you to assign the computer's internal representation of a speech word to a variable.

SPGET is especially useful if you want to add a suffix to a word in the computer's resident vocabulary.

Word-string is a string-expression whose value is any of the words or phrases in the computer's resident vocabulary. If you use a string constant, you must enclose it in quotes.

The computer substitutes "UHOH" for a word-string not in the vocabulary.

A speech phrase (more than one word) must be enclosed in pound signs (#).

See Appendix H for a list of the computer's resident vocabulary.

The internal representation of the word-string (the direct-string) is returned in the string-variable. See Appendix I for information on adding suffixes to direct-strings.

You can specify multiple word-strings and direct-strings by alternating them.

Program

The following program illustrates using CALL SPGET.

```
110 CALL SPGET("COMPUTER",Y$)
120 CALL SAY("I AM A",Y$)
```

SPRITE --Subprogram**SPRITE**

Format

CALL SPRITE(#sprite-number,character-code,foreground-color,
pixel-row,pixel-column[, vertical-velocity,horizontal-velocity][,...])

Cross Reference

CHAR, COINC, COLOR, DELSPRITE, DISTANCE, GRAPHICS, LOCATE, MAGNIFY, MOTION,
PATTERN, POSITION, SCREEN

Description

The SPRITE subprogram enables you to create sprites.

Sprites are graphics that can be assigned any valid color and placed anywhere on the screen. Sprites treat the screen as a grid 256 pixels high and 256 pixels wide. However, only the first 192 pixels are visible on the screen.

You can create up to 32 sprites in all graphics modes except Text Modes, which do not allow sprites (the SPRITE subprogram has no effect in Text Modes).

Sprites can be set in motion in any direction at a variety of speeds. A sprite continues its motion until it is specifically changed by the program or until program execution stops. Because sprites move from pixel to pixel, their motion can be smoother than that of characters, which can be moved only one character position (6 or 8 pixels) at a time.

Sprites "pass over" characters on the screen. When two or more sprites are coincident (occupying the same screen pixel position), the sprite with the lowest sprite-number covers the other sprite(s).

At any given time, only four sprites (in Graphics(1,1) and (1,2)) or eight sprites (in the other graphics modes) can be on the same horizontal pixel-row. Once this limit is exceeded the row of pixels in the sprite(s) with the highest sprite-number(s) disappears.

You can use the DELSPRITE subprogram to delete one or more sprites. All sprites are deleted when your program ends (either normally or because of an error), stops at a breakpoint, or changes graphics mode.

Sprite Specifications

The sprite-number is a numeric-expression with a value from 1 to 32. If you specify the value of a previously defined sprite, the old sprite is replaced by the new sprite. If the old sprite had a vertical- or horizontal-velocity and you do not specify a new velocity, the new sprite retains the old velocity.

Character-code is a numeric-expression with a value from 0-255, specifying the character that defines the sprite pattern.

If you use the MAGNIFY subprogram to change to double-sized sprites, the

sprite definition includes the character specified by the character-code and three additional characters (see MAGNIFY).

Once defined by the SPRITE subprogram, the character-code of a sprite can be changed by the PATTERN subprogram.

The foreground-color is a numeric-expression with a value from 1 to 16, specifying one of the 16 available colors. Once defined by the SPRITE subprogram, the foreground-color of a sprite can be changed by the COLOR subprogram.

The background-color of a sprite is always transparent.

The pixel-row and pixel-column are numeric-expressions whose values specify the screen pixel position of the pixel at the upper-left corner of the sprite.

Once defined by the SPRITE subprogram, the pixel-row and pixel-column of a sprite can be changed by the LOCATE subprogram, and the current pixel-row and pixel column of a sprite can be ascertained by the POSITION subprogram. Also, the distance between sprites or between a sprite and a specified screen pixel can be ascertained by the DISTANCE subprogram, and the COINC subprogram can be used to ascertain whether sprites are coincident with each other or with a specified screen pixel.

Sprite Motion

The optional vertical- and horizontal-velocity are numeric-expressions with values from -128 to 127. If both values are zero, the sprite is stationary. The speed of a sprite is in direct linear proportion to the absolute value of the specified velocity.

A positive vertical-velocity causes the sprite to move toward the top of the screen; a negative vertical-velocity causes the sprite to move toward the bottom of the screen.

A positive horizontal-velocity causes the sprite to move to the right; a negative horizontal-velocity causes the sprite to move to the left.

If neither the vertical- nor horizontal-velocity are zero, the sprite moves at an angle, in a direction and at a speed determined by the velocity values.

The velocity of a sprite can be changed by the MOTION subprogram.

When a moving sprite reaches an edge of the screen, it disappears. The sprite reappears in the corresponding position at the opposite edge of the screen.

The motion of a sprite may be affected by the computer's internal processing and by input to, and output from, external devices.

Program

The following three programs show some possible uses of sprites.

```

100 CALL CLEAR
110 CALL CHAR(244,"FFFFFFFFFFFFFFFF")
120 CALL CHAR(246,"183C7EFFFF7E3C18")
130 CALL CHAR(248,"FOOFFOOFFOOFFOOF")
140 CALL SPRITE(#1,244,5,92,124,#2,248,7,1,1)
150 CALL SPRITE(#28,33,16,12,48,1,1)
160 CALL SPRITE(#15,246,14,1,1,127,-128)
170 GOTO 170
(Press CLEAR to stop the program.)

```

Line 140 creates a dark blue sprite in the center of the screen and a red striped sprite in the upper-right corner of the screen. Line 150 creates a white sprite near the upper-left corner of the screen and starts it moving slowly at a 45-degree angle down and to the right. The sprite is an exclamation point.

Line 160 creates a dark red sprite at the upper-right corner of the screen and starts it moving very fast at a 45 degree angle down and to the left.

The following program makes a rather spectacular use of sprites.

```

100 CALL CLEAR
110 CALL CHAR(244,"0008081C7F1C0808")
120 RANDOMIZE
130 CALL SCREEN(2)
140 FOR A=1 TO 28
150 CALL SPRITE(#A,244,INT(A/3)+3,92,124,A*INT(RND*4.5)
-2.25+A/2*SGN(RND-.5),A*INT(RND*4.5)-2.25+A/2*SGN(RND-.5))
160 NEXT A
170 GOTO 140
(Press CLEAR to stop the program.)

```

Line 110 defines character 244.

Line 150 defines the sprites, 28 in all. The sprite-number is the current value of A. The character-value is 244. The sprite-color is $\text{INT}(A/3)+3$. The starting dot-row and dot-column are 92 and 124, the center of the screen. The row- and column-velocities are chosen randomly using the value of $A*\text{INT}(\text{RND}*4.5)-2.25+A/2*\text{SGN}(\text{RND}-.5)$.

Line 170 causes the sequence to repeat.

The following program uses all the subprograms that relate to sprites except for COLOR. They are CHAR, COINC, DELSPRITE, LOCATE, MAGNIFY, MOTION, PATTERN, POSITION, and SPRITE.

The program creates two double-sized magnified sprites in the shapes of two people walking along a floor. There is a barrier that one of them passes through and the other jumps through. The one that jumps through goes a


```
560 CALL MOTION(#2,0,4)
570 CALL DELSPRITE(#1)
580 FOR STEP1=1 TO 20
590 CALL PATTERN(#2,248)
600 FOR DELAY=1 TO 40 :: NEXT DELAY
610 CALL PATTERN(#2,244)
620 FOR DELAY=1 TO 40 :: NEXT DELAY
630 NEXT STEP1
640 CALL CLEAR
```

Lines 110, 120, 140, 150, 250, and 260 define the sprites.

Line 130 sets the meeting counter to zero.

Lines 170 through 200 build the floor.

Lines 210 through 240 build the barrier.

Line 270 sets the starting speed of the sprite that will speed up.

Line 290 sets the sprites in motion.

Line 300 creates the illusion of walking.

Line 320 checks to see if the sprites have met. Line 330 transfers control if the sprites have met.

Lines 340 and 350 check to see if the sprite has reached the barrier and transfer control if it has.

Line 360 loops back to continue the walk.

Lines 370 through 460 handle the sprites running into each other. Lines 380 and 390 stop them.

Line 400 checks to see if it is the first meeting.

Line 410 increments the meeting counter.

Line 420 finds the sprites position.

Line 430 makes them smaller.

Line 440 puts them on the floor and moves the fast one slightly ahead.

Line 450 starts them moving again.

Lines 470 through 530 handle the fast sprite jumping through the barrier. Line 480 stops it.

Line 490 finds where it is.

Line 500 puts it at the new location beyond the barrier.

Lines 510 and 520 start it moving again, a little faster.

Lines 540 through 640 handle the second meeting.

Line 560 starts the slow sprite moving.

Line 570 deletes the fast sprite.

Lines 580 through 630 make the slow sprite walk 20 steps.

SQR --Function--Square Root

SQR

Format

SQR(numeric-expression)

Type

REAL

Description

The SQR function returns the positive square root of the value of the numeric-expression.

The value of the numeric-expression cannot be negative.

Examples

```
100 PRINT SQR(4)
```

Prints 2.

```
100 X=SQR(2.57E5)
```

Sets X equal to the square root of 257,000, which is 506.9516742255.

STOP**STOP**

Format
STOP

Cross Refernce
END

Description
The STOP statement stops the execution of your program.

When your computer encounters a STOP statement, the computer performs the following operations:

It closes all open files.

It restores the default character definitions of all characters.

Restores the default foreground-color (black) and background-color (transparent) to all characters.

Restores the default screen color (cyan).

Deletes all sprites.

Resets the sprite magnification level to 1.

The graphics colors (see DCOLOR) and current position (see DRAWTO) are not affected. If the computer is in Pattern or Text Mode the graphics mode and margin settings remain unchanged.

A STOP statement is not necessary to stop your program; the program automatically stops after the highest-numbered line is executed.

STOP is frequently used before a subprogram that follows the main portion of a program, to ensure that the subprogram is not executed after the execution of the highest-numbered line in the main program.

STOP can be used interchangeably with the END statement, except that you cannot use STOP to end a subprogram.

Program

The following program illustrates a use of the STOP statement. The program adds the numbers from 1 to 100.

```
100 CALL CLEAR
110 TOT=0
120 NUMB=1
130 TOT=TOT+NUMB
140 NUMB=NUMB+1
150 IF NUMB>100 THEN PRINT TOT::STOP
160 GOTO 130
```

STR\$ --Function--String-Number**STR\$**

Format

STR\$(numeric-expression)

Type

String

Cross Reference

VAL

Description

The STR\$ function returns the string representation of the value of the numeric-expression.

STR\$ enables you to use the string representation of the numeric-expression with an instruction that requires a string-expression as a parameter.

STR\$ is the inverse of the VAL function.

STR\$ removes leading and trailing spaces.

Examples

```
100 NUM$=STR$(78.6)
```

Sets NUM\$ equal to "78.6".

```
100 LL$=STR$(3E15)
```

Sets LL\$ equal to "3.E+15".

```
100 X$=STR$(A*4)
```

Sets X\$ equal to a string representation of whatever value is obtained when A is multiplied by 4. For instance, if A is equal to -8, X\$ is set equal to "-32".

SUB --Subprogram

SUB

Format

SUB subprogram-name[(parameter[,...])]

Cross Reference

CALL, SUBEND, SUBEXIT

Description

The SUB statement is the first statement in a subprogram.

You can use a subprogram to separate a group of statements from the main program. Subprograms are generally used to perform a specific operation several times in the same program or in different programs, or to isolate variables that are specific to the subprogram.

Subprograms are accessed from your main program with a CALL statement. The subprogram-name in the SUB statement is the same name that you use in the CALL statement that transfers control to the subprogram.

The maximum length of a subprogram-name is 15 characters.

A user-written subprogram may have the same subprogram-name as a built-in subprogram. In such a case, a CALL statement will access the user-written subprogram instead of the built-in one.

You can use parameters to pass values to a subprogram. Parameters must be valid names of variables or arrays.

SUBEND must be the last statement executed in a subprogram. When the computer encounters a SUBEND or a SUBEXIT statement in a subprogram, program control returns to the statement immediately following the CALL statement that called the subprogram.

It is recommended that you do not use any statement other than SUBEND or SUBEXIT to leave a subprogram. If you use another statement to leave a subprogram you may still be using variables local to the subprogram, which may cause unexpected results.

Subprograms must have higher line numbers than any part of your main program. A SUB statement cannot be part of an IF THEN statement.

Subprogram Variables

The variables used in a subprogram (other than those used as parameters) are local to the subprogram; that is, even if a variable in your main program has the same name as a variable in a subprogram, the value of that variable outside the subprogram is not affected by changes to its value in the subprogram. If a subprogram is called more than once, any local variables used in the subprogram retain their values from one call to the next.

Parameters

When your program executes a subprogram beginning with a SUB statement with parameters, the parameter values (constants or variables) are passed from the parameter-list of the CALL statement to the subprogram. The parameter-list in the CALL statement must contain the same number of parameters as the SUB statement. Values are passed in the order in which they are listed.

A numeric parameter must be passed a numeric value. A string parameter must be passed a string value.

An array parameter must be passed an array. A string-array parameter must be passed a string array.

To pass an entire array as one parameter, follow the array name with left and right parentheses. If the array has more than one dimension, place one comma between the parentheses for each additional dimension.

Passing Parameters by Reference and Value

When a subprogram manipulates the value of a parameter passed to it, the new parameter value may or may not be passed back to the main program. When a parameter is passed to a subprogram "by reference", the new value is passed back to the main program after the subprogram has executed.

When a parameter is passed to a subprogram "by value", the new value is not passed back to the main program.

Variables, array elements, and arrays are normally passed by reference. However, if a numeric variable or array element is of a different data-type in the main program than it is in the subprogram, the parameter is passed by value.

To specify that a variable or array element is to be passed by value rather than by reference, enclose it in parentheses in the CALL statement's parameter-list. Note that this option is not available for arrays.

If you use an expression as a parameter, it is evaluated and passed by value.

Examples

```
100 SUB MENU
```

Marks the beginning of a subprogram. No parameters are passed or returned.

```
100 SUB MENU(COUNT,CHOICE)
```

Marks the beginning of a subprogram. The variables COUNT and CHOICE may be used and/or have their values changed in the subprogram and returned to the variables in the same position in the calling statement.

```
100 SUB PAYCHECK( DATE,Q,SSN,PAYRATE, TABLE(,))
```

Marks the beginning of a subprogram. The variables DATE, Q, SSN, PAYRATE, and the array TABLE with two dimensions may be used and/or have their values changed in the subprogram and returned to the variables in the same position in the calling statement.

Program

The following program illustrates a use of SUB. The subprogram MENU had been previously saved with the MERGE option. It prints a menu and requests a choice. The main program tells the subprogram how many choices there are and what the choices are. It then uses the choice made in the subprogram to determine what program to run.

```
100 CALL MENU(5,R)
110 ON R GOTO 120,130,140,150,160
120 RUN "DSK1.PAYABLES"
130 RUN "DSK1.RECEIVE"
140 RUN "DSK1.PAYROLL"
150 RUN "DSK1.INVENTORY"
160 RUN "DSK1.LEDGER"
170 DATA ACCOUNTS PAYABLE,ACCOUNTS RECEIVABLE,PAYROLL,INVENTORY,GENERAL
LEDGER
```

Beginning of subprogram MENU.

Note that this R is not the same as the R used in lines 100 and 110 in the main program.

```
10000 SUB MENU(COUNT,CHOICE)
10010 CALL CLEAR
10020 IF COUNT>22 THEN PRINT "TOO MANY ITEMS" :: CHOICE=0 :: SUBEXIT
10030 RESTORE
10040 FOR R=1 TO COUNT
10050 READ TEMP$
10060 TEMP$=SEG$(TEMP$,1,25)
10070 DISPLAY AT(R,1):R;TEMP$
10080 NEXT R
10090 DISPLAY AT(R+1,1):"YOUR CHOICE: 1"
10100 ACCEPT AT(R+1,14)BEEP VALIDATE(DIGIT)SIZE(-2):CHOICE
10110 IF CHOICE>COUNT OR CHOICE<1 THEN 10100
10120 SUBEND
```


SUBEND --Subprogram End

SUBEND

Format
SUBEND

Cross Reference
SUB, SUBEXIT

Description

The SUBEND statement marks the end of a subprogram.

SUBEND must be the last statement executed in a subprogram. When the computer encounters a SUBEND statement in a subprogram, program control returns to the statement immediately following the CALL statement that called the subprogram.

It is recommended that you do not use any statement other than SUBEND or SUBEXIT to leave a subprogram. If you use another statement to leave a subprogram you may still be using variables local to the subprogram, which may cause unexpected results.

A SUBEND statement cannot be part of an IF THEN statement.

The only statements that can immediately follow a SUBEND statement are REM, END, or the SUB statement for the next subprogram.

SUBEXIT --Subprogram Exit

SUBEXIT

Format
SUBEXIT

Cross Reference
SUB, SUBEND

Description
The SUBEXIT statement enables you to leave a subprogram before the computer executes the SUBEND statement that ends the subprogram.

SUBEXIT enables you to have more than one exit from a subprogram.

When the computer encounters a SUBEXIT statement in a subprogram, program control returns to the statement immediately following the CALL statement that called the subprogram.

It is recommended that you do not use any statement other than SUBEND or SUBEXIT to leave a subprogram. If you use another statement to leave a subprogram you may still be using variables local to the subprogram, which may cause unexpected results.

SWAP

SWAP

Format

```
CALL SWAP(var1,var2)
```

Description

The SWAP statement is used to exchange the values of two variables, provided they are of the same type and same precision. If they are not of the same type a "TYPE MISMATCH" error will occur.

The SWAP statement cannot be used to "SWAP" the contents of two arrays, except as individual elements.

```
100 FOR I=1 to 100
110 CALL SWAP(A$(I),B$(I))
120 NEXT I
```

The SWAP statement can also be used to alphabetize two strings.

```
100 INPUT "STRING #1 >":A$
110 INPUT "STRING #2 >":B$
120 IF A$>B$ THEN CALL SWAP(A$,B$)
130 PRINT A$,B$
```

TAB --Function--Tabulate

TAB

Format

TAB(numeric-expression)

Cross Reference

DISPLAY, PRINT

Description

The TAB function specifies the starting position of the next item to be printed by a PRINT or DISPLAY instruction.

The numeric-expression specifies the starting position of the next print item in a print-list of a PRINT or DISPLAY instruction.

If the value of the numeric-expression is not an integer, it is rounded to the nearest integer. If the value of the numeric-expression is less than 1, it is replaced by 1.

If the value of the numeric-expression is greater than the record length of the screen or device, it is repeatedly reduced by the record length until it is less than or equal to the record length. The record length of the screen is the width of the screen window defined by the margins. For more information about the record length of a particular device, refer to the owner's manual that comes with that device.

Because the TAB function itself is treated as a separate print item, it must be preceded and/or followed by a print separator (usually a semicolon), unless it is the only item in the print-list.

If the number of characters already printed in the current record is greater than or equal to the position indicated by the value of the numeric-expression, the print item following the TAB is printed in the next record, beginning in the position specified by the value of the numeric-expression.

TAB can be used to print to a device or file only if the device or file has been opened in DISPLAY format.

TAB cannot be used with PRINT USING or DISPLAY USING.

Examples

```
100 PRINT TAB(12);35
```

Prints the number 35 at the twelfth position.

```
100 PRINT 356;TAB(18);"NAME"
```

Prints 356 at the beginning of the line and NAME at the eighteenth position of the line.

```
100 PRINT "ABCDEFGHijklm";TAB(5);"NOP"
```

Prints ABCDEFGHijklm at the beginning of the line and NOP at the fifth position of the next line.

TAN --Function--Tangent

TAN

Format

TAN(numeric-expression)

Type

REAL

Cross Reference

ATN, COS, SIN

Description

The TAN function returns the tangent of the angle whose measurement in radians is the value of the numeric-expression.

The numeric-expression cannot be less than $-1.5707963269514E10$ or greater than $1.5707963266374E10$.

To convert the measure to radians, multiply by $\pi/180$.

Program

The following program gives the tangent for each of several angles.

```
100 A=.7853981633973
110 B=26.565051177
120 C=45*PI/180
130 PRINT TAN(A);TAN(B)
140 PRINT TAN(B*PI/180)
150 PRINT TAN(C)
RUN
1. 7.17470553
.5
1
```

TERMCHAR --Function--Termination Character**TERMCHAR**

Format
TERMCHAR

Type
DEFINT

Cross Reference
ACCEPT, INPUT, LINPUT

Description
The TERMCHAR function returns the character code of the key pressed to exit from the previously executed INPUT, ACCEPT, or LINPUT statement.

In a program, the value returned by TERMCHAR depends on the key pressed to exit from the last instruction that accepted input from the keyboard.

VALUE RETURNED	KEY
1	AID
2	CLEAR
10	DOWN ARROW
11	UP ARROW
12	PROC'D
13	ENTER
14	BEGIN
15	BACK

If you use TERMCHAR as part of a command (unless it is preceded by ACCEPT, INPUT, or LINPUT), the value returned depends on the key pressed to enter the command (ENTER, UP ARROW, or DOWN ARROW).

Note that pressing CLEAR during keyboard input normally causes a break in the program. However, if your program includes an ON BREAK NEXT statement, you can use CLEAR to exit from an input field.

Program

The following program illustrates a use of TERMCHAR. The program displays name, address, and city, state, and zip code information entered from the keyboard. Line 160 enables you to correct errors in previously entered lines by pressing UP ARROW. This returns the cursor to the beginning of the line that immediately precedes the one from which UP ARROW was entered.

```

100 CALL CLEAR
110 R=5::C=12
120 DISPLAY AT(R,C-10):"NAME :"
130 DISPLAY AT(R+1,C-10):"ADDRESS:"
140 DISPLAY AT(R+2,C-10):"C,S,Z:"

```

```
150 ACCEPT AT(R,C)SIZE(-20):A$(R)
160 IF TERMCHAR=11 THEN R=R-1 ELSE R=R+1
170 IF R=7 THEN 150
180 DISPLAY AT(20,1):A$(5):A$(6):A$(7)
190 GOTO 110
(Press CLEAR to stop the program.)
```

TIME/TIME\$

TIME/TIME\$

Description

The computer has an internal clock that can be accessed from BASIC.

TIME\$ can be used to read the clock and TIME to set the clock.

TO SET CLOCK

Format

```
CALL TIME("hh:mm:ss")
```

The string length is always 8 characters. Therefore an hour less than 10 must be preceded by a 0.

The clock works on 24 hour time so all times after 12 noon must have 12 hours added to them.

Example

```
CALL TIME$("06:15:00")      6:15 A.M.
CALL TIME$("18:15:00")     6:15 P.M.
```

TO READ CLOCK

Format

```
PRINT TIME$
OR
T$=TIME$
```

Example

```
10 CALL CLEAR
20 DISPLAY AT(24,1):TIME$
30 GOTO 20
```

```
10 CALL TIME("00:00:00")
20 FOR I=1 TO 10000
30 NEXT I
40 PRINT TIME$
```

This is an easy way to see how long a program takes to execute.

TRACE

TRACE

Format
TRACE ON
TRACE OFF

Description

The TRACE ON instruction causes the computer to display the line number of each line in your program before it is executed.

TRACE ON enables you to see the order in which the computer performs statements as it runs your program. It is valuable as a debugging aid to help you find errors (such as unwanted infinite loops) in your program.

TRACE OFF removes the effect of the TRACE ON command.

You can use TRACE ON or TRACE OFF either as a program statement or a command.

Programs

The following programs display a trace of the order of execution of the program lines.

```
100 FOR J=1 TO 3
110 PRINT "WORD"
120 NEXT J
130 TRACE ON
RUN
```

```
100 FOR J=1 TO 3
110 PRINT "WORD"
120 NEXT J
TRACE ON
RUN
```

UNBREAK**UNBREAK**

Format

UNBREAK [line-number-list]

Cross Reference

BREAK

Description

The UNBREAK instruction removes a breakpoint from each program statement you specify.

You can use UNBREAK as either a program statement or a command.

The line-number-list consists of one or more line numbers separated by commas. When an UNBREAK instruction is executed, breakpoints are removed from the specified program lines.

If you do not include a line-number-list, UNBREAK removes all breakpoints, except for a breakpoint that occurs when a BREAK statement with no line-number-list is encountered in a program.

If the line-number-list includes an invalid line number (0 or a value greater than 32767), the message Bad line number is displayed. If the line-number-list includes a fractional or negative line number, the message Syntax error is displayed. In both cases, the UNBREAK instruction is ignored; that is, breakpoints are not removed even at valid line numbers in the line-number-list. If you were entering UNBREAK as a program statement, it is not entered into your program.

If the line-number-list includes a line number that is valid (1-32767) but is not the number of a line in your program, or a fractional number greater than 1, the message

* WARNING
LINE NOT FOUND

is displayed. (If you were entering UNBREAK as a program statement, the line-number is included in the warning message). A breakpoint is, however, removed from any valid line in the line-number-list that precedes the line number that caused the warning.

Examples

UNBREAK

450 UNBREAK

Removes all breakpoints (except those resulting from a BREAK statement with no line-number-list).

UNBREAK 100,130

350 UNBREAK 100,130

Removes the breakpoints from lines 100 and 130.

VAL --Function--Value

VAL

Format

VAL(string-expression)

Type

REAL

Cross Reference

STR\$

Description

The VAL function returns the numeric value of the string-expression.

VAL enables you to use the numeric value of the string-expression with an instruction that requires a numeric-expression as a parameter.

VAL is the inverse of the STR\$ function.

The string-expression must be a valid representation of a number. The length of the string-expression must be greater than 0 and less than 255. If you use a string constant, it must be enclosed in quotation marks.

Examples

```
100 NUMB=VAL("78.6")
```

```
110 PRINT NUMB
```

Prints 78.6.

```
100 LL=VAL("3E15")
```

Sets LL equal to 3E+15, or 315.

VALHEX --Function--Value of Hexadecimal Number

VALHEX

Format

VALHEX(string-expression)

Type

DEFINT

Description

VALHEX returns the numeric value of the hexadecimal number represented by the string-expression.

The string-expression specifies the hexadecimal (base 16) number to be converted to a decimal (base 10) number. If you use a string constant, it must be enclosed in quotation marks.

The string-expression must contain only valid hexadecimal digits (0-9,A-F). Alphabetic hexadecimal digits must be upper-case letters. VALHEX can convert a hexadecimal number from one to four digits long. If the length of the string-expression is greater than four, VALHEX uses only the last four characters.

VALHEX returns an integer greater than or equal to -32768 (hexadecimal 8000) and less than or equal to 32767 (hexidecimal 7FFF).

Examples

100 A=VALHEX("400A")
Sets A equal to 16394.

100 PRINT VALHEX("8200")
Prints -32256.

VCHAR --Subprogram--Vertical Character**VCHAR**

Format

CALL VCHAR(row,column,character-code[,number-of-repetitions])

Cross Reference

DCOLOR, GCHAR, GRAPHICS, HCHAR

Description

The VCHAR subprogram enables you to place a character on the screen and repeat it horizontally.

Row and column are numeric-expressions whose values specify the position on the screen where the character is displayed.

The value of row must be greater than or equal to 1 and not exceed the total number of rows in the present graphics mode.

The value of column must be greater than or equal to 1 and must not exceed the total number of columns in the present graphics mode.

VCHAR is not affected by margin settings.

Character-code is a numeric-expression with a value from 0-255, specifying the number of the character. See Appendix B for a list of ASCII character codes.

The optional number-of-repetitions is a numeric-expression whose value specifies the number of times the character is repeated horizontally. If the repetitions extend past the end of a column, they continue from the first character of the next column. If the repetitions extend past the end of the last column, they continue from the first character of the first column.

If you use VCHAR to display a character on the screen, and then later use CHAR, COLOR, or DCOLOR to change the appearance of that character, the result depends on the graphics mode.

In Pattern and Text Modes, the displayed character changes to the newly specified pattern and/or color(s).

In High-Resolution Mode, the displayed character remains unchanged.

Examples

```
100 CALL VCHAR(12,16,33)
```

Places character 33 (an exclamation point) in row 12, column 16.

```
100 CALL VCHAR(1,1,ASC("!"),768)
```

Places an exclamation point in row 1, column 1, and repeats it 768 times, which fills the screen in Pattern Mode.

```
100 CALL VCHAR(R,C,K,T)
```

Places the character with an ASCII code specified by the value of K in row R, column C, and repeats it T times.

VERSION**VERSION**

Format

```
CALL VERSION(numeric-variable)
```

Description

The VERSION subprogram returns a value indicating the version of BASIC being used.

In MYARC Advanced BASIC, VERSION returns a value of 300 to the numeric-variable you specify.

Example

```
100 CALL VERSION(V)  
Sets V equal to 300.
```

WEND**WEND**

The WEND statement terminates the loop that begins with WHILE.

Statements between WHILE and WEND are executed repeatedly until the condition stated in the WHILE statement is no longer true.

Unlike FOR-NEXT statements, WHILE-WEND loops may NOT be nested. WEND always continues the most recent WHILE loop until the WHILE statement's condition becomes false.

See WHILE for detailed description of the WHILE-WEND loop.

WHILE

WHILE

Format

```
WHILE condition :: ...program .... :: WEND
```

Cross Reference

WEND, FOR-NEXT, IF-THEN-ELSE

Description

The WHILE statement starts a loop which is executed repeatedly while the WHILE 'condition' is true. The loop is terminated with a WEND statement.

'Condition' is a logical expression, numeric or variable that WHILE evaluates. If the 'condition' is TRUE (or a non/zero value, ie., condition<>0), the program then loops between the WHILE and the WEND statements. When the condition is no longer TRUE (false or condition=0) WHILE passes execution to the statement after WEND.

Unlike FOR-NEXT statements WHILE-WEND loops may NOT be nested.

Example

```
100 A$=INKEY$
110 WHILE A$<>" " THEN 100
120 WEND
130 ..... program lines
140 END
RUN
```

This short routine checks the entries into the keyboard buffer until it is empty then proceeds to the rest of the program. The keyboard is said to be "Flushed".

```
100 REM WHILE TEST
110 WHILE NAME$<>"LAST"
120 READ NAME$,PHONE$
130 COUNT=COUNT +1
140 PRINT NAME$;TAB(20);PHONE$
150 WEND
160 PRINT "NUMBER OF NAMES=";COUNT
170 PRINT "WHILE HAS BEEN EVALUATED TO FALSE "
180 DATA MYARC,201-766-1700
190 DATA BILL REISS,201-000-0000
200 DATA LAST, LAST
```


I/O DEFAULTS

MYARC Advanced BASIC includes several features to simplify the direction of input and output to certain devices. These devices are your main mass storage device (typically DSK1), your main printer (typically PIO), your main communications port (typically RS232.BA=4800), and your main modem port (typically RS232.BA=1200). These defaults are set from the operating system defaults when BASIC is initially started.

The following names are used for reassigning a particular default device:

NAME	DEVICE
LPT	Default printing device
CHDIR	Default disk drive or directory
COM	Default communications port
MDM	Default modem port

To change a default, type the command and follow it by the desired device name. For example, the LLIST command prints the program to the main printer port. If your printer is connected to the RS232 port and not the PIO port you would need to redirect the output from LLIST. To do this, type LPT "RS232.BA=4800".

You may also check to see what a particular default is set for at any time. To do this, type any one of the following commands and BASIC will list what that device is set for:

COMMAND	DEVICE
PPT	Lists the default printer
PWD	Lists the default working directory
PCM	Lists the default communications port
PMD	Lists the default modem port

11-08-70

... ..
... ..
... ..
... ..
... ..

... ..
... ..
... ..
... ..
... ..

... ..
... ..
... ..
... ..
... ..

... ..
... ..
... ..
... ..
... ..

... ..
... ..
... ..
... ..
... ..

APPENDICES

- Appendix A: List of Commands, Statements, and Functions
- Appendix B: ASCII Codes
- Appendix C: Musical Tone Frequencies
- Appendix D: Character Sets
- Appendix E: Pattern-Identifier Conversion Table
- Appendix F: Color Codes
- Appendix G: Mathematical Functions
- Appendix H: List of Speech Words
- Appendix I: Adding Suffixes to Speech Words
- Appendix J: Error Messages
- Appendix K: Graphics Modes - Summary
- Appendix L: Program - Illustrating MOUSE Commands
- Appendix M: Additional Extended ASCII Codes for Keyboard Mode 6

APPENDIX A:

COMMANDS, STATEMENTS, AND FUNCTIONS

The following is a list of all MYARC Advanced BASIC commands, statements, and functions. Commands are listed first; if a command can also be used as a statement, the letter "S" is listed to the right of the command. Commands that can be abbreviated have the acceptable abbreviations underlined>. Next is a list of all MYARC Advanced BASIC statements; those that can also be used as commands have a "C" after them. Finally, there is a list of all MYARC Advanced BASIC functions.

MYARC Advanced BASIC Commands

BREAK	S	LIST		PCM	PMD	} I/O	Default
BYE		LLIST		PPT	PWD		Commands
CHDIR		LPT	S	RESEQUENCE			
CLOSE		LTRACE	S	RUN			S
CLS	S	MERGE		SAVE			
CONTINUE		NEW		CALL SPEED			
DELETE	S	NUMBER		TRACE ON/OFF			S
KEY	S	OLD		UNBREAK			S

MYARC Advanced BASIC Statements

ACCEPT	C	END		CALL MEMSET	
BEEP	C	CALL ERR	C	CALL MOTION	C
CALL	C	CALL FILES	C	MOUSE	C
CALL BCOLOR	C	CALL FILL		NEXT	C
CALL BTIME	C	FOR TO	C	ON BREAK	
CALL CHAR	C	CALL GCHAR	C	ON ERROR	
CALL CHARPAT	C	GOSUB		ON GOSUB	
CALL CHARSET	C	GOTO		ON GOTO	
CALL CIRCLE	C	CALL GRAPHICS	C	ON WARNING	
CALL CLEAR	C	CALL HCHAR	C	OPEN	C
CLOSE	C	IF THEN ELSE		OPTION BASE	
CALL COINC	C	IMAGE		CALL OUT	C
CALL COLOR	C	CALL INIT	C	CALL PATTERN	C
DATA		INPUT		CALL PEEK	C
CALL DATE	C	INPUT		CALL PEEKV	C
CALL DCOLOR	C	CALL JOYST	C	CALL POINT	C
DEF		CALL KEY	C	CALL POKEV	C
DEFvartype		KILL	C	CALL POSITION	C
CALL DESPRITE	C	LET	C	PRINT	C
DIM	C	CALL LINK	C	PRINT USING	C
DISPLAY	C	LINPUT		RANDOMIZE	C
DISPLAY USING	C	CALL LOAD	C	READ	C
CALL DISTANCE	C	CALL LOCATE	C	CALL RECTANGLE	
CALL DRAW		CALL MAGNIFY	C	REM	C
CALL DRAWTO		CALL MARGIN	C	RESTORE	C

RETURN		STOP	C	CALL VCHAR	C
CALL SAY	C	SUB		CALL VERSION	C
CALL SCREEN	C	SUBEND		WEND	
CALL SOUND	C	SUBEXIT		WHILE	
CALL SPGET	C	CALL SWAP			
CALL SPRITE	C	CALL TIME	C		

MYARC Advanced BASIC Functions

ABS	HEX\$	SEG\$
ASC	LEFT\$	SGN
ATN	LEN	SIN
CDBL	LOG	SQR
CHR\$	MAX	STR\$
CINT	MIN	TAB
COS	MOD	TAN
CREAL	PI	TERMCHAR
CSING	POS	TIME\$
DATE\$	REC	VAL
EOF	RND	VALHEX
EXP	RIGHT\$	
FREESPACE	RPT\$	
INT		

APPENDIX B:

ASCII CODES

The following predefined characters may be printed or displayed on the screen.

ASCII CODE	CHARACTER	ASCII CODE	CHARACTER
30	(cursor)	63	? (question mark)
31	(space)	64	@ (at sign)
32	(space)	65	A
33	! (exclamation point)	66	B
34	" (quote)	67	C
35	# (number or pound sign)	68	D
36	\$ (dollar)	69	E
37	% (percent)	70	F
38	& (ampersand)	71	G
39	' (apostrophe)	72	H
40	((open parenthesis)	73	I
41) (close parenthesis)	74	J
42	* (asterisk)	75	K
43	+ (plus)	76	L
44	, (comma)	77	M
45	- (minus)	78	N
46	. (period)	79	O
47	/ (slash)	80	P
48	0	81	Q
49	1	82	R
50	2	83	S
51	3	84	T
52	4	85	U
53	5	86	V
54	6	87	W
55	7	88	X
56	8	89	Y
57	9	90	Z
58	: (colon)	91	[(open bracket)
59	; (semicolon)	92	\ (reverse slant)
60	< (less than)	93] (close bracket)
61	= (equals)	94	^ (exponentiation)
62	> (greater than)	95	_ (underline)

ASCII Codes (continued)

96	` (accent grave)	112	p
97	a	113	q
98	b	114	r
99	c	115	s
100	d	116	t
101	e	117	u
102	f	118	v
103	g	119	w
104	h	120	x
105	i	121	y
106	j	122	z
107	k	123	{ (left brace)
108	l	124	(vertical bar)
109	m	125	} (right brace)
110	n	126	~ (tilde)
111	o	127	DEL (appears as a blank)

When key unit = 3 or = 5, the following key presses may also be detected by CALL KEY.

1	Alt 7 (AID)	3	Alt 1 (DEL)
4	Alt 2 (INS)	6	Alt 8 (REDO)
7	Alt 3 (ERASE)	8	Alt S (LEFT ARROW)
9	Alt D (RIGHT ARROW)	10	Alt X (DOWN ARROW)
11	Alt E (UP ARROW)	12	Alt 6 (CMD)
13	ENTER	14	Alt 5 (BEGIN)
15	Alt 9 (BACK)		

APPENDIX C:

MUSICAL TONE FREQUENCIES

The following table gives the frequencies (rounded to integers) of four octaves of the tempered scale (one half step between notes). While this list does not represent the entire range of notes that the computer can produce, it can be helpful for programming music.

FREQUENCY	NOTE	FREQUENCY	NOTE
110	A	440	A (above middle C)
117	A#,Bb	466	A#,Bb
123	B	494	B
131	C (low C)	523	C (high C)
139	C#,Db	554	C#,Db
147	D	587	D
156	D#,Eb	622	D#,Eb
165	E	659	E
175	F	698	F
185	F#,Gb	740	F#,Gb
196	G	784	G
208	G#,Ab	831	G#,Ab
220	A (below middle C)	880	A (above high C)
220	A (above middle C)	880	A (above high C)
223	A#,Bb	932	A#,Bb
247	B	988	B
262	C (middle C)	1047	C
277	C#,Db	1109	C#,Db
294	D	1175	D
311	D#,Eb	1245	D#,Eb
330	E	1319	E
349	F	1397	F
370	F#,Gb	1480	F#,Gb
392	G	1568	G
415	G#,Ab	1661	G#,Ab
440	A (above middle C)	1760	A

APPENDIX D:

CHARACTER SETS

SET	ASCII CODES	SET	ASCII CODES
29	0-7	13	128-135
30	8-15	14	136-143
31	16-23	15	144-151
0	24-31	16	152-159
1	32-39	17	160-167
2	40-47	18	168-175
3	48-55	19	176-183
4	56-63	20	184-191
5	64-71	21	192-199
6	72-79	22	200-207
7	80-87	23	208-215
8	88-95	24	216-223
9	96-103	25	224-231
10	104-111	26	232-239
11	112-119	27	240-247
12	120-127	28	248-255

APPENDIX E:

PATTERN-IDENTIFIER CONVERSION TABLE

BLOCK	BINARY CODE (0=OFF; 1=ON)	HEXADECIMAL NOTATION
—	0000	0
— X	0001	1
— X	0010	2
— XX	0011	3
X —	0100	4
X X	0101	5
XX —	0110	6
XXX —	0111	7
X —	1000	8
X X	1001	9
X X	1010	A
X XX	1011	B
XX —	1100	C
XX X	1101	D
XXX —	1110	E
XXXX	1111	F

APPENDIX F:

COLOR CODES

COLOR	CODE	COLOR	CODE
Transparent	1	Medium Red	9
Black	2	Light Red	10
Medium Green	3	Dark Yellow	11
Light Green	4	Light Yellow	12
Dark Blue	5	Dark Green	13
Light Blue	6	Magenta	14
Dark Red	7	Gray	15
Cyan	8	White	16

APPENDIX G:

MATHEMATICAL FUNCTIONS

The following mathematical functions may be defined with DEF as shown.

Function	MYARC Extended BASIC II statement
Secant	DEF SEC(X)=1/COS(X)
Cosecant	DEF CSC(X)=1/SIN(X)
Cotangent	DEF COT(X)=1/TAN(X)
Inverse Sine	DEF ARCSIN(X)=ATN(S/SQR(1/X*X))
Inverse Cosine	DEF ARCCOS(X)=ATN(X/SQR(1/X*X))+PI/2
Inverse Secant	DEF ARCSEC(X)=ATN(SQR(X*X/1))+(SGN(X)/1)*PI/2
Inverse Cosecant	DEF ARCCSC(X)=ATN(1/SQR(X*X/1))+(SGN(X)-1)*PI/2
Inverse Cotangent	DEF ARCCOT(X)=PI/2-ATN(X) or =PI/2+ATN(-X)
Hyperbolic Sine	DEF SINH(X)=(EXP(X)-EXP(-X))/2
Hyperbolic Cosine	DEF COSH(X)=(EXP(X)+EXP(-X))/2
Hyperbolic Tangent	DEF TANH(X)=2*EXP(-X)/(EXP(X)+EXP(-X))+1
Hyperbolic Secant	DEF SECH=2/(EXP(X)+EXP(-X))
Hyperbolic Cosecant	DEF CSCH=2/(EXP(X)-EXP(-X))
Hyperbolic Cotangent	DEF COTH(X)=2*EXP(-X)/(EXP(X)-EXP(-X))+1
Inverse Hyperbolic Sine	DEF ARCSINH(X)=LOG(X+SQR(X*X+1))
Inverse Hyperbolic Cosine	DEF ARCCOSH(X)=LOG(X+SQR(X*X-1))
Inverse Hyperbolic Tangent	DEF ARCTANH(X)=LOG((1+X)/(1-X))/2
Inverse Hyperbolic Secant	DEF ARCSECH(X)=LOG((1+SQR(1-X*X))/X)
Inverse Hyperbolic Cosecant	DEF ARCCSCH(X)=LOG((SGN(X)*SQR(X*X+1)+1)/X)
Inverse Hyperbolic Cotangent	DEF ARCCOTH(X)=LOG((X+1)/(X-1))/2

APPENDIX H:

LIST OF SPEECH WORDS

The following is a list of all the letters, numbers, words, and phrases that can be accessed with CALL SAY and CALL SPGET. See Appendix M for instructions on adding suffixes to anything in this list.

/ (NEGATIVE)	CENTER	F
+ (POSITIVE)	CHECK	FIFTEEN
0	CLEAR	FIGURE
1	COLOR	FIND
2	COME	FINE
3	COMES	FINISH
4	COMMA	FINISHED
5	COMMAND	FIRST
6	COMPLETE	FIT
7	COMPLETED	FIVE
8	COMPUTER	FOR
9	CONNECTED	FORTY
A (a)	CONSOLE	FOUR
A1 ()	CORRECT	FOURTEEN
ABOUT	COURSE	FOURTH
AFTER	CYAN	FROM
AGAIN	D	FRONT
ALL	DATA	G
AM	DECIDE	GAMES
AN	DEVICE	GET
AND	DID	GETTING
ANSWER	DIFFERENT	GIVE
ANY	DISKETTE	GIVES
ARE	DO	GO
AS	DOES	GOES
ASSUME	DOING	GOING
AT	DONE	GOOD
B	DOUBLE	GOOD WORK
BACK	DOWN	GOODBYE
BASE	DRAW	GOT
BE	DRAWING	GRAY
BETWEEN	E	GREEN
BLACK	EACH	GUESS
BLUE	EIGHT	H
BOTH	EIGHTY	HAD
BOTTOM	ELEVEN	HAND
BUT	ELSE	HANDHELD UNIT
BUY	END	HAS
BY	ENDS	HAVE
BYE	ENTER	HEAD
C	ERROR	HEAR
CAN	EXACTLY	HELLO
CASSETTE	EYE	HELP

List of Speech Words (continued)

HERE	MEMORY	PRINTER
HIGHER	MESSAGE	PROBLEM
HIT	MESSAGES	PROBLEMS
HOME	MIDDLE	PROGRAM
HOW	MIGHT	PUT
HUNDRED	MODULE	PUTTING
HURRY	MORE	Q
I	MOST	R
I WIN	MOVE	RANDOMLY
IF	MUST	READ (read)
IN	N	READ1 (red)
INCH	NAME	READY TO START
INCHES	NEAR	RECORDER
INSTRUCTION	NEED	RED
INSTRUCTIONS	NEGATIVE	REFER
IS	NEXT	REMEMBER
IT	NICE TRY	RETURN
J	NINE	REWIND
JOYSTICK	NINETY	RIGHT
JUST	NO	ROUND
K	NOT	S
KEY	NOW	SAID
KEYBOARD	NUMBER	SAVE
KNOW	O	SAY
L	OF	SAYS
LARGE	OFF	SCREEN
LARGER	OH	SECOND
LARGEST	ON	SEE
LAST	ONE	SEES
LEARN	ONLY	SET
LEFT	OR	SEVEN
LESS	ORDER	SEVENTY
LET	OTHER	SHAPE
LIKE	OUT	SHAPES
LIKES	OVER	SHIFT
LINE	P	SHORT
LOAD	PART	SHORTER
LONG	PARTNER	SHOULD
LOOK	PARTS	SIDE
LOOKS	PERIOD	SIDES
LOWER	PLAY	SIX
M	PLAYS	SIXTY
MADE	PLEASE	SMALL
MAGENTA	POINT	SMALLER
MAKE	POSITION	SMALLEST
ME	POSITIVE	SO
MEAN	PRESS	SOME
	PRINT	SORRY

List of Speech Words (continued)

SPACE	THIRTEEN	WANT
SPACES	THIRY	WANTS
SPELL	THIS	WAY
SQUARE	THREE	WE
START	THREW	WEIGH
STEP	THROUGH	WEIGHT
STOP	TIME	WELL
SUM	TO	WERE
SUPPOSED	TOGETHER	WHAT
SUPPOSED TO	TONE	WHAT WAS THAT
SURE	TOO	WHEN
T	TOP	WHERE
TAKE	TRY	WHICH
TEEN	TRY AGAIN	WHITE
TELL	TURN	WHO
TEN	TWELVE	WHY
TEXAS INSTRUMENTS	TWENTY	WILL
THAN	TWO	WITH
THAT	TYPE	WON
THAT IS INCORRECT	U	WORD
THAT IS RIGHT	UHOH	WORDS
THE (the)	UNDER	WORK
THE1(th)	UNDERSTAND	WORKING
THEIR	UNTIL	WRITE
THEN	UP	X
THERE	UPPER	Y
THESE	USE	YELLOW
THEY	V	YES
THING	VARY	YET
THINGS	VERY	YOU
THINK	W	YOU WIN
THIRD	WAIT	YOUR
		Z
		ZERO

APPENDIX I:

ADDING SUFFIXES TO SPEECH WORDS

This appendix describes how to add ING, S, and ED to any word available in the Solid State Speech™ resident vocabulary.

The code for a word is first read using SPGET. The code consists of a number of characters, one of which tells the speech unit the length of the word. Then, by means of the subprograms listed here, additional codes can be added to give the sound of a suffix.

Words often have trailing-off data that make the word sound more natural but prevent the easy addition of suffixes. In order to add suffixes this trailing-off data must be removed.

The following program allows you to input a word and, by trying different truncation values, make the suffix sound like a natural part of the word. The subprograms DEFING (lines 1000 through 1130), DEFS1 (lines 2000 through 2100), DEFS2 (lines 3000 through 3090), DEFS3 (lines 4000 through 4120), DEFED1 (lines 5000 through 5070), DEFED2 (lines 6000 through 6110), DEFED3 (lines 7000 through 7130), and MENU (lines 10000 through 10120) should be input separately and saved with the MERGE option. (The subprogram MENU is the same one used in the illustrative program with SUB.) You may wish to use different line numbers. Each of these subprograms (except MENU) defines a suffix.

DEFING defines the ING sound. DEFS1 defines the S sound as it occurs at the end of "cats." DEFS2 defines the S sound as it occurs at the end of "cads." DEFS3 defines the S sound as it occurs at the end of "wishes." DEFED1 defines the ED sound as it occurs at the end of "passed." DEFED2 defines the ED sound as it occurs at the end of "caused." DEFED3 defines the ED sound as it occurs at the end of "heated."

In running the program, enter a 0 for the truncation value in order to leave the truncation sequence.

```

100 REM *****
110 REM REQUIRES MERGE OF:
120 REM MENU (LINES 10000 THROUGH 10120)
130 REM DEFING (LINES 1000 THROUGH 1130)
140 REM DEFS1 (LINES 2000 THROUGH 2100)
150 REM DEFS2 (LINES 3000 THROUGH 3090)
160 REM DEFS3 (LINES 4000 THROUGH 4120)
170 REM DEFED1 (LINES 5000 THROUGH 5070)
180 REM DEFED2 (LINES 6000 THROUGH 6110)
190 REM DEFED3 (LINES 7000 THROUGH 7130)

```

Adding Suffixes to Speech Words (continued)

```
200 REM *****
210 CALL CLEAR
220 PRINT "THIS PROGRAM IS USED TO"
230 PRINT "FIND THE PROPER TRUNCATION"
240 PRINT "VALUE FOR ADDING SUFFIXES"
250 PRINT "TO SPEECH WORDS.": :
260 FOR DELAY=1 TO 300::NEXT DELAY
270 PRINT "CHOOSE WHICH SUFFIX YOU"
280 PRINT "WISH TO ADD.": :
290 FOR DELAY=1 TO 800::NEXT DELAY
300 CALL MENU(8,CHOICE)
310 DATA 'ING','S' AS IN CATS,'S' AS IN CADS,'S' AS IN WISHES,
      'ED' AS IN PASSED,'ED' AS IN CAUSED,'ED' AS IN HEATED, END
320 IF CHOICE=0 OR CHOICE=8 THEN STOP
330 INPUT "WHAT IS THE WORD? ":WORD$
340 ON CHOICE GOTO 350,379,390,410,430,450,470
350 CALL DEFING(D$)
360 GOTO 480
370 CALL DEFS1(D$)!CATS
380 GOTO 480
390 CALL DEFS2(D$)!CADS
400 GOTO 480
410 CALL DEFS3(D$)!WISHES
420 GOTO 480
430 CALL DEFED1(D$)!PASSED
440 GOTO 480
450 CALL DEFED2(D$)!CAUSED
460 GOTO 480
470 CALL DEFED3(D$)!HEATED
480 REM TRY VALUES
490 CALL CLEAR
500 INPUT "TRUNCATE HOW MANY BYTES?":L
510 IF L=0 THEN 300
520 CALL SPGET(WORDS$,B$)
530 L=LEN(B$)-L-3
540 C$=SEG$(B$,1,2)&CHR$(L)&SEG$(B$,4,L)
550 CALL SAY(,C$&D$)
560 GOTO 500
```

Adding Suffixes to Speech Words (continued)

The data has been given in short DATA statements to make it as easy as possible to input. The data statements may be consolidated to make the program shorter.

```

1000 SUB DEFING(A$)
1010 DATA 96,0,52,174,30,65
1020 DATA 21,186,90,247,122,214
1030 DATA 179,95,77,13,202,50
1040 DATA 153,120,117,57,40,248
1050 DATA 133,173,209,25,39,85
1060 DATA 225,54,75,167,29,77
1070 DATA 105,91,44,157,118,180
1080 DATA 169,97,161,117,218,25
1090 DATA 119,184,227,222,249,238,1
1100 RESTORE 1010
1110 A$=""
1120 FOR I=1 TO 55::READ A::A$=A$&CHR$(A)::NEXT I
1130 SUBEND

```

```

2000 SUB DEFS1(A$)!CATS
2010 DATA 96,0,26
2020 DATA 14,56,130,204,0
2030 DATA 223,177,26,224,103
2040 DATA 85,3,252,106,106
2050 DATA 128,95,44,4,240
2060 DATA 35,11,2,126,16,121
2070 RESTORE 2010
2080 A$=""
2090 FOR I=1 TO 29::READ A::A$=A$&CHR$(A)::NEXT I
2100 SUBEND

```

```

3000 SUB DEFS2(A$)!CADS
3010 DATA 96,0,17
3020 DATA 161,253,158,217
3030 DATA 168,213,198,86,0
3040 DATA 223,153,75,128,0
3050 DATA 95,139,62
3060 RESTORE 3010
3070 A$=""
3080 FOR I=1 TO 20::READ A::A$=A$&CHR$(A)::NEXT I
3090 SUBEND

```


Adding Suffixes to Speech Words (continued)

```
4000 SUB DEFS3(A$)!WISHES
4010 DATA 96,0,34
4020 DATA 173,233,33,84,12
4030 DATA 242,205,166,55,173
4040 DATA 93,222,68,197,188
4050 DATA 134,238,123,102
4060 DATA 163,86,27,59,1,124
4070 DATA 103,46,1,2,124,45
4080 DATA 138,129,7
4090 RESTORE 4010
4100 A$=""
4110 FOR I=1 TO 37::READ A::A$=A$&CHR$(A)::NEXT I
4120 SUBEND
```

```
5000 SUB DEFED1(A$)!PASSED
5010 DATA 96,0,10
5020 DATA 0,224,128,37
5030 DATA 204,37,240,0,0
5040 RESTORE 5010
5050 A$=""
5060 FOR I=1 TO 13::READ A::A$=A$&CHR$(A)::NEXT I
5070 SUBEND
```

```
6000 SUB DEFED2(A$)!CAUSED
6010 DATA 96,0,26
6020 DATA 172,163,214,59,35
6030 DATA 109,170,174,68,21
6040 DATA 22,201,220,250,24
6050 DATA 69,148,162,166,234
6060 DATA 75,84,97,145,204
6070 DATA 15
6080 RESTORE 6010
6090 A$=""
6100 FOR I=1 TO 29::READ A::A$=A$&CHR$(A)::NEXT I
6110 SUBEND
```

Adding Suffixes to Speech Words (continued)

```
7000 SUB DEFED3(A$)!HEATED
7010 DATA 96,0,36
7020 DATA 173,233,33,84,12
7030 DATA 242,205,166,183
7040 DATA 172,163,214,59,35
7050 DATA 109,170,174,68,21
7060 DATA 22,201,92,250,24
7070 DATA 69,148,162,38,235
7080 DATA 75,84,97,145,204
7090 DATA 178,127
7100 DATA 7010
7110 A$=""
7120 FOR I=1 TO 39::READ A::A$=A$&CHR$(A)::NEXT I
7130 SUBEND
```

```
10000 SUB MENU(COUNT,CHOICE)
10010 CALL CLEAR
10020 IF COUNT>22 THEN PRINT "TOO MANY ITEMS" :: CHOICE=0 :: SUBEXIT
10030 RESTORE
10040 FOR I=1 TO COUNT
10050 READ TEMP$
10060 TEMP$=SEG$(TEMP$,1,25)
10070 DISPLAY AT (I,1):I;TEMP$
10080 NEXT I
10090 DISPLAY AT(I+1,1):"YOUR CHOICE: 1"
10100 ACCEPT AT(I+1,14)BEEP VALIDATE(DIGIT)SIZE(-2):CHOICE
10110 IF CHOICE<1 OR CHOICE>COUNT THEN 10100
10120 SUBEND
```

Adding Suffixes to Speech Words (continued)

You can use the subprograms in any program once you have determined the number of bytes to truncate. The following program uses the subprogram DEFING in lines 1000 through 1130 to have the computer say the word DRAWING using DRAW plus the suffix ING. Note that it was found that DRAW should be truncated by 41 characters to produce the most natural sounding DRAWING. The subprogram DEFING in lines 1000 through 1130 is the program you saved with the MERGE option.

```

100 CALL DEFING(ING$)
110 CALL SPGET("DRAW",DRAW$)
120 L=LEN(DRAW$)-3-41! 3 BYTES OF SPEECH OVERHEAD, 41 BYTES TRUNCATED
130 DRAW$=SEG$(DRAW$,1,2)&CHR$(L)&SEG$(DRAW$,4,L)
140 CALL SAY("WE ARE",DRAW$&ING$,"A1 SCREEN")
150 GOTO 140
1000 SUB DEFING(A$)
1010 DATA 96,0,52,174,30,65
1020 DATA 21,186,90,247,122,214
1030 DATA 179,95,77,13,202,50
1040 DATA 153,120,117,57,40,248
1050 DATA 133,173,209,25,39,85
1060 DATA 225,54,75,167,29,77
1070 DATA 105,91,44,157,118,180
1080 DATA 169,97,161,117,218,25
1090 DATA 119,184,227,222,249,238,1
1100 RESTORE 1010
1110 A$=""
1120 FOR I=1 TO 55::READ A::A$=A$&CHR$(A)::NEXT I
1130 SUBEND
(Press SHIFT C to stop the program.)

```

APPENDIX J:

ERROR MESSAGES

The following lists all the error messages that MYARC Advanced BASIC gives. The first list is alphabetical by the message that is given, and the second list is numeric by the number of the error that is returned by CALL ERR. If the error occurs in the execution of a program, the error message is often followed by IN line-number.

Sorted by Message

#	Message	Descriptions of Possible Errors
74	BAD ARGUMENT	<ul style="list-style-type: none"> * Bad value given in ASC, ATN, COS, EXP, INT, LOG, SIN, SOUND, SQR, TAN, or VAL. * An array element specified in a SUB statement. * Bad first parameter or too many parameters in LINK.
61	BAD LINE NUMBER	<ul style="list-style-type: none"> * Line number less than 1 or greater than 32767. * Omitted line number. * Line number outside the range 1 through 32767 produced by RES.
57	BAD SUBSCRIPT	<ul style="list-style-type: none"> * Use of too large or small subscript in an array. * Incorrect subscript in DIM.
79	BAD VALUE	<ul style="list-style-type: none"> * Incorrect value given in AND, CHAR, CHR\$, CLOSE, EOF, FOR, GOSUB, GOTO, HCHAR, INPUT, MOTION, NOT, OR, POS, PRINT, PRINT USING, REC, RESTORE, RPT\$, SEG\$, SIZE, VCHAR, or XOR. * Array subscript value greater than 32767. * File number greater than 255 or less than zero. * More than three tones and one noise generator specified in SOUND. * A value passed to a subprogram is not acceptable in the subprogram. For example, a sprite velocity value less than -128 or a character value greater than 143. * Value in ON...GOTO or ON...GOSUB greater than the number of lines given. * Incorrect position given after the AT clause in ACCEPT or DISPLAY.
67	CAN'T CONTINUE	<ul style="list-style-type: none"> * Program has been edited after being stopped by a breakpoint. * Program was not stopped by a breakpoint.
69	COMMAND ILLEGAL IN PROGRAM	<ul style="list-style-type: none"> * BYE, CON, LIST, MERGE, NEW, NUM, OLD, RES, or SAVE used in a program.
84	DATA ERROR	<ul style="list-style-type: none"> * READ or RESTORE with data not present or with a

- string where a number value is expected.
 - * Line number after RESTORE is higher than the highest line number in the program.
 - * Error in object file in LOAD.
- 109 FILE ERROR
- * Wrong type of data read with a READ statement.
 - * Attempt to use CLOSE, EOF, INPUT, OPEN, PRINT, PRINT USING, REC, or RESTORE with a file that does not exist or does not have the proper attributes.
 - * Not enough memory to use a file.
- 44 FOR-NEXT NESTING
- * The FOR and NEXT statements of loops do not align properly.
 - * Missing NEXT statement.
- 130 I/O ERROR
- * An error was detected in trying to execute CLOSE, DELETE, LOAD, MERGE, OLD, OPEN, RUN, or SAVE.
 - * Not enough memory to list a program.
- 16 ILLEGAL AFTER SUBPROGRAM
- * Anything but END, REM, or SUB after a SUBEND.
- 36 IMAGE ERROR
- * An error was detected in the use of DISPLAY USING, IMAGE, or PRINT USING.
 - * More than 10 (E-format) or 14 (numeric format) significant digits in the format string.
 - * IMAGE string is longer than 254 characters.
- 28 IMPROPERLY USED NAME
- * An illegal variable name was used in CALL, DEF, or DIM.
 - * Using a MYARC Advanced BASIC reserved word in LET.
 - * Using a subscripted variable or a string variable in a FOR.
 - * Using an array with the wrong number of dimensions.
 - * Using a variable name differently than originally assigned.
A variable can be only an array, a numeric or string variable, or a user defined function name.
 - * Dimensioning an array twice.
 - * Putting a user defined function name on the left of the equals sign in an assignment statement.
 - * Using the same variable twice in the parameter list of a SUB statement.
- 81 INCORRECT ARGUMENT LIST
- * CALL and SUB mismatch of arguments.
- 83 INPUT ERROR
- * An error was detected in an INPUT.
- 60 LINE NOT FOUND
- * Incorrect line number found in BREAK, GOSUB, GOTO, ON ERROR, RUN, or UNBREAK, or after THEN or ELSE.
 - * Line to be edited not found.

- 62 LINE TOO LONG
 - * Line too long to be entered into a program.
- 39 MEMORY FULL
 - * Program too large to execute one of the following: DEF, DELETE, DIM, GOSUB, LET, LOAD, ON...GOSUB, OPEN, or SUB.
 - * Program too large to add a new line, insert a line, replace a line, or evaluate an expression.
- 49 MISSING SUBEND
 - * SUBEND missing in a subprogram.
- 47 MUST BE IN SUBPROGRAM
 - * SUBEND or SUBEXIT not in a subprogram.
- 19 NAME TOO LONG
 - * More than 15 characters in variable or subprogram name.
- 43 NEXT WITHOUT FOR
 - * FOR statement missing, NEXT before FOR, incorrect FOR-NEXT nesting, or branching into a FOR-NEXT loop.
- 78 NO PROGRAM PRESENT
 - * No program present when issuing a LIST, RESEQUENCE, RESTORE, RUN, or SAVE command.
- 10 NUMERIC OVERFLOW
 - * A number too large or too small resulting from a *,+,-,/ operation or in ACCEPT, ATN, COS, EXP, INPUT, INT, LOG, SIN, SQR, TAN, or VAL.
 - * A number outside the range -32768 to 32767 in PEEK or LOAD.
- 70 ONLY LEGAL IN A PROGRAM
 - * One of the following statements was used as a command: DEF, GOSUB, GOTO, IF, IMAGE, INPUT, ON BREAK, ON ERROR, ON...GOSUB, ON...GOTO, ON WARNING, OPTION BASE, RETURN, SUB, SUBEND, or SUBEXIT.
- 25 OPTION BASE ERROR
 - * OPTION BASE executed more than once, or with a value other than 1 or zero.
- 97 PROTECTION VIOLATION
 - * Attempt to save, list, or edit a protected program.
- 48 RECURSIVE SUBPROGRAM CALL
 - * Subprogram calls itself, directly or indirectly.
- 51 RETURN WITHOUT GOSUB
 - * RETURN without GOSUB or an error handled by the previous execution of an ON ERROR statement.
- 56 SPEECH STRING TOO LONG
 - * Speech string returned by SPGET is longer than 255 characters.
- 40 STACK OVERFLOW
 - * Too many sets of parentheses.
 - * Not enough memory to evaluate an expression or assign a value.

54 STRING TRUNCATED

- * A string created by RPT\$, concatenation ("&" operator), or a user defined function is longer than 255 characters.
- * The length of a string expression in the VALIDATE clause is greater than 254 characters.

24 STRING-NUMBER MISMATCH

- * A string was given where a number was expected or vice versa in a MYARC Advanced BASIC supplied function or subprogram.
- * Assigning a string value to a numeric value or vice versa.
- * Attempting to concatenate ("&" operator) a number.
- * Using a string as a subscript.

135 SUBPROGRAM NOT FOUND

- * A subprogram called does not exist or an assembly language subprogram named in LINK has not been loaded.

14 SYNTAX ERROR

- * An error such as a missing or extra comma or parenthesis, parameters in the wrong order, missing parameters, missing keyword, misspelled keyword, keyword in the wrong order, or the like was detected in a MYARC Advanced BASIC command, statement, function, or subprogram.
- * DATA or IMAGE not first and only statement on a line.
- * Items after final ")".
- * Missing "#" in SPRITE.
- * Missing ENTER, tail comment symbol (!), or statement separator symbol (::).
- * Missing THEN after IF.
- * Missing TO after FOR.
- * Nothing after CALL, SUB, FOR, THEN, or ELSE.
- * Two E's in a numeric constant.
- * Wrong parameter list in a MYARC Advanced BASIC supplied subprogram.
- * Going into or out of a subprogram with GOTO, GOSUB, ON ERROR, etc.
- * Calling INIT without the Memory Expansion peripheral attached.
- * Calling LINK or LOAD without first calling INIT.
- * Using a constant where a variable is required.
- * More than seven dimensions in an array.

17 UNMATCHED QUOTES

- * Odd number of quotes in an input line.

20 UNRECOGNIZED CHARACTER

- * An unrecognized character such as ? or % is not in a quoted string.
- * A bad field in an object file accessed by LOAD.

Error Messages (concluded)

Sorted by #

#	Message
10	NUMERIC OVERFLOW
14	SYNTAX ERROR
16	ILLEGAL AFTER SUBPROGRAM
17	UNMATCHED QUOTES
19	NAME TOO LONG
20	UNRECOGNIZED CHARACTER
24	STRING-NUMBER MISMATCH
25	OPTION BASE ERROR
28	IMPROPERLY USED NAME
36	IMAGE ERROR
39	MEMORY FULL
40	STACK OVERFLOW
43	NEXT WITHOUT FOR
44	FOR-NEXT NESTING
47	MUST BE IN SUBPROGRAM
48	RECURSIVE SUBPROGRAM CALL
49	MISSING SUBEND
51	RETURN WITHOUT GOSUB
54	STRING TRUNCATED
56	SPEECH STRING TOO LONG
57	BAD SUBSCRIPT
60	LINE NOT FOUND
61	BAD LINE NUMBER
62	LINE TOO LONG
67	CAN'T CONTINUE
69	COMMAND ILLEGAL IN PROGRAM
70	ONLY LEGAL IN A PROGRAM
74	BAD ARGUMENT
78	NO PROGRAM PRESENT
79	BAD VALUE
81	INCORRECT ARGUMENT LIST
83	INPUT ERROR
84	DATA ERROR
97	PROTECTION VIOLATION
109	FILE ERROR
130	I/O ERROR
135	SUBPROGRAM NOT FOUND

APPENDIX K

GRAPHICS MODES - Summary

GRAPHICS MODE	SCREEN DIMEN.	SCREEN SIZE	DEFAULT MARGINS	MODE NAME	NO. OF PATNS.	COLORS PER SCREEN	PATTERN SIZE	SPRITE MODE	MEMORY/SCREEN
1,1	256,192	32x24	3,30 1,24	Pattern Graphic1	256	16	8 x 8	1 4/line	4K/scr 32 pgs
1,2	256,192	32x24	3,30 1,24	Graphic2	768	16	8 x 8	1 4/line	16K/scr 8 pgs
1,3	256,192	32x24	3,30 1,24	Graphic3	768	16	8 x 8	2 8/line	16K/scr 8 pgs
2,1	256,192	40x24	1,40 1,24	Text- 1	256	2	6 x 8	None	4K/scr 3 pgs
2,2	256x212	40x26	1,40 1,24	Bitmap-1	???	16	6 x 8	2 8/line	32K/scr 4 pgs
2,3	256,212	40x26	1,40 1,24	Bitmap-4	???	256	6 x 8	2 8/line	64K/scr 2 pgs
3,1	512,212	80x26	1,80 1,24	Text- 2	256	2+2	6 x 8	None	8K/scr 16 pgs
3,2	512x212	80x26	1,80 1,24	Bitmap-2	???	4	6 x 8	2 8/line	32K/scr 4 pgs
3,3	512x212	80x26	1,80 1,24	Bitmap-3	???	16	6 x 8	2 8/line	64K/scr 2 pgs

APPENDIX L

PROGRAM - ILLUSTRATING MOUSE COMMANDS

The following program illustrates the use of several MOUSE Commands to draw lines on the screen. Press MOUSE button 1 to start drawing a line and hold it down until you are done drawing.

```
100 CALL GRAPHICS(2,3) :: REM 256 COLOR BIT MAPPED MODE
110 CALL SPRITE(#1,33,16,1,1) :: REM DEFINE MOUSE AS !
120 CALL SEEMOUSE :: REM MAKE SURE MOUSE IS VISIBLE ON SCREEN
130 CALL MOUSE(Y,1) :: REM TEST FOR BUTTON PRESS
140 IF Y=0 THEN 130 :: REM WAIT FOR A BUTTON PRESS
150 CALL MOUSEDRAG(ON) :: REM BUTTON PRESSED SO START DRAWING
160 CALL MOUSE(Y,1) :: REM TEST BUTTON STATUS
170 IF Y=1 THEN 160 :: REM DRAW UNTIL RELEASED
180 CALL MOUSEDRAG(OFF) :: REM STOP DRAWING WHEN RELEASED
190 GO TO 130 :: REM GO TO WAIT FOR NEXT BUTTON PRESS
```

APPENDIX M:

ADDITIONAL EXTENDED ASCII CODES FOR KEYBOARD MODE 6

In addition to the normal ASCII codes returned in keyboard mode 5, the following additional Extended Codes are also returned in keyboard mode 6:

EXTENDED CODE(HEX)	FUNCTION
3	NUL Character
F	Back Arrow
10-19	ALT Q,W,E,R,T,Y,U,I,O,P
1E-26	ALT A,S,D,F,G,H,J,K,L
2C-32	ALT Z,X,C,V,B,N,M
3B-44	F1-F10 Function Keys (Base Case)
47	Home
48	Up Arrow
49	Page Up
4B	Left Arrow
4D	Right Arrow
4F	End
50	Down Arrow
51	Page Down
52	INS
53	DEL
54-5D	F11-F20 (Upper Case F1-F10)
5E-67	F21-F30 (CTRL F1-F10)
68-71	F31-F40 (ALT F1-F10)
72	CTRL PRTSC(Start/Stop Echo to Printer)
73	CTRL Right Arrow (Reverse Word)
74	CTRL Left Arrow (Advance Word)
75	CTRL END (Erase to End of Line)
76	CTRL PG DN (Erase to End of Screen)
77	CTRL HOME (Clear Screen and Home)
78-83	ALT 1,2,3,4,5,6,7,8,9,0,-,=
84	CTRL PG UP (Top 25 Lines of Text and Home Cursor)

- continue on next page -

The following combinations of keys produce special effects:

- o ALT CTRL DEL makes the keyboard routine initiate the equivalent of a system reset/reboot.
- o CTRL BREAK makes the keyboard routine invoke the (Keyboard break) interrupt.
- o CTRL NUM-LOCK makes the keyboard routine wait for you to press any key but NUM-LOCK. This gives you a way to suspend an operation temporarily, then resume.
- o SHIFT PRNESC makes the keyboard routine invoke the Print Screen Interrupt.
- o The keyboard treats the following keys as a group, rather than individually: CTRL, SHIFT, NUM-LOCK, CAPS-LOCK, and INS. The service routine for the Keyboard I/O interrupt returns a "shift status" byte that tells you when one of these keys are pressed.